

## 前 言

水是人类生活与生产劳动所必需的物质。水文学是研究自然界各水体的运动、变化和分布规律的一门学科。随着工、农业的迅速发展和人口的急剧增长,人们对水的需求日益增加,因此对水文学的要求也就更高。

计算机是一门新生科学,其 FORTRAN 语言是目前国内外科技计算中使用最广泛的一种计算语言。计算机技术已广泛用于水文领域,在解决水文实际问题和进行水文规律研究上都起着重要的作用,大大促进了水文学学科的发展。

为了使计算机技术更多更好地在水文领域发挥作用,便于实现联机预报,提高洪水预报的精度和增长洪水预报的有效预见期,为水利水电工程的设计、调度、管理提供方便,特编著《常用水文预报算法和计算程序》一书。

该书主要介绍一些国内常用的水文预报算法,包括流域降雨径流模型、纳须时段单位线、河道汇流模型、水库调洪计算等,并提供了它们相应的 FORTRAN 语言子程序。为了使程序具有灵活和通用性,在程序中没有使用任何库软件,语言尽可能简化和通俗。可用 FORTRAN66 或 FORTRAN77 在各类小型机和微机上编译运行。

本书共分十章,第一章至第五章分别介绍了国内常用的几种流域降雨径流模型,它们是河海大学的新安江(三水源)

模型与陕北模型、美国的萨克拉门托模型、日本的水箱模型,以及降雨径流经验相关( $P \sim P_a \sim R$ )模型。以上五个模型的相应程序都是一个单元面积上的降雨径流计算模型,输入是单元面积上的平均降雨量和蒸发能力序列,输出除了  $P \sim P_a \sim R$  是单元面积上的净雨过程外,其它都是单元出口处的流量过程。各程序都是遵照原文编制的,并且尽可能使之容易调用。不过,新安江(三水源)模型的三水源划分有别于参考文献[1],这是因为参考文献[1]中水源划分公式有误,本书作了改正。陕北模型中采用了流域下渗能力分布曲线,使之可以用于较大的流域,并具有实用价值。第六章是水库调洪计算。它是由水库入库流量序列,根据初始库水位和水库在一次洪水期间的实际或计划泄水建筑物启闭情况,进行水库调洪计算,求出该调度方式下的水库出库流量、库水位、库蓄水量序列。该程序适用于水库的实时调度。第七章是马斯京根法。主要介绍马斯京根法(以下简称马法)在国内几种不同的应用形式,即基本马法、分段马法、变参数马法、非线性马法、分层马法,以及各方法的特点和相互关系,并分别给出了各种形式的计算程序。第八章和第九章分别是纳须单位线法和反馈模拟实时校正及它们相应的计算程序。第十章给出了直线插值、抛物线插值、水位与流量的换算、单位线推流计算、时段单位线的转换、马斯京根演算参数  $K$  与  $x$  及汇流系数的计算、误差统计与分析、等时段流量(或水位、雨量)序列的生成与打印过程线、径流量与输沙量的计算、流量序列的特征值统计等 13 种其它常用的水文计算子程序。

为了便于使用,前九章除了包括方法介绍、计算程序及说

明外,还分别给出了计算实例。

河海大学教授芮孝芳、水利部水利信息中心高级工程师刘金清、黄河水利委员会教授级高工辛国荣对该书进行审阅,避免了疏漏;冯立亚同志在本书的编写中做了大量的工作;本书的出版得到黄河水利委员会及委河务局等单位领导和周围同志们的大力支持,在此一并表示衷心感谢。

由于作者水平所限,不妥之处在所难免,恳请读者批评指正。

翟家瑞

1994年6月

# 目 录

## 前 言

<b>第一章 新安江(三水源)模型</b> .....	(1)
第一节 模型结构 .....	(1)
一、蒸散发计算 .....	(1)
二、产流量计算 .....	(4)
三、水源划分 .....	(5)
四、汇流计算 .....	(8)
五、不同时段长度的转化 .....	(12)
第二节 模型计算程序 .....	(13)
一、计算程序 .....	(13)
二、程序说明 .....	(18)
第三节 实例 .....	(20)
一、主程序 .....	(20)
二、实例说明 .....	(21)
<b>第二章 萨克拉门托模型</b> .....	(23)
第一节 模型结构 .....	(23)
一、产流量计算 .....	(23)
二、蒸散发计算 .....	(26)
三、水分交换计算 .....	(27)
四、下渗计算 .....	(27)
五、模型参数的意义 .....	(29)
六、几个主要参数对洪水的影响 .....	(30)
第二节 模型计算程序 .....	(30)

一、计算程序 .....	(31)
二、程序说明 .....	(38)
第三节 实例 .....	(39)
一、主程序 .....	(39)
二、实例说明 .....	(40)
<b>第三章 水箱模型 .....</b>	<b>(42)</b>
第一节 模型结构 .....	(42)
第二节 模型计算程序 .....	(45)
一、计算程序 .....	(46)
二、程序说明 .....	(47)
第三节 实例 .....	(49)
一、主程序 .....	(49)
二、实例说明 .....	(50)
<b>第四章 陕北模型 .....</b>	<b>(52)</b>
第一节 模型结构 .....	(52)
一、点降雨径流模型 .....	(52)
二、降雨和下垫面空间分布不均匀的考虑 .....	(55)
三、陕北模型推流步骤 .....	(56)
第二节 模型计算程序 .....	(58)
一、计算程序 .....	(58)
二、程序说明 .....	(61)
第三节 实例 .....	(62)
一、主程序 .....	(62)
二、实例说明 .....	(63)
<b>第五章 降雨径流相关图法 .....</b>	<b>(65)</b>
第一节 降雨径流相关图 .....	(65)
一、概述 .....	(65)
二、 $P_s$ 的计算 .....	(66)
三、相关图推流计算 .....	(66)
第二节 $P \sim P_s \sim R$ 计算程序 .....	(68)
一、 $P \sim P_s \sim R$ 程序 .....	(68)

二、程序说明 .....	(71)
第三节 实例 .....	(72)
一、主程序 .....	(72)
二、实例说明 .....	(73)
<b>第六章 水库调洪演算 .....</b>	<b>(76)</b>
第一节 计算方法 .....	(76)
第二节 水库调洪计算程序 .....	(78)
一、计算程序 .....	(78)
二、程序说明 .....	(80)
第三节 实例 .....	(81)
一、主程序 .....	(81)
二、实例说明 .....	(83)
<b>第七章 马斯京根法 .....</b>	<b>(86)</b>
第一节 基本马法 .....	(86)
第二节 分段马法 .....	(87)
一、方法介绍 .....	(87)
二、分段马法计算程序 .....	(87)
三、实例 .....	(89)
第三节 变参数马法 .....	(91)
一、方法介绍 .....	(91)
二、变参数马法计算程序 .....	(92)
三、实例 .....	(94)
第四节 非线性马法 .....	(96)
一、非线性马法计算公式 .....	(96)
二、非线性马法计算程序 .....	(97)
三、实例 .....	(101)
第五节 分层马法 .....	(103)
一、方法简介 .....	(103)
二、方法特点 .....	(104)
三、分层马法计算程序 .....	(106)
四、实例 .....	(109)

第六节 讨论 .....	(110)
一、各方法间的关系 .....	(111)
二、区间来(引)水的处理 .....	(111)
三、马法的预见期和预报误差 .....	(111)
<b>第八章 纳须单位线法 .....</b>	<b>(114)</b>
第一节 基本概念 .....	(114)
一、纳须瞬时单位线 .....	(114)
二、纳须时段单位线 .....	(114)
三、纳须单位线 $n$ 、 $k$ 的计算 .....	(116)
第二节 计算程序 .....	(118)
一、求 $n$ 和 $k$ 的计算程序 .....	(118)
二、纳须时段单位线的计算程序 .....	(120)
第三节 实例 .....	(121)
<b>第九章 反馈模拟实时校正 .....</b>	<b>(124)</b>
第一节 模型简介 .....	(124)
第二节 实时校正计算程序 .....	(126)
一、计算程序 .....	(126)
二、程序说明 .....	(128)
第三节 实例 .....	(129)
<b>第十章 其它常用水文计算子程序 .....</b>	<b>(131)</b>
一、直线插值 .....	(131)
二、抛物线插值 .....	(132)
三、水位与流量的换算 .....	(135)
四、单位线推流计算 .....	(137)
五、时段单位线的转换 .....	(139)
六、马斯京根演算参数 $K$ 与 $x$ 的计算 .....	(141)
七、马斯京根汇流系数的计算 .....	(142)
八、误差统计与分析 .....	(144)
九、等时段流量(或水位)序列的生成 .....	(147)
十、等时段雨量序列的生成 .....	(150)
十一、打印过程线 .....	(155)



十二、径流量、输沙量的计算 .....	(159)
十三、流量序列的特征值统计 .....	(165)
<b>附录 黄河洪水预报及调度系统简介 .....</b>	<b>(171)</b>
一、系统综述 .....	(171)
二、数据库及其管理 .....	(171)
三、方法库及其管理 .....	(172)
四、模型库及其管理 .....	(173)
五、三库间的关系 .....	(176)
<b>参考文献 .....</b>	<b>(178)</b>



# 第一章 新安江(三水源)模型

## 第一节 模型结构

新安江模型是河海大学(原华东水利学院)水文系 1973 年对新安江水库作入库流量预报时提出来的,是一个分布式的概念性流域降雨径流模型。10 多年来在我国湿润地区和半湿润地区多有应用,并于 80 年代中期发展改进成为新安江(三水源)模型。

本节仅介绍作一个单元面积降雨径流计算的新安江(三水源)模型。其相应的程序流程图如图 1-1 所示。下面就图中各部分的处理方法和计算公式分别作介绍。

### 一、蒸散发计算

新安江(三水源)模型中的蒸散发计算采用的是三层蒸发计算模式,它的输入是蒸发器实测水面蒸发和流域蒸散发能力的折算系数  $K$ ,模型参数是上、下、深三层的蓄水容量  $W_{UM}$ 、 $W_{LM}$ 、 $W_{DM}$  ( $W_M = W_{UM} + W_{LM} + W_{DM}$ ) 和深层蒸散发系数  $C$ 。输出是上、下、深各层的流域蒸散发量  $EU$ 、 $EL$ 、 $ED$  ( $E = EU + EL + ED$ )。计算中包括三个时变参数,即各层土壤含水量  $W_U$ 、 $W_L$ 、 $W_D$  ( $W = W_U + W_L + W_D$ )。 $W_M$ 、 $E$ 、 $W$  分别表示总的土壤蓄水容量、蒸散发量和土壤含水量。

各层蒸散发的计算原则是,上层按蒸散发能力蒸发,上层含水量不够蒸发时,剩余蒸散发能力从下层蒸发,下层蒸发与剩余蒸散发能力及下层含水量成正比,与下层蓄水容量成反比。要求计算的下层蒸发量与剩余蒸散发能力之比不小于深层蒸散发系数  $C$ ,否则,不足部分由下层含水量补给,当下层水量不够补给时,




图1-1 新安江 (三水源) 模型流程图




图1-2 蒸汽发计算程序框图




图1-3 流域蓄水容量曲线

用深层含水量补。三层蒸散发  
计算程序框图见图1-2。其中  
 $PE = P - K \cdot EM$ ， $P$  是降雨。

## 二、产流量计算

产流量计算系根据蓄满产  
流理论得出的。所谓蓄满，是指  
包气带的含水量达到田间持水  
量。在土壤湿度未达到田间持  
水量时不产流，所有降雨都被  
土壤吸收，成为张力水。而当土

壤湿度达到田间持水量后，所有降雨(减去同期蒸发)都产流。

上面是对流域上某一点而言的，一般说来，流域内各点的蓄  
水容量并不相同。新安江(三水源)模型把流域内各点的蓄水容量  
概化成如图1-3所示的一条抛物曲线。用  $W'_{mm}$  表示流域内最大的  
点蓄水容量， $W'_m$  表示流域内某一点的蓄水容量， $f$  表示蓄水能  
力  $\leq W'_m$  值的流域面积， $F$  表示流域面积， $B$  表示抛物线指数。其  
公式为

$$\frac{f}{F} = 1 - \left(1 - \frac{W'_m}{W'_{mm}}\right)^B \quad (1-1)$$

流域平均蓄水容量为

$$WM = \int_0^{W'_{mm}} \left(1 - \frac{f}{F}\right) dW'_m = \frac{W'_{mm}}{B+1} \quad (1-2)$$

与流域初始平均蓄水量( $W_0$ )相应的纵坐标( $A$ )为

$$A = W'_{mm} \left[1 - \left(1 - \frac{W_0}{WM}\right)^{\frac{1}{B+1}}\right] \quad (1-3)$$

当  $PE > 0$ ，则产流；否则不产流。产流时，

当  $PE + A < W'_{mm}$

$$R = PE - WM + W_0 + WM \left[ 1 - \frac{PE + A}{W'_{mm}} \right]^{B+1} \quad (1-4)$$

当  $PE + A \geq W'_{mm}$

$$R = PE - (WM - W_0) \quad (1-4')$$

作产流计算时,模型的输入为  $PE$ ,参数包括流域平均蓄水容量  $WM$  和抛物线指数  $B$ ;输出为流域产流量  $R$  及流域时段末平均含水量  $W$ 。

### 三、水源划分

新安江(三水源)模型采用一个自由水蓄水库进行水源划分,自由水蓄水库设置两个出口,其出流系数分别为  $KSS$  和  $KG$ 。产流量  $R$  进入自由水水库内,通过两个出流系数和溢流的方式把它分成地面径流( $RS$ )、壤中流( $RSS$ )和地下径流( $RG$ )。自由水蓄水库的结构见图1-4。




图1-4 自由水蓄水库的结构

自由水的蓄水能力在产流面积( $FR$ )上的分布也是不均匀的。模仿张力水分布不均匀的处理方式,把自由水蓄水能力在产流面积上的分布也用一条抛物线来表示(见图1-5)。用  $SMMF$  表示产流面积上最大一点的自由水蓄水容量,  $SMF$  表示产流面积上的自由水平均蓄水容量深,  $SMF'$  表示产流面积上某一点的自由水容量,  $FS$  表示自由水蓄水能力  $\leq SMF'$  值的流域面积占产流面积( $FR$ )的百分数。 $S$  表示自由水在产流面积上的平均蓄水深,  $EX$  表示流域自由水蓄水容量曲线的指数, 产流面积上各点的自由水蓄水容量关系为




图1-5 流域自由水蓄水容量曲线

$$\frac{FS}{FR} = 1 - \left(1 - \frac{SMF'}{SMMF}\right)^{EX} \quad (1-5)$$

产流面积上的平均蓄水容量深( $SMF$ )为

$$SMF = \frac{SMMF}{1 + EX} \quad (1-6)$$

与  $S$  对应的纵坐标( $AU$ )为

$$AU = SMMF \left[1 - \left(1 - \frac{S}{SMF}\right)^{\frac{1}{1+EX}}\right] \quad (1-7)$$

当  $PE + AU < SMMF$  时,地面径流量( $RS$ )为

$$RS = FR \cdot \int_{AU}^{PE+AU} \frac{FS}{FR} dSMF'$$

$$= FR \cdot \{PE - SMF + S + SMF [1 - \frac{(PE + AU)}{SMMF}]^{EX+1}\}$$

(1-8)

当  $PE + AU \geq SMMF$

$$RS = FR \cdot (PE + S - SMF) \quad (1-8')$$

显然,  $SMMF$  和  $SMF$  都是产流面积( $FR$ )的函数,是变化的,无法确定。我们假定  $SMMF$  与产流面积( $FR$ )及全流域上最大点的自由水蓄水容量( $SMM$ )的关系仍为抛物线分布

$$FR = 1 - (1 - \frac{SMMF}{SMM})^{EX} \quad (1-9)$$

则

$$SMMF = [1 - (1 - FR)^{\frac{1}{EX}}] SMM \quad (1-10)$$

$$SMM = SM \cdot (1 + EX) \quad (1-11)$$

在用式(1-5)~(1-8)进行计算时,必须首先用式(1-10)和式(1-11)计算出  $SMMF$  来。而流域的平均自由水容量  $SM$  和  $EX$  对一个流域来说是固定的,属模型率定参数。

已知上时段的产流面积( $FR0$ )和产流面积上的平均自由水深( $S0$ ),根据时段产流量( $R$ ),计算时段地面径流、壤中流、地下径流及本时段产流面积( $FR$ )和  $FR$  上的平均自由水深( $S$ )的步骤是

$$FR = R / PE$$

$$S = S0 \cdot FR0 / FR$$

$$SMM = SM \cdot (1 + EX)$$

$$SMMF = SMM \cdot [1 - (1 - FR)^{1/EX}]$$

$$SMF = SMMF / (1 + EX)$$

$$AU = SMMF \cdot [(1 - (1 - S / SMF)^{1/(1+EX)})]$$



(1) 当  $PE + AU \leq 0$  时

$$RS = 0$$

$$RSS = 0$$

$$RG = 0$$

$$S = 0$$

(2) 当  $PE + AU \geq SMMF$  时

$$RS = (PE + S - SMF) \cdot FR$$

$$RSS = SMF \cdot KSS \cdot FR$$

$$RG = SMF \cdot KG \cdot FR$$

$$S = SMF - (RSS + RG) / FR$$

(3) 当  $0 < PE + AU < SMMF$  时

$$RS = \{PE - SMF + S + SMF[1 - (PE + AU) / SMMF]^{(EX+1)}\} \cdot FR$$

$$RSS = (PE + S - RS / FR) \cdot KSS \cdot FR$$

$$RG = (PE + S - RS / FR) \cdot KG \cdot FR$$

$$S = S + PE - (RS + RSS + RG) / FR$$


在自由水蓄水库的计算中, 存在差分计算的误差问题。为了消除这些影响, 采用5mm 净雨分一段, 进一步分段计算。其程序的计算流程见图1—6。

#### 四、汇流计算

流域汇流计算包括坡地和河网两个汇流阶段。

坡地汇流是指水体在坡面上的汇集过程。在该汇流阶段, 水流不但发生水平运动, 而且还有垂向运动。在流域的坡面上, 地面径流的调蓄作用不大, 地下径流受到大的调蓄, 壤中流所受调蓄介于二者之间。

河网汇流是指水流由坡面进入河槽后, 继续沿河网的汇集过程。在河网汇流阶段, 汇流特性受制于河槽水力学条件, 各种水源






图1-6 新安江(三水源)模型水源划分计算框图

是一致的。新安江(三水源)模型中的河网汇流,仅指各单元面积上的水体从进入河槽汇至单元出口的过程。而不包括单元出口到流域出口处的河网汇流阶段。

### 1. 坡地汇流计算

新安江(三水源)模型中把经过水源划分得到的地面径流直接进入河网,成为地面径流对河网的总入流( $TRS$ )。壤中流( $RSS$ )流入壤中流水库,经过壤中流蓄水库的消退(壤中流水库的消退系数为  $KKSS$ ),成为壤中流对河网的总入流( $TRSS$ )。地下径流( $RG$ )进入地下水蓄水库,经过地下水蓄水库的消退(地下水蓄水库的消退系数为  $KKG$ ),成为地下水对河网的总入流( $TRG$ )。其计算公式为

$$TRS(t) = RS(t) \cdot U$$

$$TRSS(t) = TRSS(t-1) \cdot KKSS + RSS(t) \cdot (1 - KKSS) \cdot U$$

$$TRG(t) = TRG(t-1) \cdot KKG + RG(t) \cdot (1 - KKG) \cdot U$$

$$TR(t) = TRS(t) + TRSS(t) + TRG(t)$$

式中,  $U$  为单位转换系数,可将径流深转化成流量,  $U = \frac{F}{3.6 \cdot \Delta t}$ , 其中  $F$  为流域面积,  $\text{km}^2$ ;  $\Delta t$  为时段长,  $\text{h}$ ;  $TR$  为河网总入流,  $\text{m}^3/\text{s}$ 。

### 2. 河网汇流计算

新安江(三水源)模型中用无因次单位线模拟水体从进入河槽到单元出口的河网汇流。在本流域或邻近流域,找一个有资料的、面积与单元流域大体相近的流域,分析出地面径流单位线,就可作初值应用。

计算公式为

$$Q(t) = \sum_{i=1}^N UH(i) \cdot TR(t-i+1)$$

式中,  $Q(t)$  为单元出口处  $t$  时刻的流量值;  $UH$  为无因次时段单位线;  $N$  为单位线的历时时段数。其它符号意义同前。

流域汇流计算的输入是单元上的地面径流( $RS$ )、壤中流( $RSS$ )、地下径流( $RG$ )及计算开始时的单元面积上的壤中流流量和地下径流流量值。参数包括壤中流水库的日消退系数( $K'KSS$ )、地下水蓄水库日消退系数( $KKG$ )及单位线转换系数( $U$ )、无因次单位线( $UH$ )和历时( $N$ )。输出为单元出口的流量过程。

## 五、不同时段长度的转化

一般情况下,模型的出流系数和消退系数都是按日模型给定的。但是,进行实时洪水预报时,计算时段 $\Delta t$ 都小于24h。另外,在模型计算中,为了消除非线性影响,减少计算时段取的过长所引起的误差,模型的水源划分中又用5mm净雨作一个量级,进一步作分步长计算。那么,模型的出流系数和消退系数都必须作相应的变化。

设模型计算所取时段长为 $\Delta t(h)$ , $R$ 为 $\Delta t$ 内的净雨,

则

$$M = \frac{24}{\Delta t}$$

$$N = \frac{24}{\Delta t} \cdot [\text{INT}(\frac{R}{5}) + 1]$$

计算步长内的壤中流蓄水库的消退系数( $KKSSD$ )和地下水蓄水库的消退系数( $KKGD$ )分别与其相应的日模型消退系数( $KKSS$ )和( $KKG$ )的关系为

$$KKSSD = KKSS^{\frac{1}{M}}$$

$$KKGD = KKG^{\frac{1}{M}}$$

计算步长内流域自由水蓄水库的壤中流出流系数( $KSSD$ )和地下水出流系数( $KGD$ )与其日模型的出流系数( $KSS$ )和 $KG$ 的关系为

$$KSSD = \frac{1 - [1 - (KG + KSS)]^{\frac{1}{N}}}{1 + \frac{KG}{KSS}}$$

$$KGD = KSSD \cdot \frac{KG}{KSS}$$

## 第二节 模型计算程序

对于一个大的流域，为了考虑降水和下垫面条件的不均匀分布，一般把它分成许多单元，先用新安江（三水源）模型对每个单元进行产汇流计算，求出各单元的出流过程，然后根据单元出口至流域出口的距离和河槽的水力特性，通常用分段马斯京根法把各单元的出流量演进至流域出口处，再作线性叠加。这里的新安江（三水源）模型程序仅是单元流域上的产汇流计算，不包括单元出口至流域出口的河道演进部分。以后介绍的萨克拉门托模型、陕北模型等都是如此。

### 一、计算程序

```

SUBROUTINE XAJ(N,M,PAR,AREA,UH,DT,P,EP,QR,
+ W,FR,S,QRSS0,QRG0)
REAL * 4 W(3),WM(3),E(3),KC,IMP
REAL * 4 KSSD,KGD,KG,KSS,KKSS,KKG
REAL * 4 P(N),EP(N),UH(M),QR(N),PAR(13)
INTEGER D
DO 3 I=1,3
3 WM(I)=PAR(I)
KC=PAR(4)
C=PAR(5)
B=PAR(6)
IMP=PAR(7)
SM=PAR(8)

```

```

EX=PAR(9)
KG=PAR(10)
KSS=PAR(11)
KKG=PAR(12)
KKSS=PAR(13)
DO 5 I=1,N
5 QR(I)=0.
U=AREA/(DT*3.6)
IF(DT.LE.24) THEN
D=24/DT
CI=KKSS** (1/REAL(D))
CG=KKG** (1/REAL(D))
KSSD=(1-(1-(KG+KSS))** (1.0/REAL(D)))/(1+
+ KG/KSS)
KGD=KSSD*KG/KSS
ELSE
WRITE(*,'(A50\)' )'      所取计算时段长不合适!!!'
STOP
ENDIF
C


---


DO 150 I=1,N
IF(EP(I).LT.0) EP(I)=0.
IF(P(I).LT.0) P(I)=0.
EP(I)=EP(I)*KC
C


---


WM0=WM(1)+WM(2)+WM(3)
W0=W(1)+W(2)+W(3)
PE=P(I)-EP(I)
R=0.
RIMP=0.
IF(PE.LE.0) GOTO 100

```



```

WMM = (1. + B) * WM0 / (1. - IMP)
IF ((WM0 - W0). LE. 0. 0001) THEN
A = WMM
ELSE
A = WMM * (1. - (1. - W0 / WM0) * * (1. / (1. + B)))
ENDIF
IF ((PE + A). LT. WMM) THEN
R = PE - WM0 + W0 + WM0 * ((1 - (PE + A) / WMM) * * (1 + B))
ELSE
R = PE - (WM0 - W0)
ENDIF
RIMP = PE * IMP
100 CONTINUE
C
IF ((W(1) + P(I)). GT. EP(I)) THEN
E(1) = EP(I)
E(2) = 0. 0
E(3) = 0. 0
ELSE
E(1) = W(1) + P(I)
E(2) = (EP(I) - E(1)) * W(2) / WM(2)
IF (W(2). LE. (C * WM(2))) THEN
E(2) = C * (EP(I) - E(1))
E(3) = 0. 0
IF (W(2). GE. C * (EP(I) - E(1))) THEN
E(2) = C * (EP(I) - E(1))
E(3) = 0.
ELSE
E(2) = W(2)
E(3) = C * (EP(I) - E(1)) - E(2)
ENDIF
ENDIF

```

```

      ENDIF
      ENDIF
10    CONTINUE
C    -----
      W(1)=W(1)+P(1)-R-E(1)
      W(2)=W(2)-E(2)
      W(3)=W(3)-E(3)
      IF(W(1).GT.WM(1)) THEN
      W(2)=W(1)-WM(1)+W(2)
      W(1)=WM(1)
      IF(W(2).GT.WM(2)) THEN
      W(3)=W(3)+W(2)-WM(2)
      W(2)=WM(2)
      ENDIF
      ENDIF
      X=FR
      IF(PE.LE.0) THEN
      RS=0.
      RSS=S*KSSD*FR
      RGD=S*FR*KGD
      S=S-(RSS+RGD)/FR
      ELSE
      FR=R/PE
      S=X*S/FR
      SS=S
      Q=R/FR
      NN=INT(Q/5.0)+1
      Q=Q/REAL(NN)
      KSSDD=(1.-(1.-(KGD+KSSD))**(1.0/REAL(NN)))
+ / (1+KGD/KSSD)
      KGDD=KSSDD*KGD/KSSD

```

```

RS=0.0
RSS=0.0
RG=0.0
SMM=(1.+EX)*SM
IF(EX.LT.0.001) THEN
SMMF=SMM
ELSE
SMMF=SMM*(1.-(1.-FR)**(1./EX))
ENDIF
SMF=SMMF/(1.+EX)
DO 1000 J=1,NN
IF(S.GT.SMF)S=SMF
AU=SMMF*(1.-(1.-S/SMF)**(1.0/(1.+EX)))
IF((Q+AU).LE.0) THEN
RSD=0
RSSD=0
RGD=0
S=0
ELSE IF((Q+AU).GE.SMMF) THEN
RSD=(Q+S-SMF)*FR
RSSD=SMF*KSSDD*FR
RGD=SMF*FR*KGDD
S=SMF-(RSSD+RGD)/FR
ELSE IF((Q+AU).LT.SMMF) THEN
RSD=(Q-SMF+S+SMF*(1-(Q+AU)/SMMF)**(1+EX))
+ *FR
RSSD=(S+Q-RSD/FR)*KSSDD*FR
RGD=(S+Q-RSD/FR)*KGDD*FR
S=S+Q-(RSD+RSSD+RGD)/FR
ENDIF
RS=RS+RSD

```

```

        RSS=RSS+RSSD
        RG=RG+RGD
1000  CONTINUE
        ENDIF
C
        RS=RS*(1-IMP)
        RSS=RSS*(1-IMP)
        RG=RG*(1-IMP)
C
        QRS=(RS+RIMP)*U
        QRSS=QRSS0*CI+RSS*(1-CI)*U
        QRG=QRG0*CG+RG*(1-CG)*U
        QTR=QRS+QRSS+QRG
C
        DO 200 J=1,M
        IF((I+J-1).GT.N)GOTO 120
200   QR(I+J-1)=QR(I+J-1)+QTR*UH(J)
120   CONTINUE
        QRSS0=QRSS
        QRG0=QRG
150   CONTINUE
        RETURN
        END

```

## 二、程序说明

新安江(三水源)模型计算程序中各亚元的物理意义如下:

变量	类型	大小	I/O	内 容
N	整型	1	I	降雨径流计算的时段数
M	整型	1	I	单位线的底宽时段数

- I 新安江(三水源)模型参数,其中:
- $PAR(1)$ ——上层张力水容量  
 $WUM(mm)$ ;
- $PAR(2)$ ——下层张力水容量  
 $WLM(mm)$ ;
- $PAR(3)$ ——深层张力水容量  
 $WDM(mm)$ ;
- $PAR(4)$ ——蒸发皿转换为流域蒸  
散发能力的折算系数  
 $K$ ;
- $PAR(5)$ ——深层蒸散发系数  $C$ ;
- $PAR(6)$ ——张力水蓄水容量曲线  
指数  $B$ ;
- $PAR(7)$ ——不透水面积占全流域  
面积百分数  $IMP$ ;
- $PAR(8)$ ——自由水蓄水库容量  
 $SM(mm)$ ;
- $PAR(9)$ ——自由水蓄水容量曲线  
指数  $EX$ ;
- $PAR(10)$ ——自由水蓄水库地下  
水日出流系数  $KG$ ;
- $PAR(11)$ ——自由水蓄水库壤中  
流日出流系数  $KSS$ ;
- $PAR(12)$ ——地下径流日消退系  
数  $KKG$ ;
- $PAR(13)$ ——壤中流日消退系数  
 $KKSS$

<i>AREA</i>	实型	1	<i>I</i>	单元面积(km <sup>2</sup> )
<i>UH</i>	实型	<i>M</i>	<i>I</i>	无因次时段单位线
<i>DT</i>	实型	1	<i>I</i>	计算时段步长(h)
<i>P</i>	实型	<i>N</i>	<i>I</i>	降雨时间序列(mm)
<i>EP</i>	实型	<i>N</i>	<i>I</i>	蒸发皿实测时段蒸发能力(mm)
<i>QR</i>	实型	<i>N</i>	<i>O</i>	单元出流量(m <sup>3</sup> /s)
<i>W</i>	实型	3	<i>I/O</i>	土壤含水量,其中: $W(1)$ ——上层张力水含量(mm); $W(2)$ ——下层张力水含量(mm); $W(3)$ ——深层张力水含量(mm)
<i>FR</i>	实型	1	<i>I/O</i>	初始产流面积(%)
<i>S</i>	实型	1	<i>I/O</i>	初始流域自由水水深(mm)
<i>QRSS0</i>	实型	1	<i>I/O</i>	初始壤中流流量(m <sup>3</sup> /s)
<i>QRG0</i>	实型	1	<i>I/O</i>	初始地下径流流量(m <sup>3</sup> /s)

### 第三节 实例

#### 一、主程序

下面是使用新安江(三水源)模型进行一个单元产汇流计算的主程序,它从 XAJ.PAR 文件中读取模型参数,从 XAJ.DAT 中读取单元面积上的蒸发能力和平均降雨量,然后调用 XAJ 子程序计算出单元出口处的流量过程。

```

INTEGER YEAR,EM,ED,EH
DIMENSION PP(100),EP(100),UH(20),QR(100),
+ PAR(13),W(3)
OPEN(1,FILE='XAJ.PAR',STATUS='OLD')
OPEN(2,FILE='XAJ.DAT',STATUS='OLD')

```

```

OPEN(3,FILE='XAJ.OUT',STATUS='NEW')
READ(1,*)M,AREA,DT
READ(1,*)PAR
READ(1,*)(UH(I),I=1,M)
READ(1,*)W,FR,S,QRSS0,QRG0
READ(2,*)YEAR,IM,ID,IH,EM,ED,EH,N
READ(2,*)(EP(I),I=1,N)
READ(2,*)(PP(I),I=1,N)
CALL XAJ(N,M,PAR,AREA,UH,DT,PP,EP,QR,W,
+ FR,S,QRSS0,QRG0)
WRITE(3,110)YEAR,IM,ID,IH,EM,ED,EH,DT
110 FORMAT(5X,I4,'年',I2,'月',I2,'日',I2,'时',
+ '-',I2,'月',I2,'日',I2,'时',' DT=',F5.2,'h',
+ '流量过程')
WRITE(3,'(8F8.1)')(QR(I),I=1,N)
STOP
END

```

## 二、实例说明

### 1. 参数文件 XAJ.PAR

```

3  537.  2.
20.  75.  80.  0.65  0.11  0.3  0.  20.  1.  0.3  0.41  0.99  0.6
0.3  0.6  0.1
0.  70.  80.  0.1  20.  40.  20.

```

上面是黄河三门峡至花园口区间东洋河流域的新安江(三水源)模型参数。第一行分别为单位线底宽( $M$ )、单元面积  $AREA$  和计算时段长  $\Delta t$ ; 第二行是参数  $PAR$  数组诸变量; 第三行为单位线; 第四行为计算开始时的状态变量, 分别为上层张力水含量、下层张力水含量、深层张力水含量、产流面积( $FR$ )、自由水水深( $S$ )、壤中流流量( $QRSS0$ )和基流流量( $QRG0$ )。



## 2. 数据文件 XAJ.DAT

1982 8 1 2 8 2 24 24

24 \* 0.

10. 24.1 20.4 18.3 10.1 5.5 0.6 3.1 1.9 4.6

5. 4.8 36.2 29. 6. 3.6 0.4 0. 0.5 3.8

0. 1.8 0.2 0.3

XAJ.DAT 文件第一行分别为计算开始年、月、日、时和结束月、日、时及计算时段数  $N$ ; 第二行往下分别为流域蒸发能力和单元平均降雨过程, 降雨量由花园、横河二站算术平均求得。

## 3. 输出文件 XAJ.OUT

1982 年 8 月 1 日 2 时 ~ 8 月 2 日 24 时  $DT=2.00h$  流量过程

82.5 401.1 712.0 733.7 746.0 643.6 376.4 186.6

233.8 258.0 377.6 410.1 1105.7 2347.2 1743.6 606.0

254.6 84.8 52.8 127.8 183.5 95.6 105.5 62.8

上面是新安江(三水源)模型计算的八里胡同站出流流量过程。

## 第二章 萨克拉门托模型

### 第一节 模型结构

萨克拉门托(Sacramento)模型(以下简称萨克模型)是在 70 年代初由美国加利福尼亚州萨克拉门托河流预报中心研制的一个确定性、概念性的集总参数模型。该模型系统建立于 1971 年。1973 年开始使用,目前美国约有 30% 的河流使用该模型。

萨克模型是在斯坦福(Stanford)IV 号模型的基础上发展的,其结构见图 2-1。模型把流域分成固定不透水面积、透水面积和可变不透水面积三部分,径流来源有固定不透水面积上的直接径流、可变不透水面积上的直接径流和地面径流、透水面积上的地面径流、壤中流、浅层基流和深层基流七个部分,其主体为透水面积。土壤含水量分上、下两层,每层蓄水量又分为张力水和自由水两种。两土层之间由一条霍尔顿(Horton)型下渗曲线沟通,这条曲线是模型的核心。流域的蒸散发也有五部分组成,即透水面积的上层张力水蒸散发、透水面积的上层自由水蒸散发、透水面积的下层张力水蒸散发、可变不透水面积上的蒸散发、河道中的水面蒸发。下面就各部分的计算及其交换关系作一简单介绍。

#### 一、产流量计算

##### 1. 直接径流量

直接径流系降落在不透水面积(包括固定不透水面积和可变不透水面积)上的雨量形成的径流,它直接进入河道。

##### (1) 固定不透水面积上的直接径流

固定不透水面积上的直接径流( $ROIMP$ )等于时段降水量




图 2-1 萨克模型基本结构

( $PP$ )乘以固定不透水面积( $PCTIM$ ),

即

$$ROIMP = PP \cdot PCTIM$$

(2) 可变不透水面积上的直接径流

$$\sum_1^{NINC} PINC \cdot \left( \frac{ADIMC - UZTWC}{LZTWM} \right)^2 \cdot ADIMP$$

式中,  $PINC$  为步长有效降雨量,  $PINC = PAI / NINC$ , 其中,  $NINC$  为时段的步长数,  $PAI = PP - (UZTWM - UZTWC)$ , 称有效降雨量;  $[(ADIMC - UZTWC) / LZTWM]^2 \cdot ADIMP$  为可变不透水面积中的不透水面积。

## 2. 地面径流量

地面径流量也有两部分组成, 一部分是来自透水面积, 另一部分来自可变不透水面积。

(1) 透水面积上的地面径流

透水面积上的地面径流等于时段超渗雨量( $PEX$ )乘以透水面积( $PAREA$ ),

即

$$\sum_1^{NINC} PEX \cdot PAREA$$

式中,  $PEX$  为超渗雨量, 它是指满足上层土壤缺水量(包括渗透和壤中流所扣出的水量)后所剩余的降雨量。

(2) 可变不透水面积上的地面径流

$$\sum_1^{NINC} PEX \cdot \left[ 1 - \left( \frac{ADIMC - UZTWC}{LZTWM} \right)^2 \right] \cdot ADIMP$$

式中,  $\{1 - [(ADIMC - UZTWC) / LZTWM]^2\} \cdot ADIMP$  为可变不透水面积的透水面积。

## 3. 透水面积上产生的壤中流

$$\sum_1^{NINC} UZFWC \cdot [1 - (1 - UZK)^{DINC}] \cdot PAREA$$

式中,  $DINC = 1 / NINC \cdot DT / 24$ ,  $DT$  是以小时计的时段步长。

#### 4. 浅层基流

$$\sum_1^{NINC} LZFSK \cdot [1 - (1 - LZSK)^{DINC}] \cdot PAPEA$$

#### 5. 深层基流

$$\sum_1^{NINC} LZFK \cdot [1 - (1 - LZPK)^{DINC}] \cdot PAPEA$$

### 二、蒸散发计算

#### 1. 上层张力水蒸散发

$$E_1 = \begin{cases} EM \cdot \frac{UZTWC}{UZTWM} & (UZTWC \geq EM) \\ UZTWC & (UZTWC < EM) \end{cases}$$

#### 2. 上层自由水蒸散发

$$E_2 = \begin{cases} EM - E_1 & (EM > UZTWC, \text{且 } UZFWC \geq EM - E_1) \\ UZFWC & (EM > UZTWC, \text{而 } UZFWC < EM - E_1) \end{cases}$$

#### 3. 下层张力水蒸散发

$$E_3 = (EM - E_1 - E_2) \cdot \frac{LZTWC}{UZTWM + LZTWM}$$

#### 4. 河网总入流的水面蒸发

$$E_4 = \begin{cases} EM \cdot SARVA & (SARVA \leq PCTIM) \\ EM \cdot SARVA - (E_1 + E_2 + E_3) \cdot SARRA & (SARVA > PCTIM) \end{cases}$$

式中,  $SARRA = SARVA - PCTIM$ 。

#### 5. 可变不透水面积上的蒸散发

$$E_5 = E_1 + (EM - E_1) \cdot \frac{ADIMC - UZTWC - E_1}{UZTWM + LZTWM}$$

式中,  $UZTWC$  是时段初始值。以上各式中  $EM$  是流域蒸散发能力。

### 三、水分交换计算

#### 1. 上层土壤内的水分交换

当  $\frac{UZFWC}{UZFWM} > \frac{UZTWC}{UZTWM}$  时, 自由水补给张力水, 调整计算公式为

$$UZTWC = UZTWM \cdot \frac{UZTWC + UZFWC}{UZTWM + UZFWM}$$

$$UZFWC = UZFWM \cdot \frac{UZTWC + UZFWC}{UZTWM + UZFWM}$$

#### 2. 下层土壤内的水分交换

当  $LZTWC / LZTWM < (LZFPC + LZFSC - SAVED + LZTWC) / (LZFPM + LZFSM - SAVED + LZTWM)$  时, 自由水补给张力水, 令

$DEL = \{(LZFPC + LZFSC - SAVED + LZTWC) / (LZFPM + LZFSM - SAVED + LZTWM) - LZTWC / LZTWM\} \cdot LZTWM$   
则调整后的张力水为

$$LZTWC = LZTWC + DEL$$

调整后的浅层自由水为

$$LZFSC = LZFSC - DEL$$

### 四、下渗计算

#### 1. 下渗率的计算

模型认为上土层向下土层的下渗率( $PERC$ )与稳定下渗能力( $PBASE$ )、下层土壤的缺水程度及上层自由水的供水能力有关, 按霍尔顿下渗曲线(见图2-2), 其计算公式为

$$PERC = PBASE \cdot (1 + ZPERC \cdot DEFR^{REXP}) \cdot \frac{UZFWC}{UZFWM}$$

式中

$$PBASE = LZFPM \cdot LZPK + LZFSM \cdot LZSK$$

$$DEF R = 1 - \frac{LZFPC + LZFS C + LZTWC}{LZFPM + LZFSM + LZTWM}$$

## 2. 下渗水量的分配

下渗的水量中按  $PFREE$  的比例补给下层自由水, 即  $PERC \cdot PFREE$ 。其余  $PERC \cdot (1 - PFREE)$  的水量补给张力水, 若张力水已满足, 即  $[PERC \cdot (1 - PFREE) + LZTWC] > LZTWM$ , 则超过部分仍补给下层自由水。

补给自由水的水量 ( $PERC \cdot PFREE$ ), 再按深层和浅层各自的相对蓄水量进行分配, 其中分给深层自由水的水量为

$$PERCP = (PERC \cdot PFREE) \cdot \frac{LZFPM}{LZFPM + LZFSM} \cdot \frac{2 \times (1 - LZFPC/LZFPM)}{(1 - LZFPC/LZFPM) + (1 - LZFS C/LZFSM)}$$

分给浅层自由水的水量为

$$PERCS = (PERC \cdot PFREE) - PERCP$$

若  $PERCS + LZFS C > LZFSM$ , 则超量部分再补给深层自由水。




图2-2 下渗曲线

当下渗的水量超过了下层总的土层缺水量时, 即  $(PERC + LZFPC + LZFS C + LZTWC) > (LZFPM + LZFSM$



+LZTWM), 则需把多出的水量反馈到上层自由水蓄量中, 反馈量  $CHECK = (PERC + LZFP C + LZFS C + LZTW C) - (LZFPM + LZFSM + LZTWM)$ 。

## 五、模型参数的意义

- (1) *PCTIM*——河槽及其邻近的不透水面积占全流域面积的百分数;
- (2) *ADIMP*——流域中当全部张力水饱和后变成不透水面积的比例, 称可变不透水面积;
- (3) *SARVA*——河道、湖泊和水生物占全流域面积的百分数;
- (4) *UZTWM*——上层张力水容量, 以 mm 计;
- (5) *UZFWM*——上层自由水容量, 以 mm 计;
- (6) *UZK*——上层自由水日出流系数, 以小数计;
- (7) *ZPERC*——下渗系数, 决定下层最干旱时的最大下渗率;
- (8) *REXP*——下渗指数, 决定下渗率随上层蓄水量变化的函数形成;
- (9) *LZTWM*——下层张力水容量, 以 mm 计;
- (10) *LZFSM*——下层浅层自由水容量, 以 mm 计;
- (11) *LZFPM*——下层深层自由水容量, 以 mm 计;
- (12) *LZSK*——下层浅层自由水日出流系数, 以小数计;
- (13) *LZPK*——下层深层自由水日出流系数, 以小数计;
- (14) *PFREE*——从上层向下层下渗水量中直接补给下层自由水的比例;
- (15) *RSEVI*——下层自由水中不蒸发部分所占比例;
- (16) *SIDE*——不闭合的地下水出流量, 以未测到的地下水占总基流百分数计;
- (17) *SSOUT*——河槽总径流中径流损失系数;

## 六、几个主要参数对洪水的影响

萨克模型中虽然有17个参数,但主要参数有7个: $UZTWM$ 、 $UZFWM$ 、 $LZTWM$ 、 $LZFPM$ 、 $LZF5M$ 、 $REXP$  和  $ZPERC$ ,表2—1给出了这7个主要参数对洪水的影响情况。

$UZTWM$ 、 $UZFWM$  主要用于改变一次洪水的流量或过程。其中, $UZTWM$  的变化主要影响峰前量,而变动  $UZFWM$  则不但改变峰前量,更主要的是通过改变壤中流使峰后值也有所变化,如果要求仅改变过程线形状而使水量不变,可增大其中一个,减小另一个。

表2—1 萨克模型几个主要参数对洪水的影响

参 数		相 应 值 变 化					主要 影响项
符号	变化	峰值	$R_s$	$R_{sub}$	$R_g$	$R$	
$UZTWM$	↑	↓	↓	↓	↓	↓	$R$
$UZFWM$	↑	↓	↓	↑	↓	↓	过程线
$LZTWM$	↑	↓	↓	↓	↓	↓	$R$
$LZF5M$	↑	↓	↓	↓	↑	↓	过程线
$LZFPM$	↑	↓	↓	↓	↑	↓	过程线
$REXP$	↑	↑	↑	↑	↓	↑	过程线
$ZPERC$	↑	↓	↓	↓	↑	↓	过程线

注:↑——增加,↓——减少, $R_s$ ——地面径流, $R_{sub}$ ——壤中流,  
 $R_g$ ——基流, $R$ ——总径流量。

$REXP$ 、 $PBASE$  的影响是性质相同,方向相反。即  $REXP$  加大,渗透曲线下凹,减少渗透,增大了地面径流和壤中流,减少小部分基流,而总径流量加大。但  $PBASE$  加大,则减少地面径流和壤中流,增加少量基流,总径流量减少。

## 第二节 模型计算程序

萨克模型计算程序同新安江(三水源)模型一样,是根据实测

降雨、蒸发能力时间序列和开始时的流域下垫面条件,计算出单元流域面积出口处的总出流量过程。如果想输出计算的单元蒸散发和地面径流、壤中流、基流各分量过程,可对程序略加变动,把EUSED、FLOSF、FLOIN、FLOBF 输出即可。这些内容已计算出来,只是放在公共语句 SACIO 中没有输出。

## 一、计算程序

```

SUBROUTINE SAC(P,N,UH,M,DT,PP,EP,QR,UZTWC,
+ UZFWC,LZTWC,LZFSC,LZFPC,ADIMC,QG0)
  REAL LZTWM,LZFSM,LZFPM,LZSK,LZPK,LZTWC,
+ LZFSC,LZFPC
  DIMENSION P(20),PP(N),EP(N),UH(M),QR(N)
  COMMON/SACPAR/UZTWM,UZFWM,LZTWM,LZFSM,
+ LZFPM,FRACT,SAVED,ZPERC,PAREA,PCTIM,REXP,
+ UZK,LZSK,LZPK,PERCM,PFREE,EKC,SIDE,SSOUT,
+ ADIMP,SARVA,RSERV,CG
  COMMON/SACIO/PLIQ,EDMND,
+ E1,E2,E3,E5,EUSED,FLOSF,FLOIN,FLOBF
  UZTWM=P(1)
  UZFWM=P(2)
  LZTWM=P(3)
  LZFSM=P(4)
  LZFPM=P(5)
  UZK=P(6)
  LZSK=P(7)
  LZPK=P(8)
  ZPERC=P(9)
  REXP=P(10)
  SIDE=P(11)
  SSOUT=P(12)

```

```

PCTIM=P(13)
ADIMP=P(14)
SARVA=P(15)
PFREE=P(16)
RSERV=P(17)
AREA=P(18)
CG=P(19)
EKC=P(20)
FRACT=DT/24.
SAVED=RSERV*(LZFSM+LZFPM)
PERCM=LZSK*LZFSM+LZPK*LZFPM
PAREA=1.-PCTIM-ADIMP
IF(SARVA.GT.PCTIM)SARVA=PCTIM
DO 10 I=1,N
10 QR(I)=0.
CGD=CG** (DT/24.)
DO 100 I=1,N
PLIQ=PP(I)
EDMND=EP(I)*EKC
CALL SACSUB(UZTWC,UZFWC,LZTWC,LZFSC,LZFPC,
+ ADIMC)
E4=EDMND*SARVA
FLOS=ROIMP+FLOSF-SSOUT*FRACT
IF(E4-FLOS)61,62,62
61 FLOS=FLOS-E4
GOTO 63
62 E4=FLOS
FLOS=0.
63 E=E4+EUSED
SID=1./(1.+SIDE)
FLOBF=FLOBF*SID

```

```

      U=AREA/(DT*3.6)
C      _____
      QG=QG0*CGD+FLOBF*(1-CGD)*U
      QG0=QG
C      _____
      QRT=(FLOS+FLOIN)*U+QG
C      _____
      DO 20 J=1,M
      IF((I+J-1).GT.N)GOTO 30
20    QR(I+J-1)=QR(I+J-1)+QRT*UH(J)
30    CONTINUE
C      _____
100   CONTINUE
      RETURN
      END

```

```

      SUBROUTINE SACSUB(UZTWC,UZFWC,LZTWC,LZFSC,
+ LZFPC,ADIMC)
      REAL LZTWM,LZFSM,LZFPM,LZSK,LZPK,LZTWC,LZFSC,
+ LZFPC
      COMMON/SACPAR/UZTWM,UZFWM,LZTWM,LZFSM,
+ LZFPM,FRACT,SAVED,ZPERC,PAREA,PCTIM,REXP,
+ DCUZ,DCLZS,DCLZP,PERCM,PFREE,EKC,SIDE,SSOUT,
+ ADIMP,SARVA,RSERV,CG
      COMMON/SACIO/PLIQ,EDMND,E1,E2,E3,E5,EUSED,
+ FLOSF,FLOIN,FLOBF
      E2=0.
      E1=EDMND*UZTWC/UZTWM
      RED=EDMND-E1
      UZTWC=UZTWC-E1
      IF(UZTWC)100,155,155

```

```

100  E1=E1+UZTWC
      UZTWC=0.
      RED=EDMND-E1
      IF(UZFWC-RED)130,140,140
130  E2=UZFWC
      UZFWC=0.
      GOTO 170
140  E2=RED
      UZFWC=UZFWC-E2
155  CONTINUE
      A=UZTWC/UZTWM
      B=UZFWC/UZFWM
      IF(A-B)160,170,170
160  A=(UZTWC+UZFWC)/(UZTWM+UZFWM)
      UZTWC=UZTWM*A
      UZFWC=UZFWM*A
170  CONTINUE
      E5=E1+RED*(ADIMC-E1-UZTWC)/(UZTWM+LZTWM)
      E3=(RED-E2)*LZTWC/(UZTWM+LZTWM)
      LZTWC=LZTWC-E3
      IF(LZTWC)180,185,185
180  E3=E3+LZTWC
      LZTWC=0.
185  CONTINUE
      A=LZTWC/LZTWM
      B=(LZFPC+LZFSC-SAVED+LZTWC)/(LZFPM+LZFSM-
+ SAVED+LZTWM)
      IF(A-B)190,210,210
190  DEL=(B-A)*LZTWM
      LZTWC=LZTMC+DEL
      LZFSC=LZFSC-DEL

```

```

        IF(LZFSC)200,210,210
200    LZFPC=LZFPC+LZFSC
        LZFSC=0.
210    CONTINUE
        ROIMP=PLIQ * PCTIM
        ADIMC=ADIMC-E5
        IF(ADIMC)211,212,212
211    E5=E5+ADIMC
        ADIMC=0.
212    E5=E5 * ADIMP
        PAV=PLIQ+UZTWC-UZTWM
        IF(PAV)220,230,230
220    UZTWC=UZTWC+PLIQ
        PAV=0.
        GOTO 240
230    UZTWC=UZTWM
240    NINC=1. +0.2 * (UZFWC+PAV)
        ADIMC=ADIMC+PLIQ-PAV
        DINC=NINC
        PINC=PAV /DINC
        FLOBF=0.
        FLOSF=0.
        FLOIN=0.
        DINC=FRACT /DINC
345    CONTINUE
        DUZ=DCUZ
        DLZP=DCLZP
        DLZS=DCLZS
        IF(DINC-1.)247,248,248
247    DUZ=1. -(1.-DCUZ) * * DINC
        DLZP=1. -(1.-DCLZP) * * DINC

```

```

DLZS = 1. - (1. - DCLZS) * * DINC
248 CONTINUE
DO 385 INC = 1, NINC
PAV = PINC
RATIO = (ADIMC - UZTWC) / LZTWM
ADDRO = PINC * RATIO * RATIO
BF = LZFPC * DLZP
FLOBF = FLOBF + BF
LZFPC = LZFPC - BF
BF = LZFSC * DLZS
LZFSC = LZFSC - BF
FLOBF = FLOBF + BF
IF (PINC + UZFWC - 0.01) 380, 380, 250
250 PERC = PERCM * DINC
PERC = PERC * UZFWC / UZFWM
PERC = PERC * (1. + (ZPERC * (1. - (LZFPC + LZFSC +
+ LZTWC) / (LZFPM + LZFSM + LZTWM)) * * REXP))
IF (PERC - UZFWC) 270, 270, 260
260 PERC = UZFWC
UZFWC = 0.
GOTO 300
270 UZFWC = UZFWC - PERC
CHECK = LZFPC + LZFSC + PERC + LZTWC - LZFPM - LZFSM
+ - LZTWM
IF (CHECK) 290, 290, 280
280 PERC = PERC - CHECK
UZFWC = UZFWC + CHECK
290 DEL = DUZ * UZFWC
FLOIN = FLOIN + DEL
UZFWC = UZFWC - DEL
300 CONTINUE

```



```

SPERC=PERC
PERC=PERC*(1.-PREE)
IF(PERC-LZTWM+LZTWC)310,310,320
310 LZTWC=LZTWC+PERC
PERC=0.
GOTO 321
320 PERC=PERC-LZTWM+LZTWC
LZTWC=LZTWM
321 PERC=PERC+SPERC*PREE
IF(PERC)310,340,323
323 CONTINUE
HPL=LZFPM/(LZFPM+LZFSM)
RATLP=LZFPC/LZFPM
PERCS=PERC-PERC*(HPL*2.*(1.-RATLP)/((1.
+ -RATLP)+(1.-LZFSC/LZFSM)))
LZFSC=LZFSC+PERCS
IF(LZFSC-LZFSM)330,330,322
322 PERCS=PERCS-LZFSC+LZFSM
LZFSC=LZFSM
330 CONTINUE
LZFPC=LZFPC+(PERC-PERCS)
340 IF(PAV)380,380,350
350 CONTINUE
IF(PAV-UZFWM+UZFWC)360,360,370
360 UZFWC=UZFWC+PAV
PAV=0.
GOTO 380
370 PAV=PAV-UZFWM+UZFWC
UZFWC=UZFWM
ADDRO=ADDRO+PAV*(1.-ADDRO/PINC)
FLOSF=FLOSF+PAV

```

```

380  CONTINUE
      ADIMC=ADIMC+PINC-ADDRO
      ROIMP=ROIMP+ADDRO*ADIMP
385  CONTINUE
      PAV=0.
390  CONTINUE
      EUSED=(E1+E2+E3)*PAREA+E5
      FLOSF=FLOSF*PAREA
      FLOIN=FLOIN*PAREA
      FLOBF=FLOBF*PAREA
      RETURN
      END

```

## 二、程序说明

变量	类型	大小	I/O	内 容
<i>P</i>	实型	20	<i>I</i>	单元面积上的萨克模型参数。依次为 <i>UZTWM</i> 、 <i>UZFWM</i> 、 <i>LZTWM</i> 、 <i>LZFSM</i> 、 <i>LZFPM</i> 、 <i>UZK</i> 、 <i>LZSK</i> 、 <i>LZPK</i> 、 <i>ZPERC</i> 、 <i>REXP</i> 、 <i>SIDE</i> 、 <i>SSOUT</i> 、 <i>PCTIM</i> 、 <i>ADIMP</i> 、 <i>SARVA</i> 、 <i>PFREE</i> 、 <i>RSERV</i> 、 <i>AREA</i> 、 <i>CG</i> 、 <i>EKC</i> 。其中 <i>AREA</i> 表示单元流域面积(km <sup>2</sup> )； <i>CG</i> 为地下径流日消退系数； <i>EKC</i> 是蒸发皿转换为流域蒸散发能力的折算系数。其它变量见上节模型参数的意义
<i>N</i>	整型	1	<i>I</i>	计算的时段数
<i>UH</i>	实型	<i>M</i>	<i>I</i>	无因次时段单位线

<i>M</i>	整型	1	1	单位线底宽(时段数)
<i>DT</i>	实型	1	1	时段长(h)
<i>PP</i>	实型	<i>N</i>	1	单元面积上的平均降雨(mm)
<i>EP</i>	实型	<i>N</i>	1	单元面积上的蒸发皿的实测时段 蒸发量(mm)
<i>QR</i>	实型	<i>N</i>	0	单元总出流量( $\text{m}^3/\text{s}$ )
<i>UZTWC</i>	实型	1	I/O	上层张力水含量(mm)
<i>UZFWC</i>	实型	1	I/O	上层自由水含量(mm)
<i>LZTWC</i>	实型	1	I/O	下层张力水含量(mm)
<i>LZFSC</i>	实型	1	I/O	下层浅层自由水含量(mm)
<i>LZFPC</i>	实型	1	I/O	下层深层自由水含量(mm)
<i>ADIMC</i>	实型	1	I/O	可变不透水面积上的含水量 (mm)
<i>QG0</i>	实型	1	I/O	地下水初始流量( $\text{m}^3/\text{s}$ )。

### 第三节 实例

#### 一、主程序

下面是萨克模型的主程序,它从SAC.PAR文件中读取模型参数,从SAC.DAT中读取流域蒸发能力和面平均降雨数据,通过调用子程序SAC计算出单元出流过程,结果写入SAC.OUT。

```

INTEGER YEAR,EM,ED,EH
REAL LZTWC,LZFSC,LZFPC
DIMENSION PP(100),EP(100),UH(20),QR(100),PAR(20)
OPEN(1,FILE='SAC.PAR',STATUS='OLD')
OPEN(2,FILE='SAC.DAT',STATUS='OLD')

```

```

OPEN(3,FILE='SAC.OUT',STATUS='NEW')
READ(1,*)M,DT
READ(1,*)PAR
READ(1,*)(UH(I),I=1,M)
READ(1,*)UZTWC,UZFWC,LZTWC,LZFSC,LZFPC,ADIMC,
+ QG0
READ(2,*)YEAR,IM,ID,IH,EM,ED,EH,N
READ(2,*)(EP(I),I=1,N)
READ(2,*)(PP(I),I=1,N)
CALL SAC(PAR,N,UH,M,DT,PP,EP,QR,UZTWC,UZFWC,
+ LZTWC,LZFSC,LZFPC,ADIMC,QG0)
WRITE(3,110)YEAR,IM,ID,IH,EM,ED,EH,DT
110 FORMAT(5X,I4,'年',I2,'月',I2,'日',I2,'时',
+ '—',I2,'月',I2,'日',I2,'时',' DT=',F5.1,'h ',
+ '流量过程')
WRITE(3,'(8F8.1)')(QR(I),I=1,N)
STOP
END

```

## 二、实例说明

### 1. 参数文件 SAC.PAR

```

6      2.
25.    25.    130.   35.    105.   0.25  0.07  0.005  20.   1.8
0.0    0.0    0.01  0.01  0.01  0.42  0.3   1695.  0.99  1.0
0.05   0.10   0.35  0.25  0.15  0.10
0.0    25.    120.   30.    105.   30.   100.

```

上面是黄河流域伊河上游潭头水文站以上的萨克模型参数。第一行分别为单位线底宽( $M$ )和计算时段的步长( $\Delta t$ );第二、三行为萨克模型产汇流计算数组  $P$  的诸变量;第四行  $M$  个数为单位线纵坐标;第五行分别是初始状态变量  $UZTWC$ 、 $UZFWC$ 、

LZTWC、LZFS C、LZFPC、ADIMC 和 QG0。

## 2. 数据文件 SAC.DAT

1982	7	30	8	8	2	6	36		
36 * 0.									
8.1	21.2	7.9	0.0	0.0	0.0	0.0	1.4	7.7	3.0
4.3	6.6	3.6	5.5	6.0	7.0	8.1	8.8	9.5	12.7
13.5	11.7	7.3	2.2	1.2	0.0	0.0	0.0	2.0	1.4
3.4	0.1	0.0	0.0	0.0	0.0				

SAC.DAT 文件中的内容与 XAJ.DAT 中的相同,降雨是由陶湾、庙子、白土、潭头四站平均求得。

## 3. 输出文件 SAC.OUT

1982 年 7 月 30 日 8 时~8 月 2 日 6 时 DT=2.0h 流量过程

8.3	44.6	178.0	377.3	646.4	531.1	406.7	308.4
190.8	181.0	179.8	185.8	192.6	201.6	213.3	253.8
350.4	601.1	867.0	1132.5	1407.8	1754.4	1951.9	1843.3
1415.1	870.6	495.2	277.8	194.4	180.6	169.9	163.5
160.0	157.8	153.1	146.4				

SAC.OUT 文件中的内容是由萨克模型计算的潭头站流量过程。

## 第三章 水箱模型

### 第一节 模型结构

水箱模型(Tank model)由日本国立防灾科学研究中心菅原正己博士于1961年提出。目前,日本各河流的水箱模型基本上都已建立。70年代在不少国家作了应用。1980年以来我国有关单位也作了一些研究和试用。




图3-1 具有一个侧孔和一个底孔的水箱

水箱模型最简单的结构如图3-1所示,水箱模型假定每个水箱中任意一个出流孔的出流量与其中蓄量呈线性关系。因此,在图3-1所示的情况下,侧孔与底孔的出流量可分别按下式计算

$$Y = A_1 X \quad (3-1)$$

$$Z = A_0 X \quad (3-2)$$

式中,  $X$  为水箱中任一时刻的蓄量;  $A_0$ 、 $A_1$  分别为底孔和侧孔的出流系数;  $Z$ 、 $Y$  分别为底孔和侧孔的出流量。

从物理意义上讲,底孔出流可模拟下渗,侧孔出流可模拟径流。如果水箱有两个侧孔(见图3-2),则按照上述相同的假设,侧孔出流公式变为

$$Y = \begin{cases} 0 & X \leq H_1 \\ A_1 \cdot (X - H_1) & H_1 < X \leq H_2 \\ A_1 \cdot (X - H_1) + A_2 \cdot (X - H_2) & X > H_2 \end{cases} \quad (3-3)$$

式中,  $H_1$ 、 $H_2$  分别为两个侧孔的高度;  $A_2$  为另一个侧孔的出流系

数;其余符号意义同前。

水箱模型的一般结构如图3—3a所示。降雨从最上面的水箱进入,蒸发能力作为负降水处理(但不累计)。各水箱侧面和底部设有出流孔和渗透孔。各水箱的蓄水或从侧孔流出,或从底孔向下一个水箱渗透,每孔的出流量与它所在水箱的蓄量呈线性关系。

各侧孔出流之和即为流量。第一层水箱出流相当于地面径流,当第一层水箱多于一个侧孔时可用于模拟地面径流的非线性性质。其余三层水箱分别用于模拟壤中流、准基流、基流。见图3—3b所示的不同径流成分示意图。





图3—2 具有二个侧孔和一个底孔的水箱



(a) 四层水箱模型 (b) 不同径流成分划分

图3—3 水箱模型模拟不同径流成分示意图



在日本,一年中经常下雨,土壤表层总是湿润的,基本上无需考虑土壤水分的影响,图3-3所示的结构就可取得很好结果。但在其他国家,即使是湿润地区,也经常必须考虑土壤水分的影响。为此在上面的水箱里加入了两个土柱。第一个土柱装在第一个水箱的底部,雨水首先充满这部分,其余的水作为第一个水箱的自由水,渗透和出流均由自由水供给。第二个土柱加在第一个土柱旁,雨量加到第一个水箱的蓄量(包括第一个土柱含水量和第一个水箱的自由水)中去,蒸发也从此蓄水量中减去。当第一个土柱不饱和时,与其干燥度成比例地从下面水箱中吸收自由水。两个土柱之间的土壤水分大小,与它们的相对湿度之差成比例地从湿的一方渗透到另一方(见图3-4)。不过,加了土柱以后,已不再是纯粹水箱的概念了,本节不作赘述。




图3-4 含土柱的水箱模型

在干旱地区,水在重力的作用下向低处运动,高处变干,低处变湿,出现了空间分布不均匀的现象。为模拟这种现象,把流域按高度分成几个地带,各地带分别使用水箱模型,就形成了  $m \times n$  排列的水箱模型。这样,一个水箱除接受上面的水箱来水及向下面水箱供水之外,还从高处的旁侧水箱中接受来水,向低处旁侧水箱供水(见图3-5)。

水箱模型包括的参数是蒸散发折算系数、侧孔和底孔高度与出流系数。由于它们的物理定义不很明确,故主要依靠调试的方法来率定。例如,对于图3-1所示的最简单的水箱模型,  $(A_0 + A_1)$  的大小决定了流量过程线的形状,加大  $(A_0 + A_1)$ , 可使过程线尖瘦;减小  $(A_0 + A_1)$ , 可使过程线矮胖。  $A_1/A_0$  的大小决定了出流总量。  $A_1$  大, 出流量大, 下渗量小;  $A_0$  大, 结果则相反。因此,在调试参数



时,若发现总量符合,过程线形状拟合欠佳,则可调整 $(A_0 + A_1)$ ;反之,应当调整 $A_1/A_0$ 。




图3-5 干旱区水箱模型结构

根据一般的经验,一、二层水箱模型可用于模拟次洪水;二、三、四层水箱可用于模拟年或多年的流量过程线。就目前国内外应用该模型的情况看,水箱模型的结构还没有超过四层的。总的说来,该模型使用的效果是好的。

## 第二节 模型计算程序

水箱模型计算程序是一个 $M \times N$ 的纯水箱模型结构式。 $M$ 和 $N$ 最大可以为5,每个水箱最多可有5个孔,并且其中必须包括一个,并且只能有一个底孔,底孔的相应孔高参数为0,如果没有底孔,孔高和出流系数都置为0。若想模拟孔顶溢流现象,可把水箱侧

孔的最上一个孔的出流系数置为1。

### 一、计算程序

```
SUBROUTINE TANK(M,N,HOLE,HGT,RATE,X,  
+ L,PP,EM,DT,AREA,Q)  
  INTEGER HOLE  
  DIMENSION HOLE(5,N),HGT(5,5,5),RATE(5,5,5),X(5,N),  
+ PP(L),EM(L),Q(L)  
  DO 10 I=1,L  
    Q(I)=0.  
10  CONTINUE  
  DO 1000 IL=1,L  
    PI=PP(IL)  
    EI=EM(IL)  
    X(1,1)=X(1,1)+PI-EI  
    DO 500 IN=1,N  
      DO 200 IM=1,M  
        SIDEQ=0.  
        BOTQT=0.  
        SIDEE=0.  
        BOTTE=0.  
        K=HOLE(IM,IN)-1  
        IF(X(IM,IN).LT.0)GOTO 100  
        DO 50 IK=1,K  
          GHT=X(IM,IN)-HGT(IM,IN,IK)  
          IF(GHT.GT.0)THEN  
            SIDEQ=SIDEQ+RATE(IM,IN,IK)*GHT  
          ELSE  
            ENDIF  
50  CONTINUE
```

```

      K=HOLE(IM,IN)
      BOTTQ=RATE(IM,IN,K)*X(IM,IN)
      GOTO 180
100  CONTINUE
      EV=0.-X(IM,IN)
      DO 150 IK=1,K
      SIDE=SIDE+RATE(IM,IN,IK)*EV
150  CONTINUE
      K=HOLE(IM,IN)
      BOTTE=RATE(IM,IN,K)*EV
180  CONTINUE
      IF(IN.EQ.N)THEN
      Q(IL)=Q(IL)+SIDEQ
      ENDIF
      X(IM,IN)=X(IM,IN)-SIDEQ-BOTTQ-SIDE-BOTTE
      IF(X(IM,IN).LT.0)X(IM,IN)=0.
      IF(IM.LT.M)X(IM+1,IN)=X(IM+1,IN)+BOTTQ-BOTTE
      IF(IN.LT.N)X(IM,IN+1)=X(IM,IN+1)+SIDEQ-SIDE
200  CONTINUE
500  CONTINUE
      Q(IL)=Q(IL)*AREA/3.6/DT
1000 CONTINUE
      RETURN
      END

```

## 二、程序说明

变量	类型	大小	I/O	内 容
M	整型	1	I	水箱的行(层)数
N	整型	1	I	水箱的列数

<i>HOLE</i>	整型	$5 \times N$	<i>I</i>	各水箱实有出流孔数, <i>HOLE(I, J)</i> 表示第 <i>I</i> 行 第 <i>J</i> 列个水箱的出流孔 数。每个水箱有一个且只 有一个底孔
<i>HGT</i>	实型	$5 \times 5 \times 5$	<i>I</i>	出流孔的高度(mm): <i>HGT(I, J, K)</i> ——第 <i>I</i> 行第 <i>J</i> 列水箱第 <i>K</i> 个孔 距孔底的距离
<i>RATE</i>	实型	$5 \times 5 \times 5$	<i>I</i>	每个出流孔的出流系数: <i>RATE(I, J, K)</i> ——第 <i>I</i> 行第 <i>J</i> 列水箱第 <i>K</i> 个孔 的时段出流系数
<i>X</i>	实型	$5 \times N$	<i>I/O</i>	各个水箱的初始水深 (mm)
<i>L</i>	整型	1	<i>I</i>	计算时段数
<i>PP</i>	实型	<i>L</i>	<i>I</i>	时段降雨量(mm)
<i>EM</i>	实型	<i>L</i>	<i>I</i>	时段流域蒸发能力(mm)
<i>DT</i>	实型	1	<i>I</i>	计算时段长(h)
<i>AREA</i>	实型	1	<i>I</i>	流域面积(km <sup>2</sup> )
<i>Q</i>	实型	<i>L</i>	<i>O</i>	出流量(m <sup>3</sup> /s)

注:① 在每个水箱中,底孔的出流系数和孔高(0)应放在最后。

②该程序可以模拟最大5行,最大5列,每个水箱最多5个孔的水箱模型结构。

### 第三节 实例

#### 一、主程序

下面是水箱模型主程序,它从 TANK.PAR 中读取水箱模型参数,从 TANK.DAT 中读取流域蒸发能力和面平均降水数据,通过调用子程序 TANK 计算出单元出流过程,结果写入 TANK.OUT。

```
      INTEGER YEAR,EM,ED,EH,HOLE
      DIMENSION PP(100),EP(100),Q(100),
+ HGT(5,5,5),RATE(5,5,5),X(5,5),HOLE(5,5)
      OPEN(1,FILE='TANK.PAR',STATUS='OLD')
      OPEN(2,FILE='TANK.DAT',STATUS='OLD')
      OPEN(3,FILE='TANK.OUT',STATUS='NEW')
      READ(1,*)M,N,DT,AREA
      READ(1,*)((HOLE(I,J),I=1,M),J=1,N)
      DO 10 J=1,N
      DO 10 I=1,M
      NK=HOLE(I,J)
      DO 10 K=1,NK
      READ(1,*)HGT(I,J,K),RATE(I,J,K)
10    CONTINUE
      READ(1,*)((X(I,J),I=1,M),J=1,N)
      READ(2,*)YEAR,IM,ID,IH,EM,ED,EH,L
      READ(2,*)(EP(I),I=1,L)
      READ(2,*)(PP(I),I=1,L)
      CALL TANK(M,N,HOLE,HGT,RATE,X,L,PP,EP,DT,
+ AREA,Q)
      WRITE(3,110)YEAR,IM,ID,IH,EM,ED,EH,DT
```

```

110    FORMAT(5X,I4,'年',I2,'月',I2,'日',I2,'时',
      + '—',I2,'月',I2,'日',I2,'时','DT=',F4.0,'h',
      + '流量过程')
      WRITE(3,'(8F6.0)')(Q(I),I=1,L)
      STOP
      END

```

## 二、实例说明

### 1. 参数文件 TANK.PAR

```

2  1  6.  3000.
3  2
45.  0.20
20.  0.15
0.0  0.10
10.  0.08
0.0  0.05
15.  20.

```

上面是某流域的水箱模型参数。第一行分别是水箱行数  $M$ 、列数  $N$ 、计算时段长  $DT$ 、单元流域面积  $AREA$ ；第二行是每个水箱的实有孔数；接下来是每个出流孔的孔高和时段出流系数；最后一行是每个水箱的初始水深。在读每个水箱参数时是先读第一列（自上而下）、再读第二列，依次类推。

### 2. 数据文件 TANK.DAT

1982	8	7	8	8	11	2	16
1.5	1.	1.	0.5	0.5	1.	1.	9 * 1.5
0.0	3.	5.	25.	30.	10.	5.	9 * 0.

TANK.DAT 文件中第一行是计算开始的年、月、日、时和结束月、日、时及计算历时，第二行和第三行分别为流域蒸发能力和

单元平均降水过程。

### 3. 输出文件 TANK.OUT

1982年 8 月 7 日 8 时~ 8 月 11 日 2 时 DT=6. h 流量过程

126.	120.	117.	548.	1541.	1234.	822.	572.
411.	283.	210.	194.	177.	159.	140.	122.

这是水箱模型计算的单元出流量过程。

## 第四章 陕北模型

### 第一节 模型结构

陕北模型是河海大学建立的，也称超渗模型。在陕北地区，由于气候干旱，雨量稀少，地下水位低，包气带缺水量大，一般降雨不可能使包气带蓄满，不会形成地下径流。但由于土壤贫瘠，植被较差，根系不发达。地面下渗能力小，雨强很容易超过地面下渗能力，而形成地面径流。因此，干旱地区的产流方式主要是雨强超过地面下渗能力而形成的地面径流。

#### 一、点降雨径流模型

所谓点降雨径流模型，也就是降雨和下垫面条件在计算流域上的空间分布都是均匀的，可视为一点来考虑的极小流域的降雨径流模型。

由于在陕北模型中，对产流起主导作用的是降雨强度与地面下渗能力之间的对比关系。当雨强小于地面下渗能力时，所有的降雨被土壤吸收，不产流。而当雨强大于地面下渗能力时，吸收率只等于下渗能力，其余部分产流，即

$$\text{当 } i \leq f, \quad R = 0, \quad W = W_0 + i$$

$$\text{当 } i > f, \quad R = RS = i - f, \quad W = W_0 + f$$

式中， $i$  为减去可能蒸发率后剩余的雨强； $f$  为地面下渗能力； $R$  为产流率； $RS$  为地面径流； $W_0$  和  $W$  分别表示时段初和时段末影响土壤层的含水量。

下渗能力的大小，是地表附近土壤含水量的函数（见图4—1）。常用的下渗能力计算公式有霍尔顿型和菲利浦型，下面分别予以



介绍。

### 1. 霍尔顿下渗曲线

霍尔顿下渗曲线  $f \sim t$  关系为

$$f = f_c + (f_0 - f_c)e^{-kt} \quad (4-1)$$

式中,  $f_0$ 、 $f_c$  分别为土壤最干和最湿时的地表下渗能力;  $k$  为衰减系数, 因次为  $[T^{-1}]$ 。




图 4-1 下渗曲线与土壤湿度

将  $f \sim t$  转换成  $W \sim t$  的形式为

$$W = \int_0^t f dt = f_c t + \frac{1}{k}(f_0 - f_c)(1 - e^{-kt}) \quad (4-2)$$

联解式(4-1)及式(4-2), 消去  $t$ , 得到  $W \sim f$  的关系为

$$f = f_c + (f_0 - f_c)e^{(f_0 - kW - f)/f_c} \quad (4-3)$$

### 2. 菲利浦下渗曲线

如果采用菲利浦下渗曲线,  $f \sim t$  关系为

$$f = \frac{B}{\sqrt{t}} + A \quad (4-4)$$

式中,  $B$  和  $A$  为两个待定参数,  $B$  的因次为  $[LT^{-\frac{1}{2}}]$ ;  $A$  的因次为  $[LT^{-1}]$ 。

$W \sim t$  的关系为

$$W = 2B\sqrt{t} + At \quad (4-5)$$

联解式(4-4)和式(4-5), 得到  $W \sim f$  关系为

$$f = B^2(1 + \sqrt{1 + AW/B^2})/W + A \quad (4-6)$$

### 3. 蒸散发计算

通常的陕北模型计算中不考虑蒸散发问题, 它仅适用于一场洪水过程的模拟, 一般称之为事件模型。不考虑土壤蒸散发, 土

壤含水量就只增不减,显然,难以进行长期过程或一场不连续降雨径流过程的模拟。为了使陕北模型适用于长时期计算,就需要考虑蒸散发,使它成为连续模型。程序中的蒸发计算采用一层蒸发模式。

#### 4. 推流计算

点降雨径流模型的输入为时段降雨和时段流域蒸发能力。为了使降雨强度和下渗能力不被均化,所以  $\Delta t$  要取的很短。模型的参数包括用于计算下渗能力的参数,对于霍尔顿下渗曲线,下渗能力计算参数包括稳定下渗能力( $f_c$ )、最大下渗能力( $f_0$ )、消退系数( $K$ )和进行  $f$  迭代计算的允许误差( $DW$ )。对于菲利浦型下渗曲线,下渗能力计算参数是  $A$  和  $B$ 。另外,还有计算时段长( $\Delta t$ )和流域面积;时变参数有土壤含水量( $W$ );模型输出为出流量( $R$ )的时间序列。其推流步骤是:

首先根据时段降雨( $P$ )和蒸发能力( $EM$ ),计算有效降雨  $PE$ ,  $PE = P - EM$ 。

若  $PE \leq 0$ , 则  $R = 0$ , 用  $W$  进行蒸发计算,求下时段的土壤含水量。

否则,根据初始土壤含水量( $W$ ),用式(4-3)或式(4-6)计算下渗能力( $f$ )。把求得的  $f$  值与  $PE$  相对比,计算产流量和下渗量。下渗量补给土壤含水量。

对于由菲利浦公式转化的公式(4-6),只要给出一组系数  $A$ 、 $B$  的值,便可直接得出  $f \sim W$  的关系,以计算径流量。而用霍尔顿转化后的关系式(4-3),需用迭代才能求出  $f \sim W$  的关系。迭代过程是,以  $T = W/f_0$  作为  $t$  的第一次近似值,即可由式(4-2)得出土壤含水量( $W$ )的第一个近似值  $ST$ , 如果  $|ST - W| > \text{允许误差}(DW)$ , 则由公式(4-1)计算  $f$  的第一个近似值  $U$ , 然后  $T = T + (W - ST)/U$ , 这样迭代多次,直到  $|ST - W| \leq DW$ , 即可求得所需的  $f$  值。

## 二、降雨和下垫面空间分布不均匀的考虑

实际上,降雨和下垫面条件在流域上的分布是不均匀的。尤其是在干旱地区,暴雨多为小尺度天气系统所形成,笼罩面积小,历时短,强度大。所以,要求计算时段( $\Delta t$ )和流域的单元面积都要尽可能的小。尽管如此,也难以满足点降雨径流模型的要求。假定流域平均下渗能力( $f_i$ )与流域平均蓄水量( $W_i$ )的关系为霍尔顿或菲利浦型,即符合式(4-3)或式(4-6),而在任何时刻,各点的下渗能力在流域上的分布为抛物线(见图4-2),这种曲线称作流域下渗能力分布曲线。其公式为

$$\frac{F_A}{F} = 1 - \left(1 - \frac{f'_m}{f_{mm}}\right)^{BX} \quad (4-7)$$

流域平均下渗能力 $f_i$ 为

$$f_i = \frac{f_{mm}}{1 + BX} \quad (4-8)$$

式中, $f'_m$ 为流域平均下渗能力为 $f_i$ 时的某一点的下渗能力; $f_{mm}$ 为流域平均下渗能力为 $f_i$ 时流域内最大的点下渗能力; $F_A$ 为下渗能力 $\leq f'_m$ 的流域面积; $F$ 为流域面积; $BX$ 为抛物线指数。

只有在下渗能力小于降雨强度( $PE$ )的地区才产流。用 $R$ 表示流域产流量,其计算公式为




图4-2 某时刻的流域下渗能力分布曲线

当  $PE \leq 0$  时, 不产流, 即

$$\left. \begin{aligned} R &= 0 \\ W_{i+1} &= W_i + PE \end{aligned} \right\} \quad (4-9)$$

当  $PE > 0$  时,

若  $PE < f_{mm}$ , 局部产流,

$$\left. \begin{aligned} R &= \int_0^{PE} \frac{F_A}{F} df_m = PE - f_i + f_i \left(1 - \frac{PE}{f_{mm}}\right)^{BX+1} \\ W_{i+1} &= W_i + PE - R \end{aligned} \right\} \quad (4-10)$$

若  $PE \geq f_{mm}$ , 全流域产流,

$$\left. \begin{aligned} R &= PE - f_i \\ W_{i+1} &= W_i + f_i \end{aligned} \right\} \quad (4-11)$$

上面讨论的仅是在流域平均下渗能力为某一定值  $f_i$  时的流域下渗能力分布曲线。换言之, 图 4-2 仅是对应于某一流域平均蓄水量 ( $W_i$ ) 时的一条流域下渗能力分布曲线。所以说, 流域的下渗能力分布曲线不只一条, 而是一簇, 如图 4-3。其中最上面一条和最下面一条分别对应于流域平均下渗能力为最大下渗能力 ( $f_0$ ) 和稳定下渗能力 ( $f_c$ ) 时的流域下渗能力分布曲线。其它任何时刻的分布曲线都介于这两条线之间。在  $BX=0$  时, 就代表了点降雨径流模型。




图 4-3 流域下渗能力分布曲线

下渗能力分布曲线的参数  $BX$ 。

### 三、陕北模型推流步骤

陕北模型的输入, 输出以及参数与点降雨径流模型相同。只是参数输入中多了一个描述流域下渗能力分布曲线的参数  $BX$ 。




图 4-4 陕北模型计算程序框图

其推流步骤为:

(1)根据时段降雨( $P$ )和时段的流域可能蒸发能力( $EM$ ),计算有效降雨( $PE$ ),  $PE = P - EM$ 。

(2)若  $PE \leq 0$ , 径流量  $R = 0$ , 用流域的土壤含水量( $W_t$ )进行蒸发计算, 求出时段末土壤含水量( $W_{t+1}$ ), 结束本时段计算。

(3)若  $PE > 0$ , 由  $W_t$  用式(4-3)或式(4-6)计算流域平均下渗能力( $f_t$ )。

(4)根据  $f_t$  和  $BX$ , 用式(4-8)计算  $f_{mm}$ 。

(5)根据  $PE$  和  $f_{mm}$  值, 选用式(4-10)或式(4-11)计算时段产流量( $R$ )和时段末土壤含水量( $W_{t+1}$ )。

(6)重复上述 1~5 的步骤, 进行下时段计算。详细情况见图 4-4 的陕北模型计算程序框图。

## 第二节 模型计算程序

陕北模型虽然用下渗能力分布曲线考虑下渗能力在流域上的不均匀分布, 但由于干旱区降水的时空分布极不均匀, 所以, 要使计算结果具有一定的精度, 必须在降雨资料许可的情况下尽可能使计算时段  $\Delta t$  取的小。在保证每个单元的流域面积大致相等, 各单元至少有一个雨量站的情况下, 尽可能把流域单元面积划分的小。对于具有多个雨量站的单元, 最好用单元内的全部雨量站计算出单元的面平均降雨, 作单元降雨径流模型的降雨输入。

把每个单元计算出来的产流量( $R$ ), 直接用单元面积的无因次时段单位线把产流演进到单元出口。

### 一、计算程序

```
SUBROUTINE SBMX(PAR, IK, N, M, P, EM, UH, Q)
REAL * 4 K, PAR(8), P(N), EM(N), UH(M), Q(N)
IF (IK.EQ. 1) THEN
```

```

F0=PAR(1)
FC=PAR(2)
K=PAR(3)
DW=PAR(4)
BX=PAR(5)
AREA=PAR(6)
WT=PAR(7)
DT=PAR(8)
ELSE
A=PAR(1)
B=PAR(2)
BX=PAR(5)
AREA=PAR(6)
WT=PAR(7)
DT=PAR(8)
ENDIF
DO 1000 L=1,N
PE=P(L)-EM(L)
IF(PE.LE.0) THEN
WT=WT+P(L)-EM(L)
IF(WT.LT.0) WT=0.
R=0
ELSE
IF(IK.EQ.1) THEN
F1=F0
F2=FC
T0=WT/f0
II=0
10 ST=FC*T0+(1-exp(-K*T0))*(F0-FC)/K
II=II+1
IF(II.LE.100)GOTO 15

```

```

WRITE(*,*)'已迭代 100 次, 不收敛 !!'
STOP
15  CONTINUE
    IF (ABS(ST - WT).GT.DW) THEN
        F=F0-K*(ST-FC*T0)
        T0=T0+(WT-ST)/F
        GOTO 10
    ELSE
        F=F0-K*(WT-FC*T0)
    ENDIF
    ELSE
        F=B**2*(1+SQRT(1+A*WT/B**2))/WT+A
    ENDIF
    F=F*DT
    FMM=F*(1+BX)
    IF (PE.GT.FMM) THEN
        R=PE-F
        WT=WT+F
    ELSE
        R=PE-F+F*(1-PE/FMM)**(1+BX)
        WT=WT+PE-R
    ENDIF
    ENDIF
    DO 20 J=1,M
        IF((L+J-1).GT.N)GOTO 12
20   Q(L+J-1)=Q(L+J-1)+R*UH(J)
12   CONTINUE
1000 CONTINUE
        DO 1010 I=1,L
            Q(I)=Q(I)*AREA/3.6/DT
1010 CONTINUE

```



RETURN

END

## 二、程序说明

变量	类型	大小	I/O	内 容
$IK$	整型	1	I	下渗曲线的选型变量： =1, 霍尔顿型下渗曲线； ≠1, 菲利浦型下渗曲线
$PAR$	实型	8	I	模型参数 $IK=1$ , 则： $PAR(1)$ ——流域平均最大下渗能力 ( $f_0$ )(mm/h)； $PAR(2)$ ——流域平均稳定下渗能力 ( $f_c$ )(mm/h)； $PAR(3)$ ——下渗能力衰减系数 $K$ ( $h^{-1}$ )； $PAR(4)$ ——迭代计算时, 对土壤含水量 的允许误差 (mm)； $PAR(5)$ ——流域下渗能力分布曲线 指数 $BX$ ； $PAR(6)$ ——流域面积( $km^2$ )； $PAR(7)$ ——土壤含水量(mm)； $PAR(8)$ ——计算时段步长(h)。 $IK \neq 1$ , 则： $PAR(1)$ ——菲利浦下渗曲线参数 ( $A$ )(mm/h)；

$PAR(2)$ ——菲利浦下渗曲线参数  
 $(B)(mm/h^{\frac{1}{2}})$ ;

$PAR(3)$ 和  $PAR(4)$ 无效;

$PAR(5) \sim PAR(8)$ 同  $IK=1$  时的内容

$N$	整型	1	$I$	计算时段数
$M$	整型	1	$I$	单位线底宽(时段数)
$P$	实型	$N$	$I$	时段降雨(mm)
$EM$	实型	$N$	$I$	时段流域蒸散发能力(mm)
$UH$	实型	$M$	$I$	无因次时段单位线
$Q$	实型	$N$	$O$	出流量( $m^3/s$ )

### 第三节 实 例

#### 一、主程序

下面是陕北模型主程序, 它从  $SB.PAR$  文件中读取陕北模型参数, 从  $SB.DAT$  读取单元面积上的蒸发能力和面平均降水数据, 通过调用  $SBMX$  子程序计算出单元出流量过程, 结果写入  $SB.OUT$ 。

```

INTEGER YEAR,EM,ED,EH
DIMENSION PAR(8),UH(20),PP(100),EP(100),Q(100)
OPEN(1,FILE='SB.PAR',STATUS='OLD')
OPEN(2,FILE='SB.DAT',STATUS='OLD')
OPEN(3,FILE='SB.OUT',STATUS='NEW')
READ(1,*)IK,M

```

```

      READ(1,*)PAR
      READ(1,*)(UH(I),I=1,M)
      READ(2,*)YEAR,IM,ID,IH,EM,ED,EH,N
      READ(2,*)(EP(I),I=1,N)
      READ(2,*)(PP(I),I=1,N)
      CALL SBMX(PAR,IK,N,M,PP,EP,UH,Q)
      DT=PAR(8)
      WRITE(3,110)YEAR,IM,ID,IH,EM,ED,EH,DT
110  FORMAT(5X,I4,'年',I2,'月',I2,'日',I2,'时',
      + '—',I2,'月',I2,'日',I2,'时',' DT= ',F5.2,'h',
      + '流量过程')
      WRITE(3,'(8F6.1)')(Q(I),I=1,N)
      STOP
      END

```

## 二 实例说明

### 1. 参数文件 SB.PAR

```

1      20
156.   24.   6.1   0.05  1.    187.   15.   0.2
0.0    0.0    0.0    0.02  0.03  0.08  0.12  0.15  0.12  0.10
0.08  0.07  0.06  0.05  0.04  0.03  0.02  0.01  0.01  0.01

```

上面是黄河中游无定河上某一小流域的陕北模型参数。第一行分别为下渗曲线的选型变量( $IK$ )和单位线底宽( $M$ )；第二行是模型参数变量( $PAR$ )；后面( $M$ )个数字为无因次单位线。

### 2. 数据文件 SB.DAT

```

1977  8  11  17  8  11  22  26
26 * 0.
0.0  3.5  5.7  5.7  5.7  4.3  2.5  19 * 0.

```

SB.DAT 文件中第一行分别是计算开始的年、月、日、时和结束月、日、时及计算历时；第二行和第三行分别为流域蒸发能力和单元平均降水过程。

### 3. 输出文件 SB.OUT

1977 年 8 月 11 日 17 时~8 月 11 日 22 时 DT=0.20h 流量过程

0.0	0.0	0.0	0.0	1.0	4.5	12.5	29.6
54.3	83.2	106.0	116.0	109.1	93.6	77.6	65.2
54.8	45.2	35.5	26.4	18.7	12.7	8.4	5.6
2.8	0.8						

上面是陕北模型计算的单元出流量过程。

## 第五章 降雨径流相关图法

### 第一节 降雨径流相关图

#### 一、概述

前面几章分别介绍了新安江(三水源)模型、萨克模型、水箱模型和陕北模型。这些都是多年来国内应用较为广泛的几种降雨径流计算模型。但是,不少生产单位,尤其是一些大型水库的管理单位,他们在长期的工作实践中已建立了一套适合于当地实际情况的经验降雨径流预报方案。降雨径流经验关系曲线有各种形式,一般有产流量  $R=f(\text{次雨量}(P), \text{前期影响雨量}(P_n), \text{季节, 温度})$ 、产流量  $R=f(\text{前期影响雨量}(P_n), \text{洪水起涨流量}(Q_0))$  和考虑雨强的超渗式关系曲线形式。这里仅介绍国内最为普遍使用的产流量与降雨和前期影响雨量三者的关系,即  $P \sim P_n \sim R$  相关图(见图 5—1)。

使用  $P \sim P_n \sim R$  关系曲线进行净雨量计算一般有两种处理途径:一种是根据洪水初的  $P_n$  值,把时段雨量序列变成累积雨量序列,用累积雨量查出累积净雨,由累积净雨再转化成时段净雨量序列;另一种方法是根据时段降雨序列资料直




图 5—1 降雨径流相关图

接推求时段净雨序列。第一种方法的缺点是在整个洪水过程中,使用一条  $P \sim R$  曲线,没有考虑洪水期中  $P_a$  的变化。而后者的不足是,当时段取的过小时,一般时段雨量不大,推求净雨时的查线计算易集中在曲线的下段。两种方法的结果存在差别,至于何者更接近实际也很难断言。程序中包括了两种方案,用户可以根据情况任选。

## 二、 $P_a$ 的计算

$P_a$  由前期雨量计算,也称前期影响雨量,是反映土壤湿度的参数。其计算公式为

当上时段有雨量,即  $P_{t-1} > 0$  时,

$$P_{a,t} = K \cdot (P_{a,t-1} + P_{t-1}) \quad (5-1)$$

当上时段无雨时,

$$P_{a,t} = K \cdot P_{a,t-1} \quad (5-2)$$

式中,  $K$  为土壤含水量衰减系数,对于日模型,一般地,  $K \approx 0.85$ ;  $P_{a,t-1}$ 、 $P_{a,t}$  分别为上时段和本时段的前期影响雨量;  $P_{t-1}$  为上时段降雨量。

用  $I_m$  表示土壤最大初损量,以 mm 计,通常  $I_m = 60 \sim 100\text{mm}$ 。当计算的  $P_{a,t} > I_m$  时,则以  $I_m$  作  $P_a$  值计算,即认为,此后的  $P$  不再补充初损量,全部形成  $R$ 。

当计算时段长  $\Delta t \neq 24\text{h}$  时,土壤含水量衰减系数  $K$  应该用下式换算

$$K = KD^{\frac{1}{N}} \quad (5-3)$$

式中,  $N = 24/\Delta t$ ,  $KD$  为土壤含水量日衰减系数,  $K$  为计算时段是  $\Delta t$  小时的土壤含水量衰减系数。

## 三、相关图推流计算

用  $P \sim P_a \sim R$  相关图(见图 5-1)作流域降雨径流计算的步




图 5-2  $P \sim P_a \sim R$  曲线降雨径流计算流程图

骤是：首先摘录每条  $P \sim R$  曲线的各点坐标，把  $P \sim P_a \sim R$  曲线坐标和土壤含水量衰减系数( $K$ )、土壤最大初损量( $I_m$ )、计算时段长( $\Delta t$ )、以及计算开始时的前期影响雨量( $P_a$ )，作为模型参数输入给计算程序中的相应变量，然后根据时段降雨序列计算出每个时段的前期土壤指数和净雨量。计算过程见图 5-2 所示的  $P \sim P_a \sim R$  曲线降雨径流计算流程图。

## 第二节 $P \sim P_a \sim R$ 计算程序

该程序是把  $P \sim P_a \sim R$  相关曲线程序化，即把每条  $P \sim R$  曲线的坐标点摘录成模型参数，然后根据初始的前期土壤湿度指标，由降雨序列推求出相应的净雨过程。考虑到各地区现有单位线资料不一，有的是有因次单位线，有的是无因次单位线，有些地方是有数条单位线供选用。为了用户便于使用，本程序仅计算出净雨过程，没有用单位线转换成流量。用户可把该程序与单位线推流计算程序结合使用，方便地把本区现有的  $P \sim P_a \sim R$  曲线和经验单位线改成降雨~径流计算程序。

### 一、 $P \sim P_a \sim R$ 程序

```

SUBROUTINE PPAR(LENGTH,M,N,IM,RATIO,DELTAT,
+ W0,P0,R,PA,P,RUNOFF,FLAG)
REAL * 4 R(50,N),W0(N),P0(M),PA,RESULT,IM
REAL * 4 P(LENGTH),RUNOFF(LENGTH),RATIO
INTEGER FLAG
IF(FLAG.EQ.1) THEN
DO 100 I=2,LENGTH
P(I)=P(I)+P(I-1)
100 CONTINUE
ENDIF

```



```

DO 1000 L=1,LENGTH
IF(P(L).LT.0) P(L)=0.
IF(PA.LT.W0(1)) PA=W0(1)
N1=0
DO 10 I=1,N
IF(ABS(PA-W0(I)).LT.0.001) N1=I
10 CONTINUE
IF(N1.NE.0) THEN
DO 20 I=1,M-1
IF(P(L).GE.P0(I).AND.P(L).LE.P0(I+1)) THEN
RUNOFF(L)=R(I,N1)+(P(L)-P0(I))*(R(I+1,N1)-
+ R(I,N1))/(P0(I+1)-P0(I))
ELSE IF(P(L).LT.P0(I)) THEN
RUNOFF(L)=0.
ELSE IF(P(L).GT.P0(M)) THEN
RUNOFF(L)=R(M,N1)+(P(L)-P0(M))*(R(M,N1)-
+ R(M-1,N1))/(P0(M)-P0(M-1))
ENDIF
20 CONTINUE
ELSE
DO 30 I=1,N-1
IF(PA.GT.W0(I).AND.PA.LT.W0(I+1)) THEN
N1=I
ENDIF
30 CONTINUE
IF(PA.GT.W0(N).AND.PA.LE.IM) N1=N
DO 40 I=1,M-1
IF(P(L).GE.P0(I).AND.P(L).LE.P0(I+1)) THEN
R1=R(I,N1)+(P(L)-P0(I))*(R(I+1,N1)-R(I,N1))
+ /(P0(I+1)-P0(I))
ELSE IF(P(L).LT.P0(1)) THEN

```

```

R1=0.
ELSE IF (P(L).GT.P0(M)) THEN
R1=R(M,N1)+(P(L)-P0(M))*(R(M,N1)-R(M-1,N1))
+ /(P0(M)-P0(M-1))
ENDIF
40 CONTINUE
IF(N1.LT.N) THEN
DO 50 I=1,M-1
IF(P(L).GE.P0(I).AND.P(L).LE.P0(I+1)) THEN
R2=R(I,N1)+(P(L)-P0(I))*(R(I+1,N1)-R(I,N1))
+ /(P0(I+1)-P0(I))
ELSE IF (P(L).LT.P0(1)) THEN
R2=0.
ELSE IF (P(L).GT.P0(M)) THEN
R2=R(M,N1)+(P(L)-P0(M))*(R(M,N1)
+ -R(M-1,N1))/(P0(M)-P0(M-1))
ENDIF
50 CONTINUE
ELSE
R2=P(L)
ENDIF
IF(N1.LT.N) THEN
RUNOFF(L)=(R2-R1)/(W0(N1+1)-W0(N1))*(PA-
+ W0(N1))+R1
ELSE
RUNOFF(L)=(R2-R1)/(IM-W0(N1))*(PA-
+ W0(N1))+R1
ENDIF
ENDIF
IF(FLAG.NE.1) THEN
PA=RATIO** (DELTAT/24.)*(PA+P(L))

```

```

        IF(PA.GT.IM) PA=IM
        ENDIF
1000  CONTINUE
        IF(FLAG.EQ.1) THEN
        DO 200 I=LENGTH,2,-1
        RUNOFF(I)=RUNOFF(I)-RUNOFF(I-1)
200  CONTINUE
        ENDIF
        END

```

## 二、程序说明

变量	类型	大小	I/O	内 容
<i>LENGTH</i>	整型	1	<i>I</i>	计算时段数
<i>M</i>	整型	1	<i>I</i>	每条 $P \sim R$ 曲线的节点数, $M \leq 50$
<i>N</i>	整型	1	<i>I</i>	$P \sim R$ 曲线条数
<i>IM</i>	实型	1	<i>I</i>	最大初损(mm)
<i>RATIO</i>	实型	1	<i>I</i>	前期土壤湿度的日衰减系数
<i>DELTA T</i>	实型	1	<i>I</i>	计算时段长(h)
<i>W0</i>	实型	<i>N</i>	<i>I</i>	每条曲线的 $P_0$ 值(以增序 排列)(mm)
<i>P0</i>	实型	<i>M</i>	<i>I</i>	$P \sim P_0 \sim R$ 曲线中的 $P$ 坐标 值(mm)
<i>R</i>	实型	$50 \times N$	<i>I</i>	$P \sim P_0 \sim R$ 各条曲线中对应 于 $P$ 坐标的 $R$ 值(mm)
<i>PA</i>	实型	1	<i>I/O</i>	土壤湿度指标(mm)

<i>P</i>	实型	<i>LENGTH</i>	<i>I</i>	时段降雨量(mm)
<i>RUNOFF</i>	实型	<i>LENGTH</i>	<i>O</i>	时段净雨量(mm)
<i>FLAG</i>	整型	1	<i>I</i>	查曲线方式代码, =1,用累计雨量查曲线; ≠1,用时段雨量查曲线

### 第三节 实 例

#### 一、主程序

下面是使用  $P \sim P_n \sim R$  曲线进行降雨产流计算的主程序, 它从 PPAR.PAR 文件中读取模型参数, 从 PPAR.DAT 中读取单元面积上的平均降雨, 通过调用 PPAR 子程序计算出单元净雨过程, 结果写入 PPAR.OUT。

```

INTEGER YEAR,EM,ED,EH,FLAG
REAL IMA
DIMENSION W0(20),P0(50),R(50,20),P(100),RUNOFF(100)
OPEN(1,FILE='PPAR.PAR',STATUS='OLD')
OPEN(2,FILE='PPAR.DAT',STATUS='OLD')
OPEN(3,FILE='PPAR.OUT',STATUS='NEW')
READ(1,*)M,N,IMA,RATIO,DT
READ(1,*)(W0(I),I=1,N)
READ(1,*)(P0(I),I=1,M)
READ(1,*)((R(I,J),I=1,M),J=1,N)
READ(1,*)PA0,FLAG
READ(2,*)YEAR,IM,ID,IH,EM,ED,EH,LENGTH
READ(2,*)(P(I),I=1,LENGTH)
CALL PPAR(LENGTH,M,N,IMA,RATIO,DT,W0,

```

```

+ P0,R,PA0,P,RUNOFF,FLAG)
  WRITE(3,110)YEAR,IM,ID,IH,EM,ED,EH,DT
110  FORMAT(5X,I4,'年',I2,'月',I2,'日',I2,'时','-',
+ I2,'月',I2,'日',I2,'时',' DT=' ,F5.2,'h',
+ '产流过程')
  WRITE(3,'(10F8.1)')(RUNOFF(I),I=1,LENGTH)
  STOP
  END

```

## 二、实例说明

### 1. 参数文件 PPAR.PAR

21	10	90.	0.9	2.							
0.0	10.	20.	30.	40.	50.	60.	70.	80.	90.		
0.0	10.	20.	30.	40.	50.	60.	70.	80.	90.	100.	
110.	120.	130.	140.	150.	160.	170.	180.	190.	200.		
0.0	0.0	0.0	0.0	0.0	0.2.	0.7	1.2	2.8	4.8	7.8	
11.5	16.8	23.5	32.0	41.0	51.0	61.0	71.0	81.0	91.0		
0.0	0.0	0.5	1.0	1.2	2.2	3.5	5.0	7.0	10.0	13.7	
18.5	25.0	32.5	41.5	51.5	61.5	71.5	81.5	91.5	101.		
0.0	0.5	1.0	2.2	3.5	5.0	7.0	9.5	12.5	16.2	21.0	
27.0	34.0	42.0	51.0	61.0	71.0	81.0	91.0	101.	111.		
0.0	1.5	2.2	4.0	6.0	8.0	11.0	14.0	18.0	22.5	28.5	
35.0	43.0	51.5	61.0	71.0	81.0	91.0	101.	111.	121.		
0.0	2.0	3.5	5.8	8.5	11.8	15.0	19.0	24.0	29.8	36.0	
44.0	52.0	61.0	71.0	81.0	91.0	101.	111.	121.	131.		
0.0	2.5	5.0	8.5	12.0	16.5	20.0	25.5	31.0	38.0	45.0	

53.5	62.0	71.0	81.0	91.0	101.	111.	121.	131.	141.	
0.0	3.5	7.5	11.8	16.0	21.0	26.0	32.0	39.0	46.0	54.0
63.0	72.0	81.5	91.0	101.	111.	121.	131.	141.	151.	
0.0	5.5	10.5	15.2	21.0	26.5	32.5	39.2	46.5	54.0	62.5
72.0	81.0	91.0	101.	111.	121.	131.	141.	151.	161.	
0.0	6.5	12.0	18.5	25.0	32.0	40.0	47.2	55.0	63.0	71.5
81.0	91.0	101.	111.	121.	131.	141.	151.	161.	171.	
0.0	7.5	14.7	22.0	30.0	38.0	46.5	54.5	63.0	71.5	81.0
91.0	101.	111.	121.	131.	141.	151.	161.	171.	181.	
26.	1									

上面是从黄河流域伊河上陆浑至龙门镇区间(面积  $1826\text{km}^2$ ) 的  $P \sim P_s \sim R$  曲线上摘取的坐标点。第一行参数分别为每条  $P \sim R$  曲线的节点数( $M$ )和曲线条数( $N$ ), 土壤最大初损量( $I_m$ )、土壤含水量日衰减系数( $KD$ )和计算时段步长( $DT$ ); 第二行  $N$  个参数值分别为每条  $P \sim R$  曲线对应的  $P_s$  值; 第三、四行是  $P_0(I)$  值; 接下来  $M \times N$  个数据是  $R(I, J)$  值; 最后一行的两个数分别为计算开始时的初始土壤含水量( $P_s$ )和查曲线方式代码(FLAG)。

## 2. 数据文件 PPAR.DAT

1982	7	30	0	7	31	12	19		
12.3	27.5	44.7	40.0	34.0	13.2	9.4	6.4	19.8	6.2
2.6	4.2	2.8	4.6	9.2	4.8	8.5	6.5	6.8	

PPAR.DAT 文件中第一行分别为计算开始年、月、日、时和结束月、日、时、计算的时段数。第二、三行是由陆浑、青沟、禹山、石湾镇、黑龙沟、鸣皋、白杨、上蔡店、颍阳、江左、吕店、

鸦岭、白沙、彭婆、龙门镇 15 个站算术平均的流域降雨量过程。

### 3. 输出文件 PPAR.OUT

1982 年 7 月 30 日 0 时~7 月 31 日 12 时  $DT = 2.00h$  产流过程

0.6    2.9    10.7    23.4    31.9    13.2    9.4    6.4    19.8    6.2

2.6    4.2    2.8    4.6    9.2    4.8    8.5    6.5    6.8

上面是使用  $P \sim P_s \sim R$  曲线, 由累计降雨查得的陆浑至龙门镇区间的净雨过程。

## 第六章 水库调洪演算

### 第一节 计算方法

为了充分开发、利用水资源，我国各地兴建了大量的大、中、小型水库。这些水库在防洪、灌溉、发电、养殖、航运等方面都起到了重要的作用。但是，如果入库洪水预报和水库调度得不当，也会给库区和下游两岸人民带来严重损失。所以，为了充分发挥水库效益，造福于人类，必须认真研究每个水库的预报和调度方案。本章主要介绍大中型水库洪水预报中的水库调洪演算方法。

水库调洪计算的依据是水量平衡方程。水库的入流由上游测站来水和区间来水（包括库水面降雨形成的径流）两部分组成。如果在洪水期，计算时间不长，可以忽略水面蒸发。水库的出流是各个泄水建筑物的总泄流量。水库的水量平衡方程为

$$\bar{Q}\Delta t - \bar{q}\Delta t = \Delta V \quad (6-1)$$

式中， $\bar{Q}$  为时段内入库流量平均值； $\bar{q}$  为时段内出库流量平均值； $\Delta V$  为时段内水库蓄量的变化值； $\Delta t$  为计算时段长。

如果水库的蓄量与出流关系单一，并假定入流和出流在计算时段内呈线性变化，式(6-1)可改写为

$$\frac{V_2}{\Delta t} + \frac{q_2}{2} = \frac{V_1}{\Delta t} + \frac{q_1}{2} + \bar{Q} - q_1 \quad (6-2)$$

式中，各变量的下标 1、2 分别表示其时段始、末值。

对于某一水库，可根据其泄流曲线  $H \sim q$  及库容曲线  $H \sim V$ ，计算出一定推流时段  $\Delta t$  下的水库调洪计算曲线(图 6-1)。






图 6-1 水库调洪计算曲线

使用图 6-1 作调洪计算的步骤是：已知时段初水位  $H_1$ ，分别由  $q=f(H)$  和  $\frac{V}{\Delta t} + \frac{q}{2} = f(H)$  求出  $q_1$  和  $\frac{V_1}{\Delta t} + \frac{q_1}{2}$ 。又知  $\bar{Q}$ ，因此，式(6-2)右方全系已知数，其总和等于  $\frac{V_2}{\Delta t} + \frac{q_2}{2}$ 。可由  $\frac{V_2}{\Delta t} + \frac{q_2}{2}$  值再查  $\frac{V}{\Delta t} + \frac{q}{2} = f(H)$  得时段末水位  $H_2$ ，由  $H_2$  查  $V=f(H)$  和  $q=f(H)$  曲线，得出时段末水库出流量  $q_2$  和水库蓄量  $V_2$ 。

由水量平衡方程可知，当入库流量大于出库流量时，库水位增加，反之，库水位降低。出库流量过程和库水位及蓄水量的预报精度之高低，关键取决于入库流量的预报和库水位的代表性。

## 第二节 水库调洪计算程序

水库调洪计算一般是已知入库流量过程和水库特性曲线,根据初始库水位和水库泄水建筑物的启闭情况,计算出库流量、库水位和库蓄水量过程。这种水库调洪计算可按前面所讲的调洪步骤计算。不过在通常情况下,泄水建筑物在一次洪水过程中的启闭状态是不断变化的, $q=f(H)$ 和 $\frac{V}{\Delta t} + \frac{q}{2} = f(H)$ 都非单一。所以,就不能简单地使用公式(6-2)进行水库调洪计算,必须作好换线处理。

本程序是一个灵活通用的水库调洪计算程序,它的输入参数包括水位与库容、水位与各类泄水建筑物的泄流能力曲线坐标。输入的时变参数是计算开始时的初始库水位和在一次洪水过程中拟定或实际的闸门启闭情况。输入的水文数据为入库流量过程。输出是既定调度方案下的出库流量、库水位、库蓄水量过程。它特别适合于水库的实时调度计算。

### 一、计算程序

```
SUBROUTINE RSV(MDR2,NPOINT,PA,DLT,NT,H1,NOP,  
+ OP,QI,HVQ,DMID)  
  DIMENSION PA(100,MDR2),OP(11,NOP),QI(NT),  
+ HVQ(3,NT),DMID(100,2)  
  NDRIVE=MDR2-2  
  DT1=1000./(DLT*3.6)  
  OP(1,1)=1  
  IOP=1  
  DO 500 INT=1,NT  
    IF(IOP.GT.NOP)GOTO 200
```

```

      II=OP(1,IOP)
      IF(INT.NE.II)GOTO 200
      DO 110 I=1,NPOINT
110    DMID(I,1)=0.
      DO 120 I=1,NPOINT
      DO 120 J=1,NDRIVE
      DMID(I,1)=DMID(I,1)+PA(I,J+2)*OP(J+1,IOP)
120    CONTINUE
      DO 130 I=1,NPOINT
      DMID(I,2)=PA(I,2)*DT1+DMID(I,2)/(2.*100.)
130    CONTINUE
      IOP=IOP+1
200    IF(INT.NE.1)GOTO 240.
      CALL ZXCZ(NPOINT,PA(1,1),DMID(1,1),0,H1,QO1)
      CALL ZXCZ(NPOINT,PA(1,1),PA(1,2),0,H1,V1)
      CALL ZXCZ(NPOINT,PA(1,1),DMID(1,2),0,H1,VQ1)
      HVQ(1,INT)=QO1
      HVQ(2,INT)=H1
      HVQ(3,INT)=V1
      QI1=QI(INT)
      GOTO 250
240    CONTINUE
245    QI2=QI(INT)
      QIM=(QI1+QI2)/2.
      VQ2=VQ1+(QIM-QO1)/100.
      CALL ZXCZ(NPOINT,DMID(1,2),PA(1,1),0,VQ2,H2)
      CALL ZXCZ(NPOINT,PA(1,1),DMID(1,1),0,H2,QO2)
      CALL ZXCZ(NPOINT,PA(1,1),PA(1,2),0,H2,V2)
      HVQ(1,INT)=QO2
      HVQ(2,INT)=H2
      HVQ(3,INT)=V2

```

```

        QI1=Q1(INT)
        H1=H2
        V1=V2
        VQ1=VQ2
        QO1=QO2
250    CONTINUE
500    CONTINUE
        RETURN
        END

```

## 二、程序说明

变量	类型	大小	I/O	内 容
<i>MDR2</i>	整型	1	I	参数 <i>PA</i> 数组的列数, 等于泄水建筑物种类加 2
<i>NPOINT</i>	整型	1	I	泄流能力及其它曲线每条线摘录的坐标节点数; ≤100
<i>PA</i>	实型	100× <i>MDR2</i>	I	水库特性曲线坐标, 其中: 第 1 列是库水位(m); 第 2 列是库容量( $\times 10^8 \text{m}^3$ ); 第 <i>I</i> +2 列表表示第 <i>I</i> 类泄水建筑物与水位对应的单孔泄流能力( $\text{m}^3/\text{s}$ )
<i>DLT</i>	实型	1	I	计算时段长(h)
<i>NT</i>	整型	1	I	计算时段数
<i>H1</i>	实型	1	I	初始库水位(m)

<i>NOP</i>	整型	1	<i>I</i>	<i>OP</i> 数组的列数,等于一次计算洪水过程中闸门不同开启状态数
<i>OP</i>	实型	$11 \times NOP$	<i>I</i>	闸门启闭状态参数,其中: $OP(1,J)$ 表示第 $J$ 次运行状态的开始时段。 $OP(I+1,J)$ 表示第 $J$ 次运行状态第 $I$ 类泄水建筑物开启的孔数
<i>QI</i>	实型	$NT$	<i>I</i>	入库流量( $m^3/s$ )
<i>HVQ</i>	实型	$3 \times NT$	<i>O</i>	调洪演算结果,其中: 第 1 行为库水位( $m$ ); 第 2 行为库蓄水量( $\times 10^8 m^3$ );第 3 行为 出库流量( $m^3/s$ )
<i>DMID</i>	实型	$100 \times 2$		中间变量数组

注:①程序内部调用了一个直线插值程序 ZXCZ。

②泄水建筑物的种类不得多于 10。

### 第三节 实 例

#### 一、主程序

下面是水库调洪计算主程序,它从 RSV.PAR 文件中读取水库特性曲线等参数,从 RSV.DAT 文件中读取入库流量过程、初始库水位及洪水期间的水库运用方式,调用 RSV 子程序,计算

出既定运用方案下的出库流量及蓄水过程, 结果写入 RSV.OUT。

```
      INTEGER YEAR,EM,ED,EH
      DIMENSION PA(100,12),OP(11,20),QI(120),HVQ(3,120),
+ DMID(100,2)
      OPEN(1,FILE='RSV.PAR',STATUS='OLD')
      OPEN(2,FILE='RSV.DAT',STATUS='OLD')
      OPEN(3,FILE='RSV.OUT',STATUS='NEW')
      READ(1,*)MDR,NPOINT
      MDR2=MDR+2
      DO 10 J=1,MDR2
      READ(1,*)(PA(I,J),I=1,NPOINT)
10    CONTINUE
      READ(2,*)YEAR,IM,ID,IH,EM,ED,EH,NT
      READ(2,*)(QI(I),I=1,NT)
      READ(2,*)DLT,H1,NOP
      MDR1=MDR+1
      DO 20 J=1,NOP
      READ(2,*)(OP(I,J),I=1,MDR1)
20    CONTINUE
      CALL RSV(MDR2,NPOINT,PA,DLT,NT,H1,NOP,OP,
+ QI,HVQ,DMID)
      WRITE(3,30)
30    FORMAT(1X,'时序    入流(m3/s)  出流(m3/s)',
+ '水位(m)  蓄量(×108m3)')
      DO 40 J=1,NT
40    WRITE(3,45)J,QI(J),(HVQ(I,J),I=1,3)
45    FORMAT(1X,I4,3(2X,F11.2),F11.4)
      STOP
      END
```

## 二、实例说明

### 1. 参数文件 RSV.PAR

6	22										
280.	290.	295.	300.	302.	304.	306.	308.	310.	312.	314.	
316.	318.	320.	322.	324.	326.	328.	330.	332.	334.	335.	
0.	0.01	0.10	0.14	0.25	0.41	0.67	1.02	1.48	2.09	2.85	
3.82	5.08	7.14	10.1	14.0	18.9	24.8	31.5	39.6	51.6	59.7	
0.	0.	0.	0.	17.	40.	70.	105.	144.	190.	240.	
275.	306.	337.	360.	385.	410.	430.	455.	470.	493.	503.	
0.	110.	205.	270.	310.	355.	405.	470.	535.	585.	630.	
680.	720.	760.	800.	835.	870.	905.	935.	960.	975.	985.	
0.	110.	205.	270.	292.	310.	330.	348.	365.	382.	398.	
412.	427.	440.	455.	467.	482.	494.	505.	518.	528.	535.	
0.	0.	0.	0.	0.	0.	0.	0.	199.	211.	220.	
229.	238.	246.	253.	260.	266.	271.	277.	283.	288.	290.	
0.	0.	125.	350.	465.	580.	700.	815.	935.	985.	1030.	
1070.	1110.	1150.	1190.	1220.	1260.	1290.	1330.	1330.	1330.	1330.	
0.	0.	0.	213.	212.	211.	206.	201.	198.	195.	195.	
195.	190.	180.	170.	160.	153.	147.	142.	142.	142.	142.	

上面是黄河三门峡水库的参数文件。第一行的两个数分别表示三门峡水库的泄水建筑物的种类和每条曲线的节点数 *NPOINT*。下面的 22 个数据是库水位, 接下来依次为水位相应的库容、单一深孔、双层孔、底孔、钢管、隧洞、机组的泄流能力。

## 2. 数据文件 RSV.DAT

```

1979      8      11      20      8      13      8      19
1815. 1870. 1960. 2050. 2140. 2230. 3010. 9310. 11000. 11100.
10500. 9180. 7620. 3673. 6300. 5900. 5286. 4600. 3820.
2.      300.      3
1.      7.      5.      3.      3.      2.      1.
13.     5.      4.      3.      3.      2.      0.
16.     1.      4.      3.      3.      1.      1.

```

RSV.DAT 文件中的第一行分别为计算开始年、月、日、时和结束月、日、时及计算的时段数；第二、三行为入库流量过程；第四行的三个数据分别为计算时段长、初始库水位和计算过程中泄水建筑物启闭变化次数。第五、六、七行的数据分别表示从第 1 个计算时段开始开 7 个深孔、5 对双层孔、3 个底孔、3 个钢管、2 个隧洞、1 台机组；第 13 个计算时段开始改为开 5 个深孔、4 对双层孔、3 个底孔、3 个钢管、2 个隧洞；从第 16 个计算时段开始开 1 个深孔、4 对双层孔、3 个底孔、3 个钢管、1 个隧洞、1 台机组。

## 3. 输出文件 RSV.OUT

下面是按上述水库运用方式下对 1979 年 8 月 11 日 20 时~8 月 13 日 8 时入库流量进行调洪演算结果。

时序	入流(m <sup>3</sup> /s)	出流(m <sup>3</sup> /s)	水位(m)	蓄量(×10 <sup>8</sup> m <sup>3</sup> )
1	1815.00	3073.00	300.00	0.1410
2	1870.00	1355.87	292.36	0.0524
3	1960.00	1807.65	294.59	0.0927
4	2050.00	2088.25	295.84	0.1069



5	2140.00	2102.27	295.90	0.1074
6	2230.00	2274.13	296.62	0.1133
7	3010.00	2992.67	299.66	0.1382
8	9310.00	4167.65	303.44	0.3663
9	11000.00	5425.67	306.74	0.7974
10	11100.00	6552.26	308.80	1.2023
11	10500.00	7476.86	310.09	1.5081
12	9180.00	7689.12	310.65	1.6783
13	7620.00	6666.22	310.79	1.7209
14	3673.00	6591.84	310.55	1.6478
15	6300.00	6474.74	310.17	1.5328
16	5900.00	5124.63	310.08	1.5059
17	5286.00	5145.67	310.19	1.5395
18	4600.00	5136.56	310.15	1.5250
19	3820.00	5057.58	309.90	1.4586

## 第七章 马斯京根法

马斯京根法(简称马法)是美国麦卡锡(G·T·McCarthy)于1938年在美国马斯京根河上使用的流量演算方法。经过几十年的应用和发展,已形成许多不同的应用形式,下面就其几种主要应用形式分别介绍。

### 第一节 基本马法

基本马法即是指1938年麦卡锡在马斯京根河上最初采用的方法。根据水量平衡方程

$$\frac{1}{2}(I_1 + I_2)\Delta t - \frac{1}{2}(O_1 + O_2)\Delta t = W_2 - W_1 \quad (7-1)$$

和槽蓄方程

$$W = K[xI + (1-x)O] \quad (7-2)$$

导出马斯京根流量演算公式

$$O_2 = C_0 I_2 + C_1 I_1 + C_2 O_1 \quad (7-3)$$

其中

$$\begin{aligned} C_0 &= \frac{0.5\Delta t - Kx}{K - Kx + 0.5\Delta t} \\ C_1 &= \frac{Kx + 0.5\Delta t}{K - Kx + 0.5\Delta t} \\ C_2 &= \frac{K - Kx - 0.5\Delta t}{K - Kx + 0.5\Delta t} \end{aligned} \quad (7-4)$$

$$C_0 + C_1 + C_2 = 1 \quad (7-5)$$

式中,  $I_1$ 、 $I_2$  分别表示时段始、末的河段入流量;  $O_1$ 、 $O_2$  分别表示时段始、末的河段出流量;  $K$  为蓄量常数, 具有时间因次;  $x$  为

无因次的流量比重因子； $\Delta t$  为计算时间步长。

从式(7-4)可知，当  $\Delta t < 2Kx$  时， $C_0 < 0$ ， $I_2$  对  $O_2$  是负效应，容易在出流过程线的起涨段出现负流量；当  $\Delta t > 2K - 2Kx$  时， $C_2 < 0$ ， $O_1$  对  $O_2$  是负效应，易在出流过程线的退水段出现负流量。所以要求  $\Delta t \in [2Kx, 2K - 2Kx]$ 。

## 第二节 分段马法

### 一、方法介绍

为了避免出现负出流等不合理现象，保证上、下断面的流量在计算时段内呈线性变化和在任何时刻流量在河段内沿程呈线性变化，一般要求  $\Delta t \approx K$ 。1962 年赵人俊提出了马斯京根分段连续演算法。将演算河段分成  $N$  个子河段后，每个子河段参数  $K_L$ 、 $x_L$  与未分河段时的参数  $K$ 、 $x$  的关系为

$$K_L = \frac{K}{N} \quad (7-6)$$

$$x_L = \frac{1}{2} - \frac{N}{2}(1 - 2x) \quad (7-7)$$

分段连续演算的每段推流公式仍是式(7-3)和(7-4)，但式(7-4)中的  $K$  和  $x$  必须用分段后的  $K_L$  和  $x_L$  代替。

### 二、分段马法计算程序

#### 1. 计算程序

```
SUBROUTINE MSK(K,X,N,DT,NT,QI,QO,QC,IERR)
REAL K
DIMENSION QI(NT),QO(NT),QC(N)
IERR=0
X1=2*K*X
```

```

      X2=2 * K - X1
      IF(DT .GE. X1 .AND. DT .LE. X2) GOTO 10
      IERR = 1
10    CONTINUE
      X1=K - K * X + 0.5 * DT
      C0=(0.5 * DT - K * X)/X1
      C1=(K * X + 0.5 * DT)/X1
      C2=(K - K * X - 0.5 * DT)/X1
      DO 20 I=1,N
      IF(QC(I) .LE. 0)QC(I)=QI(1)
20    CONTINUE
      IF(QO(1) .LE. 0)QO(1)=QC(N)
      DO 100 I=2,NT
      DO 50 J=1,N
      QO1=QC(J)
      IF(J .NE. 1)GOTO 40
      QI1=QI(I-1)
      QI2=QI(I)
40    CONTINUE
      QO2=C0 * QI2 + C1 * QI1 + C2 * QO1
      QI1=QO1
      QI2=QO2
      QC(J)=QO2
50    CONTINUE
      QO(I)=QO2
100   CONTINUE
      RETURN
      END

```

## 2. 程序说明

变量	类型	大小	I/O	内 容
----	----	----	-----	-----

$K$	实型	1	$I$	子河段的马斯京根演算参数 $K(h)$
$X$	实型	1	$I$	子河段的马斯京根演算参数 $x$
$N$	整型	1	$I$	分段数
$DT$	实型	1	$I$	时段步长(h)
$NT$	整型	1	$I$	计算的时段数
$QI$	实型	$NT$	$I$	上断面入流量( $m^3/s$ )
$QO$	实型	$NT$	$O$	下断面出流量( $m^3/s$ )
$QC$	实型	$N$	$I/O$	各段下断面的初始流量( $m^3/s$ ), 若其值小于 0 时, 隐含值为 $QI(1)$
$IERR$	整型	1	$O$	马斯京根演算参数的合理性测试, 若: $2Kx \leq \Delta t \leq 2K - 2Kx$ , $IERR=0$ ; 否则 $IERR=1$

### 三、实例

黄河花园口至夹河滩河段长 105.4km, 采用  $\Delta t=4h$ , 分三段进行演算, 每段的马斯京根演算参数  $K=4.2h$ ,  $x=0.1$ , 通过使用下面的主程序, 把花园口 1982 年 7 月 30 日 20 时~8 月 4 日 16 时流量过程演进到夹河滩。

#### 1. 主程序

```

REAL K
INTEGER YEAR, EM, ED, EH
DIMENSION QI(120), QO(120), QC(20)
OPEN(1, FILE='MSK.DAT', STATUS='OLD')
OPEN(2, FILE='MSK.OUT', STATUS='NEW')
READ(1, *) N, K, X, DT
READ(1, *) YEAR, IM, ID, IH, EM, ED, EH, NT
READ(1, *) (QI(I), I=1, NT)
DO 10 I=1, N

```

```

        QC(I)=QI(1)
10      CONTINUE
        CALL MSK(K,X,N,DT,NT,QI,QO,QC,IERR)
        WRITE(2,110)YEAR,IM,ID,IH,EM,ED,EH,DT
110     FORMAT(5X,I4,'年',I2,'月',I2,'日',I2,'时',
+ '—',I2,'月',I2,'日',I2,'时',' DT=',F4.1,'h',
+ '出流过程')
        WRITE(2,'(8F8.0)')(QO(I),I=1,NT)
        STOP
        END

```

上面的程序从输入文件 MSK.DAT 中读取参数和花园口入流过程，调用 MSK 子程序演算出夹河滩流量过程，结果写入 MSK.OUT。

## 2. 输入文件 MSK.DAT

```

3      4.2    0.1    4.
1982   7      30     20     8      4      16     30
1440.  1650.  4780.  6240.  6320.  6190.  5850.  5580.
5400.  5580.  5520.  7020.  8710.  10400. 11500. 11400.
13000. 14500. 15300. 15200. 15100. 14300. 13100. 12800.
11700. 11100. 10600. 8670.  7930.  7350.

```

## 3. 输出文件 MSK.OUT

1982 年 7 月 30 日 20 时~8 月 4 日 16 时 DT=4.0h 出流过程

```

1440.  1444.  1532.  1965.  2945.  4174.  5143.  5666.
5823.  5770.  5662.  5619.  5789.  6359.  7370.  8616.
9801.  10818. 11836. 12929. 13892. 14519. 14735. 14543.
14041. 13391. 12681. 11942. 11131. 10175.

```

### 第三节 变参数马法

#### 一、方法介绍

由于河道水力特性的非线性,不同数量级的洪水传播时间和洪峰流量的衰减程度也不一样。为解决这一问题,不少人是通过计算大量的历史洪水,求出  $x = f(I_m)$ 、 $K = f(I_m)$  关系,并取  $N = \text{INT}(K/\Delta t)$  (式中,  $I_m$  为入流站洪峰流量,  $\text{INT}$  为取整函数。其它符号意义同前)。计算时是根据上游站的洪峰流量选用本次洪水的参数  $K$ 、 $x$ , 然后用分段马法进行推流。

变参数马法是利用  $K$ 、 $x$  和  $N$  与入流洪峰  $I_m$  的关系来处理河道水力特性非线性问题。实质上,除了当  $x=0$  (水库) 时,  $K$  是入流的函数之外,而在大多数情况下,  $K$  和  $x$  都不仅是入流量的函数,而且也是出流量的函数。换句话说,出流量是前期入流的响应,所以  $K$  和  $x$  不仅仅是洪峰流量的函数,也是过程线的函数。我们经常遇到,在同一河段上,一些同量级的洪水(洪峰流量相当),到达下游断面的传播时间和洪峰衰减程度大不相同,主要原因就在于此。用最小二乘法求一次洪水参数  $K$  和  $x$  的计算公式,即

$$x = - \frac{\sum_{i=1}^n O_i (I_i - O_i)}{\sum_{i=1}^n (I_i - O_i)^2} \quad (7-8)$$

$$K = \frac{\Delta t \sum_{i=1}^n O_i \left[ \sum_{j=1}^{i-1} (I_j - O_j) + \frac{1}{2} (I_i - O_i) \right]}{\sum_{i=1}^n [x (I_i - O_i) + O_i]^2} \quad (7-9)$$

在一般情况下,点出的  $K \sim I_m$  和  $x \sim I_m$  关系比较散乱。尽管

如此,不少单位还是归纳出来了  $K=f(I_m)$  和  $x=f(I_m)$  函数,并且一直应用。这里仍作为一种常用方法作了介绍,并编制了相应的程序。

## 二、变参数马法计算程序

### 1. 计算程序

```
SUBROUTINE BCSMF(QP,XK,XX,M,DT,NT,QI,QO,IERR)
  REAL K
  DIMENSION QP(M),XK(M),XX(M),QI(NT),QO(NT),
+ QC(30)
  QM=QI(1)
  DO 10 I=2,NT
    QM=AMAX1(QM,QI(I))
10  CONTINUE
  IF(QM.LE.QP(1))GOTO 120
  DO 20 I=2,M
    IF(QM.LE.QP(I))GOTO 140
20  CONTINUE
  K=XK(M)
  X=XX(M)
  GOTO 200
120  K=XK(1)
  X=XX(1)
  GOTO 200
140  K=XK(I-1)+(XK(I)-XK(I-1))*(QM-QP(I-1))/
+ (QP(I)-QP(I-1))
  X=XX(I-1)+(XX(I)-XX(I-1))*(QM-QP(I-1))/
+ (QP(I)-QP(I-1))
200  CONTINUE
  N=K/DT+0.5
```



```

K=K/N
X=(1.-N*(1.-2.*X))/2.
IERR=0
X1=2*K*X
X2=2*K-X1
IF(DT.GE.X1.AND.DT.LE.X2)GOTO 210
IERR=1
210  CONTINUE
X1=K-K*X+0.5*DT
C0=(0.5*DT-K*X)/X1
C1=(K*X+0.5*DT)/X1
C2=(K-K*X-0.5*DT)/X1
DO 220 I=1,N
QC(I)=QI(I)
220  CONTINUE
IF(QO(1).LE.0)QO(1)=QC(N)
DO 300 I=2,NT
DO 250 J=1,N
QO1=QC(J)
IF(J.NE.1)GOTO 240
QI1=QI(I-1)
QI2=QI(I)
240  CONTINUE
QO2=C0*QI2+C1*QI1+C2*QO1
QI1=QO1
QI2=QO2
QC(J)=QO2
250  CONTINUE
QO(I)=QO2
300  CONTINUE
RETURN

```

END

## 2. 程序说明

变量	类型	大小	I/O	内 容
QP	实型	M	I	$K=f(Q)$ 及 $x=f(Q)$ 关系曲线的 $Q$ 值序列(以增序存放, 即: $QP(1)\leq QP(2)\leq\cdots\leq QP(M)$ )
XK	实型	M	I	与QP坐标相对应的全河段马斯京根演算参数 $K$ 值(h)
XX	实型	M	I	与QP坐标相对应的全河段马斯京根演算参数 $x$ 值
M	整型	1	I	$K=f(Q)$ 和 $x=f(Q)$ 关系曲线的节点数
DT	实型	1	I	计算时段长(h)
NT	整型	1	I	计算时段数
QI	实型	NT	I	上游站入流量( $\text{m}^3/\text{s}$ )
QO	实型	NT	O	下游站出流量( $\text{m}^3/\text{s}$ )
IERR	整型	1	O	马斯京根演算参数的合理性测试, 若: $2Kx\leq\Delta t\leq 2K-2Kx$ , $IERR=0$ ; 否则, $IERR=1$

## 三、实例

渭河临潼至华县站河段长 84.2km, 经过大量的资料率定, 得出  $x=f(Q_m)$  和  $K=f(Q_m)$ ,  $Q_m$  为临潼站洪峰流量。通过下面的主程序, 用变参数马法把临潼站 1983 年 9 月 28 日 2 时~9 月 28 日 24 时的流量演算至华县站。

## 1. 主程序

```
INTEGER YEAR,EM,ED,EH
DIMENSION QP(50),XK(50),XX(50),QI(120),QO(120)
OPEN(1,FILE='BCSMF.DAT',STATUS='OLD')
OPEN(2,FILE='BCSMF.OUT',STATUS='NEW')
READ(1,*)M
READ(1,*)(QP(I),I=1,M)
READ(1,*)(XK(I),I=1,M)
READ(1,*)(XX(I),I=1,M)
READ(1,*)YEAR,IM,ID,IH,EM,ED,EH,NT,DT
READ(1,*)(QI(I),I=1,NT)
CALL BCSMF(QP,XK,XX,M,DT,NT,QI,QO,IERR)
WRITE(2,110)YEAR,IM,ID,IH,EM,ED,EH,DT
110  FORMAT(5X,I4,'年',I2,'月',I2,'日',I2,'时',
+ '—',I2,'月',I2,'日',I2,'时',' DT=',F4.1,'h',
+ '下游断面流量')
WRITE(2,'(8F8.0)')(QO(I),I=1,NT)
STOP
END
```

主程序从输入文件 BCSMF.DAT 中读取河段参数和上游站来水过程。下面输入文件中的第一行是参数  $M$ ；第二、三行是  $QP$ ；第四、五行是参数  $K$ ；第六、七行是  $x$ ；第八行是计算开始年、月、日、时和结束月、日、时，以及  $NT$ 、 $DT$ 。计算的华县站流量过程写入文件 BCSMF.OUT。

## 2. 输入文件 BCSMF.DAT

20

500.	1000.	1500.	2000.	2500.	3000.	3500.	4000.	4500.	5000.
5500.	6000.	6500.	7000.	7500.	8000.	8500.	9000.	9500.	10000.
15.9	11.7	9.5	8.6	8.2	8.1	8.0	8.0	8.5	9.8

11.4	13.1	14.7	16.4	18.1	19.7	21.4	23.1	24.7	26.4
0.29	0.49	0.44	0.39	0.38	0.38	0.38	0.38	0.38	0.38
0.37	0.33	0.30	0.26	0.23	0.19	0.16	0.12	0.09	0.05
1983	9	28	2	9	28	24	23	1.	
1900.	3290.	4310.	4566.	4660.	4400.	4140.	4000.	3860.	3720.
3580.	3311.	3043.	2902.	2826.	2764.	2702.	2640.	2578.	2516.
2454.	2392.	2330.							

### 3. 输出文件 BCSMF.OUT

1983 年 9 月 28 日 2 时~9 月 28 日 24 时 DT=1.0h 下游断面流量

1900.	1904.	1922.	1970.	2062.	2203.	2390.	2609.
2842.	3070.	3275.	3445.	3572.	3653.	3688.	3680.
3638.	3567.	3477.	3376.	3270.	3163.	3059.	

## 第四节 非线性马法

### 一、非线性马法计算公式

随着马斯京根法的应用,人们发现参数  $K$  和  $x$  不是常数,而是随流量变化的,从而人们开始了对非线性马斯京根法的研究。

水量平衡方程

$$\frac{1}{2}(I_1 + I_2)\Delta t - \frac{1}{2}(O_1 + O_2)\Delta t = W_2 - W_1 \quad (7-10)$$

槽蓄方程

$$W_1 = K_1[x_1 I_1 + (1 - x_1)O_1] \quad (7-11)$$

$$W_2 = K_2[x_2 I_2 + (1 - x_2)O_2] \quad (7-12)$$

把式(7-11)和式(7-12)代入式(7-10),整理得

$$O_2 = C_0 I_2 + C_1 I_1 + C_2 O_1 \quad (7-13)$$

式中

$$\begin{aligned}
C_0 &= \frac{0.5\Delta t - K_2 x_2}{K_2 - K_2 x_2 + 0.5\Delta t} \\
C_1 &= \frac{0.5\Delta t + K_1 x_1}{K_2 - K_2 x_2 + 0.5\Delta t} \\
C_2 &= \frac{K_1 - K_1 x_1 - 0.5\Delta t}{K_2 - K_2 x_2 + 0.5\Delta t}
\end{aligned} \tag{7-14}$$

这里  $C_0, C_1, C_2$  不是常数,  $C_0$  是时段末  $K_2, x_2$  的函数,  $C_1$  和  $C_2$  都是时段初  $K_1, x_1$  和时段末  $K_2, x_2$  的函数。并且  $C_0, C_1, C_2$  之和不恒等于 1。

我们假定  $K$  和  $x$  分别与示储流量  $Q'$  呈线性关系, 其形式为

$$x = AQ' + B \tag{7-15}$$

$$K = CQ' + D \tag{7-16}$$

这里  $A, B, C, D$  都是常数。

$$\text{令} \quad Q' = xI + (1-x)O \tag{7-17}$$

把式(7-15)代入式(7-17),

$$\text{得} \quad Q' = \frac{O + B(I - O)}{1 - A(I - O)} \tag{7-18}$$

已知  $I_1, I_2$  和  $O_1$ , 计算  $O_2$  的步骤是: 先假设一个  $O_2$  (一般以  $O_1$  作初始值), 根据  $I_1, O_1, I_2, O_2$ , 由式(7-18)计算出  $Q'_1$  和  $Q'_2$ , 再由式(7-15)和式(7-16)求出  $x_1, x_2, K_1, K_2$ , 将它们代入式(7-14)求出  $C_0, C_1, C_2$ , 由式(7-13)计算出  $O_2$ , 把计算的  $O_2$  与假定的初值比较, 如果相差较大可用计算的  $O_2$  作初值再重新计算, 直至前后两次计算值相差在允许范围之内, 一般迭代 3~4 次即可。

当  $K, x$  与示储流量  $Q'$  的关系不能用直线表示时, 也可把式(7-15)和式(7-16)配成合适的非线性公式, 再与式(7-17)联解, 求出示储流量计算公式。

## 二、非线性马法计算程序

这里提供的程序是一个非线性马斯京根分段连续演算。它假

定每段的演算参数  $K$  和  $x$  分别与示储流量  $Q'$  呈线性关系, 即符合公式(7-15)和式(7-16)。但是, 为了使用方便, 程序的输入不是参数  $A$ 、 $B$ 、 $C$ 、 $D$ , 而是使用了两点式直线公式, 分别给出每条直线上两个点的坐标值, 由程序自动计算斜率和截距。

### 1. 计算程序

```

SUBROUTINE NLMSK(QXK,ABCD,N,DLT,NT,QC,DEQ,
+ QI,QO)
REAL K1,K2
DIMENSION QXK(6,N),ABCD(4,N),QC(N),QI(NT),
+ QO(NT)
DO 10 I=1,N
IF(QC(I).LT.0)QC(I)=QI(1)
10 CONTINUE
IF(QO(1).LE.0)QO(1)=QC(N)
DO 20 I=1,N
ABCD(1,I)=(QXK(5,I)-QXK(2,I))/(QXK(4,I)-
+ QXK(1,I))
ABCD(2,I)=QXK(5,I)-ABCD(1,I)*QXK(4,I)
ABCD(3,I)=(QXK(6,I)-QXK(3,I))/(QXK(4,I)-
+ QXK(1,I))
ABCD(4,I)=QXK(6,I)-ABCD(3,I)*QXK(4,I)
20 CONTINUE
DO 100 I=2,NT
DO 50 J=1,N
QO1=QC(J)
IF(J.NE.1)GOTO 40
QI1=QI(I-1)
QI2=QI(I)
40 CONTINUE
QO21=QI1

```

```

      III=0
45  CONTINUE
      III=III+1
      QP1=(QO1+ABCD(2,J)*(QI1-QO1))/(1.-
+ ABCD(1,J)*(QI1-QO1))
      QP2=(QO21+ABCD(2,J)*(QI2-QO21))/(1.-
+ ABCD(1,J)*(QI2-QO21))
      X1=ABCD(1,J)*QP1+ABCD(2,J)
      K1=ABCD(3,J)*QP1+ABCD(4,J)
      X2=ABCD(1,J)*QP2+ABCD(2,J)
      K2=ABCD(3,J)*QP2+ABCD(4,J)
      C3=K2-K2*X2+0.5*DLT
      IF(ABS(C3).LT.0.0001)C3=0.0001
      C0=(0.5*DLT-K2*X2)/C3
      C1=(0.5*DLT+K1*X1)/C3
      C2=(K1-K1*X1-0.5*DLT)/C3
      QO2=C0*QI2+C1*QI1+C2*QO1
      IF(ABS(QO2-QO21).LE.DEQ)GOTO 48
      IF(III.GT.50)THEN
        WRITE(*,*)'已迭代 50 次! 请加大 DEQ!'
        STOP
      ELSE
        ENDIF
      QO21=QO2
      GOTO 45
48  CONTINUE
      QI1=QO1
      QI2=QO2
      QC(J)=QO2
50  CONTINUE
      QO(I)=QO2

```

100 CONTINUE

RETURN

END

## 2. 程序说明

变量	类型	大小	I/O	内 容
$QXK$	实型	$6 \times N$	I	各河段演进参数与示储流量的关系。其中： $QXK(1, I)$ ——第 $I$ 段的第一个示储流量值； $QXK(2, I)$ ——第 $I$ 段与示储流量 $QXK(1, I)$ 对应的 $x$ 值； $QXK(3, I)$ ——第 $I$ 段与示储流量 $QXK(1, I)$ 对应的 $K$ 值； $QXK(4, I)$ ——第 $I$ 段的第二个示储流量值； $QXK(5, I)$ ——第 $I$ 段与示储流量 $QXK(4, I)$ 对应的 $x$ 值； $QXK(6, I)$ ——第 $I$ 段与示储流量 $QXK(4, I)$ 对应的 $K$ 值
$ABCD$	实型	$4 \times N$	O	根据 $QXK$ 中的坐标值计算出的演算参数与示储流量关系线的斜率和截距, 其中:



$ABCD(1, I)$ ——第  $I$  段中  $x$  与  $Q'$  关系线的斜率

$A$ ;

$ABCD(2, I)$ ——第  $I$  段中  $x$  与  $Q'$  关系线的截距

$B$ ;

$ABCD(3, I)$ ——第  $I$  段中  $K$  与  $Q'$  关系线的斜率  $C$ ;

$ABCD(4, I)$ ——第  $I$  段中  $K$  与  $Q'$  关系线的截距  $D$

$N$	整型	1	$I$	分段数
$DLT$	实型	1	$I$	时段步长(h)
$NT$	整型	1	$I$	计算的时段数
$QC$	实型	$N$	$I/O$	各段下断面的初始流量( $m^3/s$ ); 若其值小于等于 0 时,隐含值为 $QI(1)$
$DEQ$	实型	1	$I$	迭代计算中允许的流量绝对误差( $m^3/s$ )
$QI$	实型	$NT$	$I$	上游站入流量( $m^3/s$ )
$QO$	实型	$NT$	$O$	下游站计算的出流量( $m^3/s$ )

### 三、实例

仍以黄河花园口至夹河滩的“82.8”洪水为例,说明非线性马法的使用。该河段分为 3 段,  $\Delta t = 4h$ , 设每段的参数相同, 并且  $K$  和  $x$  与示储流量的关系为直线, 当示储流量为  $5000m^3/s$

时, 参数  $x = 0.3$ ,  $K = 5h$ ; 当示储流量为  $20000m^3/s$  时,  $x = 0$ ,  $K = 4.5h$ 。通过下面的主程序, 读取各河段参数, 把花园口站 1982 年 7 月 30 日 20 时~8 月 4 日 16 时流量过程演进到夹河滩站。

### 1. 主程序

```

      INTEGER YEAR,EM,ED,EH
      DIMENSION QXK(6,20),ABCD(4,20),QC(20),QI(120),
+   QO(120)
      OPEN(1,FILE='NLMSK.DAT',STATUS='OLD')
      OPEN(2,FILE='NLMSK.OUT',STATUS='NEW')
      READ(1,*)N,DT,DEQ
      READ(1,*)((QXK(I,J),I=1,6),J=1,N)
      READ(1,*)YEAR,IM,ID,IH,EM,ED,EH,NT
      READ(1,*)(QI(I),I=1,NT)
      DO 10 I=1,N
      QC(I)=QI(1)
10    CONTINUE
      CALL NLMSK(QXK,ABCD,N,DT,NT,QC,DEQ,QI,QO)
      WRITE(2,110)YEAR,IM,ID,IH,EM,ED,EH,DT
110   FORMAT(5X,I4,'年',I2,'月',I2,'日',I2,'时',
+   '—',I2,'月',I2,'日',I2,'时', ' DT=',F4.1,'h',
+   ' 出流过程')
      WRITE(2,'(8F8.0)')(QO(I),I=1,NT)
      STOP
      END
  
```

### 2. 输入文件 NLMSK.DAT

3	4.	100.			
5000.	0.3	5.	20000.	0.	4.5
5000.	0.3	5.	20000.	0.	4.5
5000.	0.3	5.	20000.	0.	4.5

1982	7	30	20	8	4	16	30
1440.	1650.	4780.	6240.	6320.	6190.	5850.	5580.
5400.	5580.	5520.	7020.	8710.	10400.	11500.	11400.
13000.	14500.	15300.	15200.	15100.	14300.	13100.	12800.
11700.	11100.	10600.	8670.	7930.	7350.		

上面程序, 从 NLMSK.DAT 文件中读取演算参数和上游站流量, NLMSK.DAT 的第一行分别为河道分段数、计算时间步长、流量迭代允许误差。第二、三、四行分别为各河段的示储流量和对应的  $x$ 、 $K$  值。从第五行起为计算开始、结束时间、计算时段数和上游站入流量。

### 3. 输出文件 NLMSK.OUT

1982 年 7 月 30 日 20 时~8 月 4 日 16 时 DT=4.0h 出流过程

1440.	1440.	1440.	1459.	1827.	3489.	4948.	5703.
5948.	5893.	5727.	5592.	5601.	5898.	6854.	8234.
9608.	10702.	11723.	12882.	13902.	14547.	14758.	14566.
14078.	13446.	12762.	12049.	11310.	10439.		

## 第五节 分层马法

### 一、方法简介

分层马法是根据河道断面特性, 确定出分层数目, 然后把入流过程进行分层。各层的入流量计算公式为

$$I_{i,t} = \begin{cases} 0 & I_t \leq Q_{i-1} \\ I_t - Q_{i-1} & Q_{i-1} < I_t \leq Q_i \\ Q_i - Q_{i-1} & I_t > Q_i \end{cases} \quad (7-19)$$

且  $Q_0=0$

式中,  $I_t$  为  $t$  时刻上游断面总入流;  $I_{i,t}$  为第  $i$  层  $t$  时刻的入流量

( $i=1, 2, \dots, L$ ),  $L$  为分层数,  $Q_i$  为第  $i$  层的上限流量。

把求得的各层入流量  $I_{i,t}$  分别用其对应层的马法演算系数  $C_{0i}$ 、 $C_{1i}$ 、 $C_{2i}$  进行推流计算, 求出各层入流量在下断面的出流  $O_{i,t}$ 。即

$$O_{i,t} = C_{0i}I_{i,t} + C_{1i}I_{i,t-1} + C_{2i}O_{i,t-1} \quad (7-20)$$

式中

$$\begin{aligned} C_{0i} &= \frac{0.5\Delta t - K_i x_i}{K_i - K_i x_i + 0.5\Delta t} \\ C_{1i} &= \frac{0.5\Delta t + K_i x_i}{K_i - K_i x_i + 0.5\Delta t} \\ C_{2i} &= \frac{K_i - K_i x_i - 0.5\Delta t}{K_i - K_i x_i + 0.5\Delta t} \end{aligned} \quad (7-21)$$

式中,  $K_i$ 、 $x_i$  分别表示第  $i$  层水体的蓄量常数和流量比重因子。把计算的各层出流量叠加起来, 即为下断面的总出流量。

$$O_t = \sum_{i=1}^L O_{i,t} \quad (7-22)$$

## 二、方法特点

在复式河道中, 水位涨到某一高度时发生漫溢, 水面突然变宽, 其水力特性及洪水传播规律发生变化, 使得洪水演进参数  $K$ 、 $x$  难以概化, 整个过程线高、低水段很难拟合, 此时可以用分层马法处理。

演算是否需要分层, 分层多少和各层参数的确定, 一般是根据河道特性和实测流量过程线分析确定。首先点绘出研究河段沿河一些站实测大断面图及几场大洪水的上、下游站流量过程线。若大断面呈明显的复式型, 或下游站流量过程线有明显的不连续性现象 (一般情况, 流量过程线的不连续点与大断面突变处的上滩流量相一致), 就需要根据大断面形状 (或下游站流量过程线) 的突变次数进行分层, 突变处所对应的流量作为分界流量。若找

不出上述明显现象，可不分层。

分层后，因各层的参数只对该层流量在其过程线中的相应段起主导作用，所以，需用以往的大、中、小洪水资料进一步率定。率定时，应先用小洪水资料对下层参数进行率定，然后依次使用中、大洪水资料从下向上逐层调整参数。在率定过程中，对分界流量需作进一步调整。如果开始没有分层，在对大洪水的计算过程中，发现高水段与实测过程模拟较差，可以在开始发生偏离处进行分层，达到使全过程都有较好的拟合效果。

前面谈了分层马法的适应情况，分层原则及参数确定的一般方法，下面就该法具有的特点简单归纳如下：

1. 分层马法实质上是对河道洪水演算的一种非线性求解。由于它直接考虑河道断面形态，抓住了问题的关键，各层的参数可以分别调整，自由度大，因此能够解决一般马法难以解决的问题。该法概念明确，编程简单，容易率定，使用方便。

2. 当分层马法的分层数目  $L=1$  时，或者虽分数层，但各层的参数相同时，其结果就是一般马法。因此可以认为，分层马法是一般马法的扩展。

3. 各层间  $K$ 、 $x$  的变化规律，与河道特性、分层的多少及各层厚度有关，选择时，应以率定结果为准。一般说来，随着河槽水位上升，水面加宽，河道调蓄作用增大，洪水波传播的时间加长， $K$  值变大， $x$  值变小。

4. 分层马法中  $K$ 、 $x$  与流量间的关系，不同于一般马法中的  $K=f(Q)$  和  $x=f(Q)$ 。一般马法中的流量是河道总流量，反映的是河道的总水体。而分层马法中的流量是层流量，反映的是部分水体。所以，分层马法演算参数与流量的关系和一般马法中的物理概念不完全一致。

### 三、分层马法计算程序

分层马法计算程序是一个分层分段连续演算的通用程序。当给定了演算的分层数和分段数、各段分界流量以及各段中各层马法演算参数  $K$ 、 $x$ ，便可调用该程序把上游站入流按分层马法的原理演算到下游，求出下游断面的总出流。

#### 1. 计算程序

```
SUBROUTINE LAYMK(N,L,QXK,C012,DT,NT,
+ QC,QI,QO)
REAL K
DIMENSION QXK(3,L),C012(3,L),QI(NT),QO(NT),QC(N)
DO 10 IL=1,L
K=QXK(3,IL)/N
X=(1.-N*(1.-2.*QXK(2,IL)))/2.
C3=K-K*X+0.5*DT
C012(1,IL)=(0.5*DT-K*X)/C3
C012(2,IL)=(K*X+0.5*DT)/C3
C012(3,IL)=(K-K*X-0.5*DT)/C3
10 CONTINUE
DO 20 I=1,N
IF(QC(I).LT.0)QC(I)=QI(1)
20 CONTINUE
IF(QO(1).LE.0)QO(1)=QC(N)
DO 300 INT=2,NT
DO 200 IN=1,N
QO1=QC(IN)
IF(IN.NE.1)GOTO 40
QI1=QI(INT-1)
QI2=QI(INT)
40 CONTINUE
```

```

      QO2=0
      DO 100  IL=1,L
50    CONTINUE
      IF(IL.EQ.1)THEN
      QK1=0
      ELSE
      QK1=QXK(1,IL-1)
      ENDIF
      QK2=QXK(1,IL)
      IF(QI1.LE.QK1)QLI1=0
      IF((QI1.GT.QK1).AND.(QI1.LE.QK2))QLI1=QI1-QK1
      IF(QI1.GT.QK2)QLI1=QK2-QK1
      IF(QI2.LE.QK1)QLI2=0
      IF((QI2.GT.QK1).AND.(QI2.LE.QK2))QLI2=QI2-QK1
      IF(QI2.GT.QK2)QLI2=QK2-QK1
      IF(QO1.LE.QK1)QLO1=0
      IF((QO1.GT.QK1).AND.(QO1.LE.QK2))QLO1=QO1-QK1
      IF(QO1.GT.QK2)QLO1=QK2-QK1
      QLO2=C012(1,IL)*QLI2+C012(2,IL)*QLI1+
+ C012(3,IL)*QLO1
      QO2=QO2+QLO2
100  CONTINUE
      QI1=QO1
      QI2=QO2
      QC(IN)=QO2
200  CONTINUE
      QO(INT)=QO2
300  CONTINUE
      RETURN
      END

```



## 2. 程序说明

变量	类型	大小	I/O	内 容
$N$	整型	1	$I$	河道演算分段数
$L$	整型	1	$I$	各段分层数
$QXK$	实型	$3 \times L$	$I$	各层的分界流量和演算参数。 其中： $QXK(1, I)$ ——第 $I$ 层全河段的上限流量； $QXK(2, I)$ ——第 $I$ 层全河段的马法参数 $x$ ； $QXK(3, I)$ ——第 $I$ 层全河段的马法参数 $K$
$C012$	实型	$3 \times L$	$O$	各层的演算系数。其中： $C012(1, I)$ ——第 $I$ 层的分段马法演算参数 $C_0$ ； $C012(2, I)$ ——第 $I$ 层的分段马法演算参数 $C_1$ ； $C012(3, I)$ ——第 $I$ 层的分段马法演算参数 $C_2$
$DT$	实型	1	$I$	计算时段长(h)
$NT$	整型	1	$I$	计算时段数
$QC$	实型	$N$	$I/O$	各段下断面的初始流量, 若其值 $\leq 0$ 时, 隐含值为 $QI(1)$
$QI$	实型	$NT$	$I$	上游站总入流量( $m^3/s$ )
$QO$	实型	$NT$	$O$	下游站总出流量( $m^3/s$ )



#### 四、实例

以黄河下游夹河滩至高村(河长 83.2km)河段为例。该段属地上悬河,是典型的复式河道,当流量达到某一标准后,洪水漫滩,河道特性变化。用分层马斯京根法对夹河滩站 1982 年 7 月 31 日 12 时~8 月 7 日 24 时的流量演进至高村。经分析,该河道断面可分为两层,0~8000m<sup>3</sup>/s 为第一层,  $K=12h$ ,  $x=0.45$ ; 8000~30000m<sup>3</sup>/s 为第二层,  $K=34h$ ,  $x=0.4$ 。计算取  $\Delta t=4h$ , 并分四个子河段进行演进。使用下面的主程序从 LAYMK.DAT 中读取参数和夹河滩流量过程,把计算结果写入 LAYMK.OUT。

##### 1. 主程序

```
REAL K
INTEGER YEAR,EM,ED,EH
DIMENSION QXK(3,5),C012(3,5),QI(120),QO(120),QC(20)
OPEN(1,FILE='LAYMK.DAT',STATUS='OLD')
OPEN(2,FILE='LAYMK.OUT',STATUS='NEW')
READ(1,*)N,L,DT
READ(1,*)((QXK(I,J),I=1,3),J=1,L)
READ(1,*)YEAR,IM,ID,IH,EM,ED,EH,NT
READ(1,*)(QI(I),I=1,NT)
DO 10 I=1,N
  QC(I)=QI(1)
10  CONTINUE
  CALL LAYMK(N,L,QXK,C012,DT,NT,QC,QI,QO)
  WRITE(2,110)YEAR,IM,ID,IH,EM,ED,EH,DT
110  FORMAT(5X,I4,'年',I2,'月',I2,'日',I2,'时',
+ '—',I2,'月',I2,'日',I2,'时',' DT=',F4.1,'h',
+ '出流过程')
```

```
WRITE(2,'(8F8.0)')(QO(I),I=1,NT)
```

```
STOP
```

```
END
```

## 2. 输入文件 LAYMK.DAT

4	2	4.					
8000.	0.45	12.	30000.	0.4	34.		
1982	7	31	12	8	7	24	46
1330.	2850.	5200.	6200.	6400.	5920.	5440.	5080.
4900.	5300.	5970.	7160.	8470.	9550.	11620.	13800.
14500.	13500.	13300.	13390.	14000.	13700.	13100.	12600.
11800.	10480.	9200.	7090.	7190.	6930.	6850.	6750.
6500.	6360.	6010.	6090.	5810.	5510.	5130.	5020.
4550.	4400.	4730.	4490.	4600.	4600.		

## 3. 输出文件 LAYMK.OUT

1982年7月31日12时~8月7日24时 DT = 4.0h 出流过程

1330.	1338.	1434.	1905.	3092.	4703.	5838.	6138.
5907.	5495.	5185.	5127.	5445.	6144.	7047.	7787.
8094.	8326.	8727.	9284.	9931.	10588.	11197.	11733.
12180.	12523.	12745.	12829.	12754.	12491.	12011.	11291.
10278.	8870.	7260.	6536.	6317.	6129.	5976.	5789.
5497.	5199.	4926.	4679.	4540.	4569.		

## 第六节 讨 论

本章前几节分别介绍了马斯京根流量演算方法的几种常用形式，并给出了相应的计算程序。下面就各方法间的关系，区间来(引)水的处理，马法的预见期与预报误差等分述如下：

## 一、各方法间的关系

分段马法、变参数马法、非线性马法、分层马法是当前国内使用较为广泛的几种河道洪水演进方法，其根本都是对基本马法线性假定作必要的修正。分段马法主要是解决基本马法的空间分布线性假定引起的误差问题。其余三种主要解决基本马法时间变化线性假定引起的误差，其实质上是解决断面流速分布线性假定问题。在后三种方法中，变参数马法是把每次洪水按线性处理，而各次洪水考虑取不同参数；非线性马法是每个时段按线性处理，而不同时段取不同参数，可见较变参数马法考虑非线性因素要精细一些；分层马法则直接从断面上考虑，随着过水断面面积的增加，不同层上取不同参数，特别适宜于复式河道。如果能把分段马法与其它三种方法中的某种方法相结合，就能综合地考虑基本马法时、空分布非线性问题，获得较好的结果。

## 二、区间来(引)水的处理

马法是基于水量平衡原理之上的，但对于一般河流，上下游站水量并不平衡，所以用马法作河道洪水演进时，区间来(引)水问题不可忽视。需要首先对区间来(引)水进行处理，即根据来(引)水所在河段，把来水流量过程加在相应河段的入流过程中(引水视为负来水)。使上下断面水量平衡后再作流量演算。

## 三、马法的预见期和预报误差

众所周知，马法只有当  $\Delta t = 2Kx$  时才有一个时段的预见期，而在一般情况下，马法本身并没有预见期。要想用于实时预报，必须用降雨径流模型预报出上游站未来流量过程，或者根据上游站的流量趋势预估出未来的入流量方可进行马法的作业预报。另外，即使有些方法可以使河道洪水演算获得预见期，但方法本身

的预见期一般也不超过一个时段,照样难以满足实际工作之需要。所以,找出误差的影响因素,作好预报误差的理论分析是十分必要的。

我们知道,在无区间入流的情况下,河道下游断面的流量是上游断面入流量和马斯京根连续演算汇流系数的卷积。用  $O(t)$  表示下断面的出流量,用  $I(t)$  表示上断面的入流量,  $r(\tau)$  表示马斯京根连续演算汇流系数,其卷积关系为

$$O(t) = \int_0^{t_m} I(t - \tau) r(\tau) d\tau \quad (7-23)$$

把式(7-23)写成离散形式,则

$$O(t) = \sum_{i=0}^m I(t-i) r(i) \quad (7-24)$$

式中,  $m$  是汇流系数的历时。

如果根据  $t-1$  时刻以前的实测资料计算  $O(t)$ , 也就是使预见期为 1 个时段, 这时只有  $I(t)$  未知, 假设  $I(t)$  的估计误差为  $\Delta I(t)$ , 那么,  $O(t)$  的误差就是  $\Delta I(t) r(0)$ ; 若根据  $t-2$  时刻以前的实测资料预报  $O(t)$ , 预见期为 2 个时段,  $O(t)$  的误差为  $\Delta O(t) = \Delta I(t) r(0) + \Delta I(t-1) r(1)$ 。以此类推, 由此可见, 预见期越长, 误差越大。误差的大小是预见期内入流的估计误差和相应的马斯京根连续演算汇流系数乘积之和。对于某一预见期条件下马法的预报误差的理论计算公式为

$$\Delta O(t/N) = \sum_{i=0}^{N-1} \Delta I(t-i) r(i) \quad (7-25)$$

式中,  $N$  为预见期 ( $N \leq m$ );  $\Delta O(t/N)$  为预见期为  $N$  个时段情况下  $t$  时刻流量的预报误差;  $\Delta I(t-i)$  为  $t-i$  时刻的上游入流的估计误差。

需要说明的是, 式(7-25)是理论误差, 是基于预见期为 0 时计算误差等于 0。实际上并不是这样, 在用马法作流量演算时, 即使上下游都是实测流量, 没有预见期, 计算结果也有一定的误

差，这是测验误差、区间来水、引水、参数  $K$ 、 $x$  取值不合理等因素所致。

## 第八章 纳须单位线法

### 第一节 基本概念

单位线属于黑色系统。尽管如此,自从 1932 年舍尔曼(L·K·Sherman)提出单位线的概念以后,它一直是水文学者经常用来进行净雨流量转换的工具。本节主要介绍纳须(J·E·Nash)单位线。一般对单位线的定义是:在一个特定的流域上,单位时段内均匀分布的单位净雨所形成的流域出口站的地面径流过程线。

#### 一、纳须瞬时单位线

1945 年克拉克(C·O·Clark)首先提出瞬时单位线的概念。所谓瞬时单位线,就是流域上均匀分布的、历时趋于无限小、强度趋于无穷大、但净雨量为 1 个单位的净雨所形成的单位线。通常用  $u(t)$  或  $u(0,t)$  表示。1957 年,纳须把流域看作是一连串的  $n$  个相同的“线性水库”,如图 8—1 所示。可以推导出一个单位的瞬时入流进入水库系统演进,其瞬时单位线的数学方程式为

$$u(0,t) = \frac{1}{k\Gamma(n)} \left(\frac{t}{k}\right)^{n-1} e^{-\frac{t}{k}} \quad (8-1)$$

式中, $\Gamma$  为伽马函数; $n$  相当于水库个数; $k$  为一个线性水库的蓄泄系数。

#### 二、纳须时段单位线

在实际应用中,需要将瞬时单位线转换成时段单位线,一般用  $s(t)$  曲线。




图 8-1 纳须模型示意图

按照  $s(t)$  曲线的定义,  $s(t)$  曲线等于瞬时单位线的积分, 即

$$s(t) = \int_0^t u(0, t) dt = \int_0^{\frac{t}{k}} \frac{1}{\Gamma(n)} \left(\frac{t}{k}\right)^{n-1} e^{-\frac{t}{k}} d\left(\frac{t}{k}\right) \quad (8-2)$$

那么, 时段长为  $\Delta t$  的单位线为

$$u(\Delta t, t) = [s(t) - s(t - \Delta t)] \quad (8-3)$$

由此可见,  $s(t)$  曲线是线性水库个数  $n$  和相对时间  $t/k$  的函数。当  $n$  为正整数时, 对式 (8-2) 进行积分, 可推导出  $s(t)$  曲线和纳须时段单位线的计算公式, 使用该式可以方便地进行纳须时段单位线的手工计算和实现计算机编程。公式推导如下:

对于式 (8-2), 令  $v = t/k$ , 当  $t = 0$  时,  $v = 0$ ; 当  $t = t$  时,  $v = t/k$ ,  $s(t)$  曲线的计算公式可写为

$$\begin{aligned} s(t) &= \int_0^v \frac{1}{\Gamma(n)} v^{n-1} e^{-v} dv \\ &= \frac{1}{\Gamma(n)} \left[ -v^{n-1} e^{-v} + (n-1) \int_0^v v^{n-2} e^{-v} dv \right] \end{aligned}$$



$$\begin{aligned}
&= \frac{1}{\Gamma(n)} \left[ -v^{n-1}e^{-v} - (n-1)v^{n-2}e^{-v} \right. \\
&\quad \left. + (n-1)(n-2) \int_0^v v^{n-3}e^{-v}dv \right] \\
&= \frac{1}{\Gamma(n)} \left[ -v^{n-1}e^{-v} - (n-1)v^{n-2}e^{-v} - (n-1)(n-2) \right. \\
&\quad \left. \cdot v^{n-3}e^{-v} + (n-1)(n-2)(n-3) \int_0^v v^{n-4}e^{-v}dv \right]
\end{aligned}$$

当  $n$  为自然数时, 在最后必然出现

$$\int_0^v v^0 e^{-v} dv = -e^{-v} \Big|_0^v = 1 - e^{-v}$$

所以

$$\begin{aligned}
s(t) &= \frac{1}{\Gamma(n)} \left\{ \Gamma(n) - \sum_{i=1}^n \left[ e^{-\frac{t}{k}} \left( \frac{t}{k} \right)^{n-i} \prod_{j=1}^{i-1} (n-j) \right] \right\} \\
&= 1 - e^{-\frac{t}{k}} \sum_{i=1}^n \frac{1}{(n-i)!} \left( \frac{t}{k} \right)^{n-i} \quad (8-4)
\end{aligned}$$

这就是  $n$  为自然数时  $s(t)$  曲线的计算公式。

把式(8-4)代入式(8-3), 得

$$\begin{aligned}
u(\Delta t, t) &= [s(t) - s(t - \Delta t)] \\
&= e^{-\frac{t}{k}} \left[ e^{\frac{\Delta t}{k}} \sum_{i=1}^n \frac{1}{(n-i)!} \left( \frac{t - \Delta t}{k} \right)^{n-i} - \sum_{i=1}^n \frac{1}{(n-i)!} \left( \frac{t}{k} \right)^{n-i} \right] \quad (8-5)
\end{aligned}$$

式(8-5)是  $n$  为自然数时, 计算步长为  $\Delta t$  时的纳须时段单位线计算公式。

### 三、纳须单位线 $n, k$ 的计算

在已知纳须单位线的参数  $n$  和  $k$  时, 根据式(8-5)可以方便地计算出某给定计算时段长( $\Delta t$ )下的纳须时段单位线。那么下面的问题就是如何根据流域的实测水文资料求出  $n$  和  $k$ 。根据参考文献[6]中给出的瞬时单位线、入流量(净雨)过程、出流量过程三者的矩之间的关系, 即



$$n = \frac{[M_Q^{(1)} - M_I^{(1)}]^2}{N_Q^{(2)} - N_I^{(2)}} \quad (8-6)$$

$$k = \frac{N_Q^{(2)} - N_I^{(2)}}{M_Q^{(1)} - M_I^{(1)}} \quad (8-7)$$

式中,  $M_I^{(1)}$ 、 $M_Q^{(1)}$  分别为入流量(净雨量)及出流量的一阶原点矩;  
 $N_I^{(2)}$ 、 $N_Q^{(2)}$  分别为入流量和出流量的二阶中心矩, 它们的计算公式为

$$\begin{aligned} M_I^{(1)} &= \frac{\int_0^\infty I(\tau) \tau d\tau}{\int_0^\infty I(\tau) d\tau} = \frac{\sum_{i=1}^n \bar{I}_i(\tau) \tau_i}{\sum_{i=1}^n \bar{I}_i(\tau)} \\ &= \frac{\sum_{i=1}^n \bar{I}_i(\tau) m_i}{\sum_{i=1}^n \bar{I}_i(\tau)} \cdot \frac{\Delta t}{2} \end{aligned} \quad (8-8)$$

$$\begin{aligned} M_Q^{(1)} &= \frac{\int_0^\infty Q(t) t dt}{\int_0^\infty Q(t) dt} = \frac{\sum_{i=1}^n \bar{Q}_i(t) t_i}{\sum_{i=1}^n \bar{Q}_i(t)} \\ &= \frac{\sum_{i=1}^n \bar{Q}_i(t) m_i}{\sum_{i=1}^n \bar{Q}_i(t)} \cdot \frac{\Delta t}{2} \end{aligned} \quad (8-9)$$

式中,  $I(\tau)$ 、 $\bar{I}_i(\tau)$  分别表示  $\tau$  时刻的入流量和入流量的时段平均值;  
 $Q(t)$ 、 $\bar{Q}_i(t)$  分别表示  $t$  时刻的出流量和出流量的时段平均值。  
 $m_i = 1, 3, 5, 7, \dots, 2i-1, \dots, 2n-3, 2n-1$ 。

同理, 计算入流量、出流量的二阶原点矩的公式为

$$M_I^{(2)} = \frac{\sum_{i=1}^n \bar{I}_i(\tau) m_i^2}{\sum_{i=1}^n \bar{I}_i(\tau)} \cdot \left(\frac{\Delta t}{2}\right)^2 \quad (8-10)$$

$$M_Q^{(2)} = \frac{\sum_{i=1}^n \bar{Q}_i(t) m_i^2}{\sum_{i=1}^n \bar{Q}_i(t)} \cdot \left(\frac{\Delta t}{2}\right)^2 \quad (8-11)$$

入流量和出流量的二阶中心矩可用原点矩计算,即

$$N_I^{(2)} = M_I^{(2)} - [M_I^{(1)}]^2 \quad (8-12)$$

$$N_Q^{(2)} = M_Q^{(2)} - [M_Q^{(1)}]^2 \quad (8-13)$$

在推求某一流域纳须时段单位线时,首先选用几场有代表性的洪水资料,根据每场洪水的相应入流和出流过程,分别求出它们的单位线参数  $n$  和  $k$ ,然后把  $n$  和  $k$  平均(如果相差较大,可以分成几组)概化出流域的单位线参数  $n$  和  $k$ ,便可用纳须时段单位线计算公式求出流域的单位线。

## 第二节 计算程序

### 一、求 $n$ 和 $k$ 的计算程序

该程序是根据入流和出流时间序列,用矩法计算出本场洪水的纳须单位线参数  $n$  和  $k$ 。

#### 1. 计算程序

```
SUBROUTINE GETNK(R,NR,Q,NQ,DT,N,K)
```

```
REAL N,K,M11,M12,M01,M02,N12,N02
```

```
DIMENSION R(NR),Q(NQ)
```

```
X1=0
```

```
X2=0
```

```
X3=0
```

```

DO 10 I=1,NR
X1=X1+R(I)*(2*I-1)
X2=X2+R(I)
10  X3=X3+R(I)*(2*I-1)* * 2
MI1=X1/X2*DT/2.
MI2=X3/X2*DT*DT/4.
X1=0
X2=0
X3=0
DO 20 I=1,NQ-1
XM=(Q(I)+Q(I+1))/2.
X1=X1+XM*(2*I-1)
X2=X2+XM
20  X3=X3+XM*(2*I-1)* * 2
MO1=X1/X2*DT/2.
MO2=X3/X2*DT*DT/4.
NI2=MI2-MI1* * 2
NO2=MO2-MO1* * 2
N=(MO1-MI1)* * 2/(NO2-NI2)
K=(NO2-NI2)/(MO1-MI1)
RETURN
END

```

## 2. 程序说明

变量	类型	大小	I/O	内 容
R	实型	NR	I	净雨过程(mm)
NR	整型	1	I	净雨的时段数
Q	实型	NQ	I	R 的相应出流量过程(m <sup>3</sup> /s)
NQ	整型	1	I	Q 的时段数
DT	实型	1	I	计算时段长(h)

N	实型	1	0	纳须单位线的参数 $n$
K	实型	1	0	纳须单位线的参数 $k(h)$

## 二、纳须时段单位线的计算程序

### 1. 计算程序

```

SUBROUTINE CUNIT(N,K,DT,UH,MAX,M)
REAL K
DIMENSION UH(MAX)
UH(1)=0
IT=2
T=DT
10  CONTINUE
X1=0
X2=0
DO 100 I=1,N
NI=N-I
JC=1
IF(NI.EQ.0)GOTO 30
DO 20 J=1,NI
20  JC=JC * J
30  CONTINUE
X1=((T-DT)/K) * * NI/JC + X1
X2=(T/K) * * NI/JC + X2
100 CONTINUE
X1=X1 * EXP(DT/K)
UH(IT)=EXP(-T/K) * (X1-X2)
IF((UH(IT).LT.UH(IT-1)).AND.
+ (UH(IT).LT.0.0005))GOTO 120
T=T+DT
IT=IT+1

```

```

        GOTO 10
120    M=IT-1
        UNIT=0
        DO 130 I=1,M
130    UNIT=UNIT+UH(I)
        WC=UNIT-1.
        IF (ABS(WC).LT.0.0005) GOTO 200
        UH(M)=UH(M)-WC
        IF (UH(M).LT.0.0005) M=M-1
200    CONTINUE
        RETURN
        END

```

## 2. 程序说明

变量	类型	大小	I/O	内 容
<i>N</i>	整型	1	<i>I</i>	纳须单位线参数 <i>n</i>
<i>K</i>	实型	1	<i>I</i>	纳须单位线参数 <i>k</i> (h)
<i>DT</i>	实型	1	<i>I</i>	计算单位线时段长 (h)
<i>M</i>	整型	1	<i>O</i>	单位线底宽 (时段数)
<i>UH</i>	实型	<i>M</i>	<i>O</i>	单位线纵坐标
<i>MAX</i>	整型	1	<i>I</i>	为 <i>UH</i> 数组开设的最大空间

## 第三节 实例

纳须时段单位线的计算程序比较简单,只要给出参数 *n* 和 *k*、计算时段长 (*DT*) 和数组 (*UH*) 的最大开辟空间 (*MAX*),调用 CUNIT 子程序即可方便地计算出单位线 (*UH*) 的每个纵坐标,这里不再举例。

下面就如何根据净雨和地面径流量推求纳须单位线的参数 *n*

和  $k$  进行举例说明。已知某站一次降水的净雨和地面径流过程如表 8—1 所示。

表 8-1

时间 (年.月.日.时)	净雨 (mm)	实测出流 (m <sup>3</sup> /s)	基 流 (m <sup>3</sup> /s)	地面径流 (m <sup>3</sup> /s)
1959. 7. 9. 2		108	108	0
8	30.0	516	109	407
14	10.8	1000	112	888
20		1100	115	985
10. 2		827	120	707
8		436	122	314
14		280	125	155
20		190	126	64
11. 2		165	127	38
8		128	128	0

通过下面的主程序,从输入文件 GETNK.DAT 读取净雨时段数( $NR$ )、地面径流量时段数( $NQ$ )、计算时间步长( $DT$ )及净雨量过程( $R$ )和地面径流量过程( $Q$ ),调用 GETNK 子程序就可计算出纳须单位线参数  $n$  和  $k$ ,并把结果在屏幕上打印出来。

## 1. 主程序

```

REAL N,K
DIMENSION R(100),Q(100)
OPEN(1,FILE='GETNK.DAT',STATUS='OLD')
READ(1,*)NR,NQ,DT
READ(1,*)(R(I),I=1,NR)
READ(1,*)(Q(I),I=1,NQ)
CALL GETNK(R,NR,Q,NQ,DT,N,K)
WRITE(*,10)N,K
10  FORMAT(5X,2F10.2)

```

STOP

END

## 2. 输入文件 GETNK.DAT

2 ,10 6.

30. 10.8

0. 407. 888. 985. 707. 314. 155. 64. 38. 0.

## 第九章 反馈模拟实时校正

### 第一节 模型简介

由于流域特性和降水分布的复杂性和多变性,用降水径流或河道汇流作出的流量序列预报,有时误差很大。反馈模拟实时校正正是最大限度地利用了预见期内所获得的各种信息对后期预报值进行校正以提高预报精度的一种有效方法。本文列出了有关反馈模拟实时校正的计算公式,详细内容请参见参考文献[7]。

已知实测流量  $Q_{ob(i)}$  ( $i=1, 2, \dots, N$ ) 和预报流量  $Q_{f(i)}$  ( $i=1, 2, \dots, LAP$ )。首先对实测流量与预报流量进行相关分析,计算出相关系数  $R_c$  和确定性系数  $D_y$ 。

实测流量的平均流量为

$$\bar{Q}_{ob} = \frac{1}{N} \sum_{i=1}^N Q_{ob(i)} \quad (9-1)$$

与实测流量对应的预报流量的平均流量为

$$\bar{Q}_f = \frac{1}{N} \sum_{i=1}^N Q_{f(i)} \quad (9-2)$$

相关系数为

$$R_c = \frac{\sum_{i=1}^N (Q_{f(i)} - \bar{Q}_f)(Q_{ob(i)} - \bar{Q}_{ob})}{\left[ \sum_{i=1}^N (Q_{f(i)} - \bar{Q}_f)^2 \cdot \sum_{i=1}^N (Q_{ob(i)} - \bar{Q}_{ob})^2 \right]^{1/2}} \quad (9-3)$$

确定性系数为

$$D_y = R_c^2 \quad (9-4)$$

求相邻时刻实测流量间的差值

$$DQ_{ob(i)} = Q_{ob(i)} - Q_{ob(i-1)} \quad (9-5)$$



式中,  $i=2, \dots, N$ 。对于  $i=1$ , 设  $DQ_{ob(1)}=0$

求预报流量间的差值

$$DQ_{f(i)} = Q_{f(i)} - Q_{f(i-1)} \quad (9-6)$$

式中,  $i=2, \dots, LAP$ 。对于  $i=1$ , 设  $DQ_{f(1)}=0$

计算相邻两个时段实测流量差值之和与预报流量差值之和的比值  $FACT$  因子, 其表达式为

$$FACT(i-1) = \frac{DQ_{ob(i)} + DQ_{ob(i-1)}}{DQ_{f(i)} + DQ_{f(i-1)}} \quad (9-7)$$

$$\text{或 } FACT(i-1) = \frac{DQ_{ob(i)} - DQ_{ob(i-1)}}{DQ_{f(i)} - DQ_{f(i-1)}} \quad (9-7')$$

求  $F_{(i,j)}$ , 其表达式为

$$F_{(i,j)} = FACT(i)^{0.75 \cdot j} \quad (9-8)$$

( $j=1, 2, \dots, 6$ )

用  $DQ_{f(i)} \geq 0$  或  $DQ_{f(i)} < 0$  把流量过程分为涨水段和退水段, 进行反馈模拟实时校正。

### 1. 涨水段反馈模拟实时校正

(1) 如果  $i-(N+6) \geq 0$ , 且  $i > 7$ , 则实时校正流量为

$$Q_{ob(i)} = Q_{ob(i-1)} + DQ_{f(i)} \quad (9-9)$$

(2) 如果  $i-(N+6) < 0$ , 则实时校正系数

$$FAC = \frac{F_{(i-6,6)} + F_{(i-5,5)} + \dots + F_{(N,i-N)}}{7 + N - i}$$

$$Q_{ob(i)} = Q_{ob(i-1)} + DQ_{f(i)} \cdot FAC \quad (9-10)$$

### 2. 退水段反馈模拟实时校正

退水段反馈模拟实时校正计算公式为

$$Q_{ob(i)} = Q_{f(i)} \cdot \frac{Q_{ob(i-1)}}{Q_{f(i-1)}} \quad (9-11)$$

### 3. 当 $N=1$ 时的涨水段反馈模拟实时校正

$$Q_{ob(i)} = Q_{ob(i-1)} + [Q_{f(i)} - Q_{f(i-1)}] \quad (9-12)$$

$i=2,3,\dots,K$  ( $K$  是洪峰对应的序数)。

## 第二节 实时校正计算程序

反馈模拟实时校正是充分利用已获得的实测流量信息,根据这些已出现的实测流量与原预报流量值的关系,对未来的预报流量值进行反馈模拟校正。程序的输入是未来  $LAP$  个时段的预报流量和预见期内前  $N$  个时段的实测流量;输出为实时校正后的流量,其中前  $N$  个为实测值,后  $LAP-N$  个是根据实测流量与预报流量的关系对未来出现的预报流量作了校正的流量值。

### 一、计算程序

```
SUBROUTINE RTADJ(QOB,N,QF,LAP,QADJ,RC,DY,F)
  DIMENSION QOB(N),QF(LAP),QADJ(LAP),F(8,LAP)
  RC=0.
  QOB0=0.
  QF0=0.
  DF=0.
  DOB=0.
  DO 10 I=1,N
    QF0=QF0+QF(I)
10  QOB0=QOB0+QOB(I)
    QF0=QF0/N
    QOB0=QOB0/N
  DO 20 I=1,N
    A0=QF(I)-QF0
    A1=QOB(I)-QOB0
    DF=DF+A0*A0
    DOB=DOB+A1*A1
```

```

20   RC=RC+A0*A1
      RC=RC/SQRT(DF*DOB+0.0001)
      DY=RC*RC
      IF(N-LAP) 120,400,400
120  F(1,1)=0.
      F(2,1)=0.
      DO 130 I=2,LAP
130  F(2,I)=QF(I)-QF(I-1)
      DO 140 I=2,N
140  F(1,I)=QOB(I)-QOB(I-1)
      DO 200 KK=2,N
      FACT=1.
      X=F(2,KK)+F(2,KK-1)+0.0001
      IF(X.EQ.0) GOTO 180
      IF(F(1,KK).LE.0.AND.F(1,KK-1).LE.0) GOTO 180
      FACT=(F(1,KK)+F(1,KK-1)+0.0001)/X
      IF(FACT-2.21) 150,180,160
150  IF(FACT-0.45) 170,180,180
160  FACT=2.21
      GOTO 180
170  FACT=0.45
180  DO 190 I=1,6
      FACT=FACT**(.75**I)
190  F(I+2,KK-1)=FACT
200  CONTINUE
C   _____
      DO 205 I=1,N
205  QADJ(I)=QOB(I)
      DO 300 I=N+1,LAP
      IF(F(2,I)) 220,210,210
210  FAC=1.

```

```

      JKN=I-N
      IF (JKN-6)230,230,260
220   FAC=(QADJ(I-1)+0.0001)/(QF(I-1)+0.0001)
      GOTO 260
230   FAC=0.
      LKN=N-1
      CNT=0.
      DO 240 LI=JKN,6
      FAC=FAC+F(LI+2,LKN)
      CNT=CNT+1
      LKN=LKN-1
      IF (LKN)250,250,240
240   CONTINUE
250   FAC=FAC/CNT
260   QADJ(I)=QADJ(I-1)+F(2,I)*FAC
300   CONTINUE
C     -----
400   CONTINUE
      RETURN
      END

```

## 二、程序说明

变量	类型	大小	I/O	内 容
QOB	实型	N	I	实测流量(m <sup>3</sup> /s)
N	整型	1	I	实测流量的个数
QF	实型	LAP	I	预报流量(m <sup>3</sup> /s)
LAP	整型	1	I	预报流量的个数
QADJ	实型	LAP	O	实时校正流量(m <sup>3</sup> /s)
RC	实型	1	O	实测流量与预报流量间的相关系数

$DY$  实型 1 0 实测流量与预报流量间的确定性系数  
 $F$  实型  $8 \times LAP$  中间变量

### 第三节 实例

下面是黄河潼关水文站 1992 年 8 月 14 日 12 时~15 日 12 时 ( $\Delta t = 2h$ ) 的一次洪水的反馈模拟实时校正的例子。通过下面的主程序从输入文件 RTADJ.DAT 中读取预报流量和已出现的实测流量,调用 RTADJ 子程序对未来预报流量值进行反馈模拟实时校正,结果写入 RTADJ.OUT。

#### 1. 主程序

```
DIMENSION QOB(50),QF(100),QADJ(100),F(8,100)
OPEN(1,FILE='RTADJ.DAT',STATUS='OLD')
OPEN(2,FILE='RTADJ.OUT',STATUS='NEW')
READ(1,*)N,LAP
READ(1,*)(QOB(I),I=1,N)
READ(1,*)(QF(I),I=1,LAP)
CALL RTADJ(QOB,N,QF,LAP,QADJ,RC,DY,F)
WRITE(2,'(2F10.2)')RC,DY
WRITE(2,'(8F8.0)')(QADJ(I),I=1,LAP)
STOP
END
```

#### 2. 输入文件 RTADJ.DAT

4 13

3210. 3370. 3520. 3800.

3900. 4100. 4200. 4220. 4230. 4250. 4600. 4590.

4520. 4130. 4000. 3800. 3620.

RTADJ.DAT 文件的第一行分别为实测流量个数  $N$ 、预报流

量个数  $LAP$ 。下面前  $N$  个数据是实测流量值,后  $LAP$  个数据为预报流量。

### 3. 输出文件 RTADJ.OUT

0.88     0.77  
 3210.   3370.   3520.   3800.   3813.   3835.   4203.   4194.  
 4130.   3773.   3655.   3472.   3307.

输出文件的前两个数据分别为实测流量与预报流量的相关系数和确定性系数。后面  $LAP$  个数据为实时校正流量。

通过反馈模拟实时校正,洪峰预报误差由 14.4%降为 4.5%,精度有明显提高,如图 9—1。采用数据系当年报汛值和实时预报值。




图 9—1 潼关站 1992 年 8 月洪水过程线

## 第十章 其它常用水文计算子程序

前面已对一些常用的流域降雨径流模型、水库调洪计算、马斯京根法、纳须单位线法和反馈模拟实时校正等水文计算方法及其相应的计算程序作了详细介绍,本章再介绍其它一些常用水文计算子程序。有些程序尽管非常简单,但由于它被前面一些模型程序调用过,这里不妨也一并列出。

### 一、直线插值

#### 1. 功能

用于直线插值。

#### 2. 计算程序

```
SUBROUTINE ZXCZ(N,X,Y,K,U,F)
  DIMENSION X(N),Y(N)
  IF(U.GE.X(1))GOTO 20
  IF(K.NE.1)THEN
    F=Y(1)
    GOTO 200
  ELSE
    I=2
    GOTO 100
  ENDIF
20  IF(U.LE.X(N))GOTO 40
  IF(K.NE.1)THEN
    F=Y(N)
    GOTO 200
  ELSE
```

```

        I=N
        GOTO 100
    ENDIF
40    CONTINUE
        DO 50 I=2,N
            IF(U.LE.X(I))GOTO 100
50    CONTINUE
100    F=Y(I-1)+(Y(I)-Y(I-1))/(X(I)-X(I-1))*(U-X(I-1))
200    CONTINUE
        RETURN
    END

```

### 3. 程序说明

变量	类型	大小	I/O	内 容
N	整型	1	I	已知曲线的节点数, $N \geq 2$
X	实型	N	I	曲线的横坐标。递增序列, 即: $X(1) < X(2) < \dots < X(N)$
Y	实型	N	I	与 X 对应的纵坐标
K	整型	1	I	插值要求类别, 若 $K=1$ , 曲线的两 端点之外, 采用直线外延; 否则, 端点之外取端点值
U	实型	1	I	已知的横坐标
F	实型	1	O	求出的与 U 对应的纵坐标

## 二、抛物线插值

### 1. 功能和计算公式

#### (1) 功能

根据曲线的  $N$  个节点, 进行抛物线插值。

#### (2) 计算公式

设已知节点的横坐标为  $X_1, X_2, \dots, X_{n-1}, X_n$  ( $X_1 < X_2 < \dots <$



$X_{i-1} < X_i$ ), 其对应的纵坐标分别为  $y_1, y_2, \dots, y_{i-1}, y_i$ 。求横坐标  $U$  值所对应的纵坐标  $F$  值

$$F = \frac{(U - X_i)(U - X_{i+1})}{(X_{i-1} - X_i)(X_{i-1} - X_{i+1})} y_{i-1} \\ + \frac{(U - X_{i+1})(U - X_{i-1})}{(X_i - X_{i+1})(X_i - X_{i-1})} y_i \\ + \frac{(U - X_{i-1})(U - X_i)}{(X_{i+1} - X_{i-1})(X_{i+1} - X_i)} y_{i+1}$$

插值时选取最靠近  $U$  的三个节点  $X_{i-1}, X_i, X_{i+1}$  进行, 选点方法是

$$i = \begin{cases} 2 & \text{当 } U \leq X_2 \\ k-1 & \text{当 } X_{k-1} < U \leq X_k \text{ 且 } |U - X_{k-1}| \leq |U - X_k| \\ & (k = 3, 4, \dots, n-1) \\ k & \text{当 } X_{k-1} < U \leq X_k \text{ 且 } |U - X_{k-1}| > |U - X_k| \\ & (k = 3, 4, \dots, n-1) \\ n-1 & \text{当 } U > X_{n-1} \end{cases}$$

## 2. 计算程序

```
SUBROUTINE PWXCZ(N,X,Y,K,U,F)
DIMENSION X(N),Y(N)
IF(U.LT.X(1).AND.K.NE.1)THEN
F=Y(1)
GOTO 100
ELSE
ENDIF
IF(U.GT.X(N).AND.K.NE.1)THEN
F=Y(N)
GOTO 100
ELSE
ENDIF
```

```

      L=N-1
      DO 10 I=2,L
      IF(U.LE.X(I))GOTO 20
10    CONTINUE
      I=L
      GOTO 30
20    IF(I.EQ.2)GOTO 30
      IF((U-X(I-1)).LE.(X(I)-U))I=I-1
30    CONTINUE
      X1=X(I-1)
      X2=X(I)
      X3=X(I+1)
      A1=(U-X2)*(U-X3)/((X1-X2)*(X1-X3))
      A2=(U-X3)*(U-X1)/((X2-X3)*(X2-X1))
      A3=(U-X1)*(U-X2)/((X3-X1)*(X3-X2))
      F=A1*Y(I-1)+A2*Y(I)+A3*Y(I+1)
100  CONTINUE
      RETURN
      END

```

### 3. 程序说明

变量	类型	大小	I/O	内 容
N	整型	1	I	已知曲线的节点数, $N \geq 3$
X	实型	N	I	曲线的横坐标, 递增序列
Y	实型	N	I	与 X 对应的纵坐标
K	整型	1	I	插值要求类别, 若 $K=1$ , 曲线的两端之外, 用最外面的三个节点外延插值; 否则, 端点之外取端点值
U	实型	1	I	已知的横坐标
F	实型	1	O	求出的与 U 对应的纵坐标

### 三、水位与流量的换算

#### 1. 功能

根据已知的水位流量关系曲线(单一或绳套型),进行水位与流量的转换。

#### 2. 计算程序

```
SUBROUTINE HQC(K1,K2,N,H,QU,QD,M,X,Y)
  DIMENSION H(N),QU(N),QD(N),X(M),Y(M)
  IF(K2.NE.1)GOTO 200
C  -----
C  已知水位求流量
100  CONTINUE
     IF(K1.EQ.1)GOTO 150
C  -----
     DO 110 I=1,M
       CALL ZXCZ(N,H,QU,1,X(I),Y(I))
110  CONTINUE
     GOTO 300
C  -----
150  CONTINUE
     M1=M-1
     DO 160 I=1,M1
       IF(X(I+1).GE.X(I))THEN
         CALL ZXCZ(N,H,QU,1,X(I),Y(I))
         II=1
       ELSE
         CALL ZXCZ(N,H,QD,1,X(I),Y(I))
         II=2
       ENDIF
160  CONTINUE
```

```

IF(II.EQ.1)THEN
CALL ZXCZ(N,H,QU,1,X(M),Y(M))
ELSE
CALL ZXCZ(N,H,QD,1,X(M),Y(M))
ENDIF
GOTO 300

```

C -----

C 已知流量求水位

```

200 CONTINUE
IF(K1.EQ.1)GOTO 250

```

C -----

```

DO 210 I=1,M
CALL ZXCZ(N,QU,H,1,X(I),Y(I))
210 CONTINUE
GOTO 300

```

C -----

```

250 CONTINUE
M1=M-1
DO 260 I=1,M1
IF(X(I+1).GE.X(I))THEN
CALL ZXCZ(N,QU,H,1,X(I),Y(I))
II=1
ELSE
CALL ZXCZ(N,QD,H,1,X(I),Y(I))
II=2
ENDIF
260 CONTINUE
IF(II.EQ.1)THEN
CALL ZXCZ(N,QU,H,1,X(M),Y(M))
ELSE
CALL ZXCZ(N,QD,H,1,X(M),Y(M))

```

```

        ENDIF
300  RETURN
      END

```

### 3. 程序说明

变量	类型	大小	I/O	内 容
K1	整型	1	I	水位~流量关系线类型 K1=1, 绳套型; K1≠1, 单一型
K2	整型	1	I	计算类别 K2=1, 已知水位求流量; K2≠1, 已知流量求水位
N	整型	1	I	单支曲线节点数
H	实型	N	I	水位
QU	实型	N	I	与水位相应的涨水段流量值
QD	实型	N	I	与水位相应的退水段流量值
M	整型	1	I	已知的水位(或流量)序列的个数
X	实型	M	I	已知水位(当 K2≠1 时为流量)值
Y	实型	M	O	求得的流量(当 K2≠1 时为水位)值

注:①该程序内部调用了直线插值子程序 ZXCZ。

②当水位~流量关系线单一时, QD 无效。

## 四、单位线推流计算

### 1. 功能

根据流域净雨(或河段上游站的入流量)和流域径流单位线(或马斯京根汇流系数),采用卷积的差分形式算出流域出口(或河段下游站)的流量过程。

## 2. 计算程序

```

SUBROUTINE UHQC(MI,QI,MU,UH,U,MO,QO)
  DIMENSION QI(MI),UH(MU),QO(MO)
  DO 10 I=1,MO
    QO(I)=0
10  CONTINUE
    DO 20 I=1,MI
      DO 20 J=1,MU
        IJ=I+J-1
        QO(IJ)=QO(IJ)+QI(I)*UH(J)*U
20  CONTINUE
    RETURN
  END

```

## 3. 程序说明

变量	类型	大小	I/O	内 容
MI	整型	1	I	入流过程 QI 的个数
QI	实型	MI	I	入流过程(单位为 mm 或 m <sup>3</sup> /s)
MU	整型	1	I	单位线历时(时段)
UH	实型	MU	I	单位线纵坐标
U	实型	1	I	单位线计算的转换系数。当入流 QI 为净雨(单位是 mm), 且 UH 是无因次单位线时, $U = \frac{\text{流域面积}}{3.6\Delta t}$ ; 当入流量单位与出流量单位相同, 或单位线本身是有因次单位线时, $U = 1$ 。流域面积(km <sup>2</sup> ), $\Delta t$ (h)
MO	整型	1	I	出流量 QO 的个数, $MO = MI + MU - 1$

QO 实型 MO O 出流量( $\text{m}^3/\text{s}$ )

## 五、时段单位线的转换

### 1. 功能

·把已知的某一时段单位线转换成另一时段单位线。

### 2. 计算程序

```
SUBROUTINE DWXZH(M1,UH1,DT1,TIME,S1,MAX2,UH2,  
+ DT2,M2)  
  DIMENSION UH1(M1),TIME(M1),S1(M1),UH2(MAX2)  
  DM2=(M1-1)*DT1/DT2  
  M2=(M1-1)*DT1/DT2  
  IF(DM2.GT.M2)M2=M2+1  
  M2=M2+1  
  TIME(1)=0  
  S1(1)=UH1(1)  
  DO 10 I=2,M1  
    TIME(I)=(I-1)*DT1  
    S1(I)=S1(I-1)+UH1(I)  
10  CONTINUE  
  W1=S1(M1)  
  UH2(1)=UH1(1)  
  DO 20 I=2,M2  
    T2=(I-1)*DT2  
    CALL PWXCZ(M1,TIME,S1,0,T2,UH2(I))  
20  CONTINUE  
  DO 30 I=1,M2  
    UH2(I)=UH2(I)*DT1/DT2  
30  CONTINUE  
  SS1=UH2(1)  
  DO 40 I=2,M2
```

```

        SS2=UH2(I)
        UH2(I)=SS2--SS1
        SS1=SS2
40    CONTINUE
        DO 50 I=1,M2
50    IF(UH2(I).LT.0)UH2(I)=0
        W2=0
        DO 60 I=1,M2
60    W2=W2+UH2(I)
        U=DT1/DT2
        DO 70 I=1,M2
        UH2(I)=UH2(I)*U*W1/W2
70    CONTINUE
        RETURN
        END

```

### 3. 程序说明

变量	类型	大小	I/O	内 容
M1	整型	1	I	原单位线历时(时段)
UH1	实型	M1	I	原单位线纵坐标
DT1	实型	1	I	原单位线时段长(h)
TIME	实型	M1		中间变量
S1	实型	M1		中间变量
MAX2	整型	1	I	为UH2开设的最大空间, $MAX2 \geq M1 \cdot DT1/DT2 + 1$
UH2	实型	M2	O	转换后的时段单位线纵坐标
DT2	实型	1	I	要求转换成的单位线时段长(h)
M2	整型	1	O	UH2的历时(时段)

注:该程序内部调用了抛物线插值程序PWXCZ。



## 六、马斯京根演算参数 $K$ 与 $x$ 的计算

### 1. 功能

根据河段上、下游断面的实测流量过程,用最小二乘法计算出该河道的马斯京根演算参数  $K$  和  $x$  值。

### 2. 计算程序

```
SUBROUTINE GETKX(N,Q1,QO,DT,K,X)
REAL K
DIMENSION QI(N),QO(N)
A1=0
A2=0
DO 10 I=1,N
DQ=QI(I)-QO(I)
A1=A1+QO(I)*DQ
A2=A2+DQ*DQ
10 CONTINUE
X=-A1/A2
A1=0
A2=0
DO 30 I=1,N
A3=0.
IF(I.EQ.1)GOTO 25
I1=I-1
DO 20 J=1,I1
DQJ=QI(J)-QO(J)
A3=A3+DQJ
20 CONTINUE
25 DQI=QI(I)-QO(I)
A1=A1+QO(I)*(A3+0.5*DQI)
A2=A2+(X*DQI+QO(I))* * 2
```

```

30  CONTINUE
    K=DT * A1/A2
    RETURN
    END

```

### 3. 程序说明

变量	类型	大小	I/O	内 容
N	整型	1	I	入流(或出流)量序列的个数
QI	实型	N	I	入流量时间序列
QO	实型	N	I	出流量时间序列
DT	实型	1	I	时间序列的步长(h)
K	实型	1	O	河段马法参数 K
X	实型	1	O	河段马法参数 x

## 七、马斯京根汇流系数的计算

### 1. 功能

根据给定的河道分段数  $N$  和一段的马法演算参数  $x$  值, 求出  $\Delta t = K$  时的河道马斯京根汇流系数, 或称河道汇流单位线。

### 2. 计算程序

```

SUBROUTINE MSKC(N,X,MX,C,M)
  DIMENSION C(MX)
  DO 10 I=1,MX
    C(I)=0.
10  CONTINUE
    C(1)=1.
    CC=1.5-X
    C0=(0.5-X)/CC
    C1=(0.5+X)/CC
    C2=(0.5-X)/CC
    DO 50 J=1,N

```

```

      I=1
20  CONTINUE
      IF(I.NE.1)GOTO 30
      Q11=0
      Q01=0
30  CONTINUE
      Q12=C(I)
      Q02=C0 * Q12+C1 * Q11+C2 * Q01
      IF(Q02.LT.0.0005.AND.Q02.LT.Q01)GOTO 40
      Q11=Q12
      Q01=Q02
      C(I)=Q02
      I=I+1
      GOTO 20
40  CONTINUE
      M=I-1
50  CONTINUE
      W=0
      DO 60 I=1,M
      W=W+C(I)
60  CONTINUE
      DW=1.-W
      IF(DW.GT.0.0005.OR.DW.LT.-0.0005)THEN
      C(M)=C(M)+DW
      ELSE
      ENDIF
      RETURN
      END

```

### 3. 程序说明

变量	类型	大小	I/O	内 容
----	----	----	-----	-----

$N$	整型	1	$I$	河道分段数
$X$	实型	1	$I$	每段的马斯京根参数 $x$ 值
$MX$	整型	1	$I$	为汇流系数 $C$ 数组开的空间
$C$	实型	$M$	$O$	河道汇流系数
$M$	整型	$I$	$O$	汇流系数的历时

## 八、误差统计与分析

### (一) 功能和计算公式

#### 1. 功能

用于统计一个时间序列(计算过程)与另一个时间序列(实测过程)的误差。

#### 2. 计算公式

##### (1) 时间序列均值

$$\bar{S} = \frac{1}{N} \sum_{i=1}^N S_i$$

##### (2) 洪量相对误差

$$WP = \frac{(\bar{S} - \bar{Q})}{\bar{Q}} \cdot 100\%$$

##### (3) 相对绝对值误差

$$AP = \frac{1}{N} \sum_{i=1}^N \frac{|S_i - Q_i|}{\bar{Q}} \cdot 100\%$$

##### (4) 均方差

$$SRD = \sqrt{\frac{1}{N} \sum_{i=1}^N (S_i - Q_i)^2}$$

##### (5) 最大绝对值差

$$MAD = \text{MAX} |S_i - Q_i|$$

式中, MAX 是求最大值函数。

##### (6) 峰值相对误差

$$MP = \frac{S_{\max} - Q_{\max}}{Q_{\max}} \cdot 100\%$$

### (7) 峰现时间差

$$TD = T_s - T_q$$

式中,  $S_i$  为计算时间序列的第  $i$  个值;  $Q_i$  为实测时间序列的第  $i$  个值;  $N$  为时间序列个数;  $\bar{Q}$  为实测时间序列的均值;  $S_{\max}$  为计算时间序列的最大值;  $Q_{\max}$  为实测时间序列的最大值;  $T_s$  为  $S_{\max}$  的出现时间;  $T_q$  为  $Q_{\max}$  的出现时间;  $TD$  为峰现时间差(h), 正值为计算滞后, 负值为计算超前。

### (二) 计算程序

#### 1. 计算程序

```

SUBROUTINE WCTJ(N,Q,S,DT,ER)
  DIMENSION Q(N),S(N),ER(7)
  SP=0
  QP=0
  AP=0
  SRD=0
  DO 10 I=1,N
    SP=SP+S(I)
    QP=QP+Q(I)
    A=S(I)-Q(I)
    AP=AP+ABS(A)
    SRD=SRD+A*A
10  CONTINUE
  SP=SP/N
  QP=QP/N
  AP=AP/N
  SRD=SQRT(SRD/N)
  ER(1)=SP
  ER(2)=(SP-QP)/QP*100.

```

```

ER(3)=AP/QP * 100.
ER(4)=SRD
ER(5)=ABS(S(1)-Q(1))
DO 20 I=2,N
A=ABS(S(I)-Q(I))
IF(ER(5).LT.A) ER(5)=A
20  CONTINUE
SMAX=S(1)
QMAX=Q(1)
IS=1
IQ=1
DO 50 I=2,N
IF(S(I).LE.SMAX)GOTO 30
SMAX=S(I)
IS=I
30  CONTINUE
IF(Q(I).LE.QMAX)GOTO 40
QMAX=Q(I)
IQ=I
40  CONTINUE
50  CONTINUE
ER(6)=(SMAX-QMAX)/QMAX * 100.
ER(7)=(IS-IQ) * DT
RETURN
END

```

## 2. 程序说明

变量	类型	大小	I/O	内 容
N	整型	1	I	时间序列的个数
Q	实型	N	I	实测时间序列过程

<i>S</i>	实型	<i>N</i>	<i>I</i>	计算时间序列过程
<i>DT</i>	实型	1	<i>I</i>	计算时段长(h)
<i>ER</i>	实型	7	<i>O</i>	统计的各种特征值,它们依次是:计算时间序列均值 $\bar{S}$ 、洪量相对误差 $WP$ 、相对绝对值误差 $AP$ 、均方差 $SRD$ 、最大绝对值差 $MAD$ 、峰值相对误差 $MP$ 、峰现时间差 $TD$

## 九、等时段流量(或水位)序列的生成

### 1. 功能

把洪水要素摘录的流量(或水位)过程,插补成等时间间隔的时间序列。有实测资料时期内的数据采用直线内插,实测资料时期外的数据用端点值平延。

### 2. 计算程序

```

SUBROUTINE EXTQ(YEAR,IM,ID,IH,EM,ED,EH,DT,
+ N,MDH,OBQ,M,TIME,TQ,OBT)
INTEGER YEAR,EM,ED,EH
INTEGER * 4 TIME
DIMENSION MDH(4,N),OBQ(N),OBT(N),TIME(M),TQ(M)
DIMENSION MON(12),MONDAY(12)
DATA MON/31,28,31,30,31,30,31,31,30,31,30,31/
DATA MONDAY/0,31,59,90,120,151,181,212,243,
+ 273,304,334/
IF(YEAR/4*4.NE.YEAR)GOTO 20
MON(2)=29
DO 10 I=3,12
10 MONDAY(I)=MONDAY(I)+1
20 CONTINUE

```

```

IDT = DT
ND = MONDAY (EM) - MONDAY (IM) + ED - ID
M = ND * 24 / DT + EH / DT - IH / DT + 1
IMM = IM
IDD = ID
HH = IH
TIME (1) = IMM * 100000 + IDD * 1000 + HH * 10
DO 30 I = 2, M
HH = HH + DT
IF (HH. LE. 24) GOTO 25
HH = HH - 24
IDD = IDD + 1
IF (IDD. LE. MON (IMM)) GOTO 25
IDD = IDD - MON (IMM)
IMM = IMM + 1
25 TIME (I) = IMM * 100000 + IDD * 1000 + HH * 10
30 CONTINUE
DO 50 I = 1, N
IMM = MDH (1, I)
IDD = MDH (2, I)
IHH = MDH (3, I)
IFF = MDH (4, I)
OBT (I) = (MONDAY (IMM) + IDD - 1) * 24. + IHH + IFF / 60.
50 CONTINUE
DO 60 I = 1, M
IT = TIME (I)
IMM = IT / 100000
IT = MOD (IT, 100000)
IDD = IT / 1000
IT = MOD (IT, 1000)
HH = IT / 10.

```



```

T = (MONDAY (IMM) + IDD - 1) * 24. + HH
CALL ZXCZ(N, OBT, OBQ, 0, T, Q)
TQ(I) = Q
60  CONTINUE
RETURN
END

```

### 3. 程序说明

变量	类型	大小	I/O	内 容
YEAR	整型	1	I	计算的年份
IM	整型	1	I	要生成的等时间间隔时间序列开始月份
ID	整型	1	I	要生成的等时间间隔时间序列开始日
IH	整型	1	I	要生成的等时间间隔时间序列开始时
EM	整型	1	I	要生成的等时间间隔时间序列结束月份
ED	整型	1	I	要生成的等时间间隔时间序列结束日
EH	整型	1	I	要生成的等时间间隔时间序列结束时
DT	实型	1	I	要生成的等时间间隔时间序列的时段长(h)
N	整型	1	I	实测时间序列的个数
MDH	整型	4 × N	I	分别表示每个实测流量对应的月、日、时、分
OBQ	实型	N	I	实测流量序列

<i>M</i>	整型	1	0	生成的等时间间隔时间序列的个数
<i>TIME</i>	整型	<i>M</i>	0	生成的等时间间隔时间序列对应的时间 格式为: <i>MMDDHHF</i> 其中 <i>MM</i> ——月; <i>DD</i> ——日; <i>HH</i> ——时; <i>F</i> ——分(十进制)
<i>TQ</i>	实型	<i>M</i>	0	生成的等时间间隔的流量(或水位)过程
<i>OBT</i>	实型	<i>N</i>		中间变量数组

注:内部调用了直线插值程序 ZXCZ。

## 十、等时段雨量序列的生成

### 1. 功能

把降雨要素摘录表中所示的降水过程,插补成等时间间隔的降雨时间序列。

### 2. 计算程序

```

SUBROUTINE EXTP(YEAR,IM,ID,IH,EM,ED,EH,DT,
+ N,MDH,OBP,M,TIME,TP,K,OBT,OBPH)
INTEGER YEAR,EM,ED,EH
INTEGER * 4 TIME
DIMENSION MDH(6,N),OBP(N),TIME(M),TP(M)
DIMENSION MON(12),MONDAY(12),OBT(K),OBPH(K)
DATA MON/31,28,31,30,31,30,31,31,30,31,30,31/
DATA MONDAY/0,31,59,90,120,151,181,212,243,
+ 273,304,334/

```

```

      IF (YEAR/4 * 4.NE. YEAR)GOTO 20
      MON(2)=29
      DO 10 I=3,12
10    MONDAY(I)=MONDAY(I)+1
20    CONTINUE
      IDT=DT
      ND=MONDAY(EM)-MONDAY(IM)+ED-ID
      M=ND * 24/DT +EH/DT -IH/DT
      IMM=IM
      IDD=ID
      HH=IH
      DO 30 I=1,M
      HH=HH+DT
      IF (HH.LE. 24)GOTO 25
      HH=HH-24
      IDD=IDD+1
      IF (IDD.LE. MON(IMM))GOTO 25
      IDD=IDD-MON(IMM)
      IMM=IMM+1
25    TIME(I)=IMM * 100000+IDD * 1000+HH * 10
30    CONTINUE
C
      IM1=MDH(1,1)
      ID1=MDH(2,1)
      IH1=MDH(3,1)
      IF1=MDH(4,1)
      IM2=IM1
      ID2=ID1
      IH2=MDH(5,1)
      IF2=MDH(6,1)
      IF (IH2.GE. IH1)GOTO 35

```

```

ID2=ID2+1
IF (ID2. LE. MON (IM2))GOTO 35
ID2=ID2-MON (IM2)
IM2=IM2+1
35  T1=(MONDAY (IM1)+ID1-1) * 24. +IH1+IF1/60.
    T2=(MONDAY (IM2)+ID2-1) * 24. +IH2+IF2/60.
    OBT (1)=T1
    OBPH (1)=0.
    OBT (2)=T2
    OBPH (2)=OBP (1)
    II=2
    DO 50  I=2,N
    IM1=MDH (1,I)
    ID1=MDH (2,I)
    IH1=MDH (3,I)
    IF1=MDH (4,I)
    T3=(MONDAY (IM1)+ID1-1) * 24. +IH1+IF1/60.
    IF (T3. LE. T2) GOTO 38
    II=II+1
    OBT (II)=T3
    OBPH (II)=OBPH (II-1)
38  CONTINUE
    IM2=IM1
    ID2=ID1
    IH2=MDH (5,I)
    IF2=MDH (6,I)
    IF (IH2. GE. IH1)GOTO 40
    ID2=ID2+1
    IF (ID2. LE. MON (IM2))GOTO 40
    ID2=ID2-MON (IM2)
    IM2=IM2+1

```

```

40·  CONTINUE
      T4=(MONDAY(IM2)+ID2-1)*24.+IH2+IF2/60.
      II=II+1
      OBT(II)=T4
      OBPH(II)=OBPH(II-1)+OBP(I)
      T1=T3
      T2=T4
50  CONTINUE
      K=II
      T=(MONDAY(IM)+ID-1)*24.+IH
      CALL ZXCZ(K,OBT,OBPH,0,T,P1)
      DO 60 I=1,M
      IT=TIME(I)
      IMM=IT/100000
      IT=MOD(IT,100000)
      IDD=IT/1000
      IT=MOD(IT,1000)
      HH=IT/10.
      T=(MONDAY(IMM)+IDD-1)*24.+HH
      CALL ZXCZ(K,OBT,OBPH,0,T,P2)
      TP(I)=P2-P1
      P1=P2
60  CONTINUE
      RETURN
      END.

```

### 3. 程序说明

变量	类型	大小	I/O	内 容
YEAR	整型	1	I	计算的年份
IM	整型	1	I	插补计算的开始月

<i>ID</i>	整型	1	<i>I</i>	插补计算的开始日
<i>IH</i>	整型	1	<i>I</i>	插补计算的开始时
<i>EM</i>	整型	1	<i>I</i>	插补计算的结束月
<i>ED</i>	整型	1	<i>I</i>	插补计算的结束日
<i>EH</i>	整型	1	<i>I</i>	插补计算的结束时
<i>DT</i>	实型	1	<i>I</i>	要生成的等时间间隔雨量序列的时段长(h)
<i>N</i>	整型	1	<i>I</i>	实测雨量序列的个数
<i>MDH</i>	整型	$6 \times N$	<i>I</i>	分别表示每个实测雨量对应的起始月、日、时、分和终止时、分
<i>OBP</i>	实型	<i>N</i>	<i>I</i>	实测雨量值
<i>M</i>	整型	1	<i>O</i>	生成的等时间间隔雨量序列的个数
<i>TIME</i>	整型	<i>M</i>	<i>O</i>	生成的等时间间隔雨量序列对应的时间。 格式为: <i>MMDDHHF</i> 其中 <i>MM</i> ——月; <i>DD</i> ——日; <i>HH</i> ——时; <i>F</i> ——分(十进制)
<i>TP</i>	实型	<i>M</i>	<i>O</i>	生成的等时间间隔的雨量过程
<i>K</i>	整型	1	<i>O</i>	中间变量数组 <i>OBT</i> 和 <i>OBPH</i> 的大小
<i>OBT</i>	实型	<i>K</i>		中间变量
<i>OBPH</i>	实型	<i>K</i>		中间变量

注: ① *OBT* 和 *OBPH* 分别开设的空间最好为  $2N$ 。

②程序内部调用了直线插值子程序 *ZXCZ*。

## 十一、打印过程线

### 1. 功能

本程序可以最多打印四条等时间间隔的流量和降水序列过程线(如图 10—1 所示),其中最多只能有一个降雨序列。流量过程线有 1 : 10, 1 : 100, 1 : 500, 1 : 1000 四个比例,程序将根据所给时间序列过程自动选择。

### 2. 计算程序

```
SUBROUTINE PLT(M,N,DATA,K)
  DIMENSION DATA(4,N)
  CHARACTER * 1  QPL(80),A1(4),A2
  DATA  A1/'1','2','3','4'/
  OPEN(10,FILE='PLT.OUT',STATUS='NEW')
  A2='I'
  DO 10  I=1,M
    DO 10  J=1,N
      IF(DATA(I,J).LT.0)DATA(I,J)=0.
10    CONTINUE
      QMAX=DATA(1,1)
      DO 20  I=1,2
        DO 20  J=1,N
          QMAX=AMAX1(QMAX,DATA(I,J))
20    CONTINUE
      RATE=10.
      MQ=QMAX/RATE
      IF(MQ.LE.78)GOTO 30
      RATE=100.
      MQ=QMAX/RATE
      IF(MQ.LE.78)GOTO 30
      RATE=500.
```




图 10-1 降水、流量过程线



```

      MQ=QMAX/RATE
      IF(MQ.LE.78)GOTO 30
      RATE=1000.
30    CONTINUE
      WRITE(10,50)RATE
50    FORMAT(/36X,'流量比例 1:',F6.0,5X,'(m3/s)',
+ '降水比例 1:1 (mm)')
      DO 70 I=1,80
      IF(I/10*10.EQ.I)THEN
      QPL(I)='*'
      ELSE
      QPL(I)='—'
      ENDIF
70    CONTINUE
      GO TO(110,120,130,140),M
110   WRITE(10,'(7X,7X,A1,80A1)')A2,QPL
      GOTO 160
120   WRITE(10,'(7X,14X,A1,80A1)')A2,QPL
      GOTO 160
130   WRITE(10,'(7X,21X,A1,80A1)')A2,QPL
      GOTO 160
140   WRITE(10,'(7X,28X,A1,80A1)')A2,QPL
160   CONTINUE
      DO 500 I=1,N
      DO 170 J=1,80
170   QPL(J)=' '
      DO 250 J=1,M
      IF(K.EQ.1.AND.J.EQ.1)GOTO 230
C     ————流 量—————
      KY=INT(DATA(J,I)/RATE+0.5)
      QPL(KY)=A1(J)

```

```

        GOTO 247
230  CONTINUE
C              降    水          
        KY=DATA(J,I)
        IF(KY.GE.79)THEN
        DO 240  K1=1,79
240  QPL(K1)='='
        QPL(80)='>'
        ELSE
        DO 245  K1=1,KY
245  QPL(K1)='='
        ENDIF
247  CONTINUE
250  CONTINUE
        GO TO(310,320,330,340),M
310  WRITE(10,410)I,(DATA(J,I),J=1,M),A2,QPL
        GOTO 360
320  WRITE(10,420)I,(DATA(J,I),J=1,M),A2,QPL
        GOTO 360
330  WRITE(10,430)I,(DATA(J,I),J=1,M),A2,QPL
        GOTO 360
340  WRITE(10,440)I,(DATA(J,I),J=1,M),A2,QPL
360  CONTINUE
410  FORMAT(I5,2X,F6.0,1X,A1,80A1)
420  FORMAT(I5,2X,2(F6.0,1X),A1,80A1)
430  FORMAT(I5,2X,3(F6.0,1X),A1,80A1)
440  FORMAT(I5,2X,4(F6.0,1X),A1,80A1)
500  CONTINUE
        RETURN
        END

```

### 3. 程序说明

变量	类型	大小	I/O	内 容
<i>M</i>	整型	1	<i>I</i>	要打印的时间序列个数
<i>N</i>	整型	1	<i>I</i>	每个时间序列的历时(时段)
<i>DATA</i>	实型	$4 \times N$	<i>I</i>	时间序列值
<i>K</i>	整型	1	<i>I</i>	时间序列中有无降雨过程的标志。

$K=1$ , *DATA* 中第一个时间序列是降水; 否则, *DATA* 中全为流量序列

注: 打印的图象过程在 PLT.OUT 文件内。

## 十二、径流量、输沙量的计算

### 1. 功能

根据实测的流量和含沙量资料, 计算指定日期内的径流量和输沙量。

### 2. 计算程序

```

SUBROUTINE HQS(YEAR,IM,ID,IH,EM,ED,EH,
+ N,MDH,OBQS,QW,SW,OBT,OB,OBS)
INTEGER YEAR,EM,ED,EH
DIMENSION MDH(4,N),OBQS(2,N),OBT(N),OB(N),OBS(N)
DIMENSION MON(12),MONDAY(12)
DATA MON/31,28,31,30,31,30,31,31,30,31,30,31/
DATA MONDAY/0,31,59,90,120,151,181,212,243,
+ 273,304,334/
IF(YEAR/4*4.NE.YEAR)GOTO 20
MON(2)=29
DO 10 I=3,12
10 MONDAY(I)=MONDAY(I)+1

```

```

20  CONTINUE
    TI=(MONDAY(IM)+ID-1)*24.+IH
    TE=(MONDAY(EM)+ED-1)*24.+EH
    II=0
    DO 50  I=1,N
      IF(OBQS(1,I).GE.0)THEN
        II=II+1
        IMM=MDH(1,I)
        IDD=MDH(2,I)
        IHH=MDH(3,I)
        IFF=MDH(4,I)
        OBT(II)=(MONDAY(IMM)+IDD-1)*24.+IHH+IFF/60.
        OB(II)=OBQS(1,I)
      ELSE
        ENDIF
50  CONTINUE
    K=II
    IF(K.EQ.1)THEN
      QI=OB(1)
      QE=OB(1)
    ELSE
      CALL ZXCZ(K,OBT,OB,0,TI,QI)
      CALL ZXCZ(K,OBT,OB,0,TE,QE)
    ENDIF
    DO 80  I=1,N
      IF(OBQS(1,I).LT.0)THEN
        II=II+1
        IMM=MDH(1,I)
        IDD=MDH(2,I)
        IHH=MDH(3,I)
        IFF=MDH(4,I)

```

```

T=(MONDAY(IMM)+IDD-1)*24.+IHH+IFF/60.
CALL ZXCZ(K,OB,T,0,T,Q)
OBT(I)=T
OB(I)=Q
ELSE
ENDIF
80  CONTINUE
    I1=0
    I2=0
    DO 90  I=1,N
    IF(OBQS(1,I).GE.0)THEN
    I1=I1+1
    OBS(I)=OB(I1)
    ELSE
    I2=I2+1
    OBS(I)=OB(K+I2)
    ENDIF
90  CONTINUE
    DO 100  I=1,N
    IMM=MDH(1,I)
    IHH=MDH(3,I)
    IFF=MDH(4,I)
    OBT(I)=(MONDAY(IMM)+MDH(2,I)-1)*24.+IHH
    + IFF/60.
100  CONTINUE
    DO 130  I=1,N
    IF(OBT(I).GE.TI)GOTO 135
130  CONTINUE
135  K1=I
    DO 140  I=1,N
    J=N-I+1

```

```

        IF(OBT(J).LE.TE)GOTO 145
140  CONTINUE
145  K2=J
        IF(K1.GE.N.OR.K2.LE.1)THEN
            QW=0
            SW=0
            GOTO 500
        ELSE
            ENDIF
            QW=0
            T1=TI
            Q1=QI
            DO 180  I=1,N
                IF(I.LE.K1)GOTO 160
                IF(I.GE.K2)GOTO 190
                T2=OBT(I)
                Q2=OBS(I)
                 $QW=QW+(T2-T1)*0.36*(Q1+Q2)*0.5*0.0001$ 
                T1=T2
                Q1=Q2
160  CONTINUE
180  CONTINUE
190  T2=TE
            Q2=QE
             $QW=QW+(T2-T1)*0.36*(Q1+Q2)*0.5*0.0001$ 
C  -----
        II=0
        DO 200  I=1,N
            IF(OBQS(2,I).GE.0)THEN
                II=II+1
                OBT(II)=OBT(I)

```

```

      OB(II)=OB(I) * OBQS(2,I) * 0.001
      ELSE
      ENDIF
200  CONTINUE
      K=II
      IF(II.EQ.0)THEN
      SW=0
      GOTO 500
      ELSE
      IF(II.EQ.1)THEN
      SI=OB(1)
      SE=OB(1)
      ELSE
      CALL ZXCZ(K,OBT,OB,0,TI,SI)
      CALL ZXCZ(K,OBT,OB,0,TE,SE)
      ENDIF
      ENDIF
      DO 230 I=1,K
      IF(OBT(I).GE.TI)GOTO 235
230  CONTINUE
235  K1=I
      DO 240 I=1,K
      J=K-I+1
      IF(OBT(J).LE.TE)GOTO 245
240  CONTINUE
245  K2=J
      IF(K1.GE.K.OR.K2.LE.1)THEN
      SW=0
      GOTO 500
      ELSE
      ENDIF

```

```

      SW=0
      T1=T1
      S1=S1
      DO 280 I=1,K
      IF(I.LE.K1)GOTO 260
      IF(I.GE.K2)GOTO 290
      T2=OBT(I)
      S2=OB(I)
      SW=SW+(T2-T1)*0.36*(S1+S2)*0.5
      T1=T2
      S1=S2
260  CONTINUE
280  CONTINUE
290  T2=TE
      S2=SE
      SW=SW+(T2-T1)*0.36*(S1+S2)*0.5
500  CONTINUE
      RETURN
      END

```

### 3. 程序说明

变量	类型	大小	I/O	内 容
YEAR	整型	1	I	计算的年份
IM	整型	1	I	统计水量和沙量的开始月
ID	整型	1	I	统计水量和沙量的开始日
IH	整型	1	I	统计水量和沙量的开始时
EM	整型	1	I	统计水量和沙量的结束月
ED	整型	1	I	统计水量和沙量的结束日
EH	整型	1	I	统计水量和沙量的结束时
N	整型	1	I	实测数据的个数



<i>MDH</i>	整型	$4 \times N$	<i>I</i>	分别表示每个实测数据对应的月、日、时、分。
<i>OBQS</i>	实型	$2 \times N$	<i>I</i>	分别表示实测的流量( $\text{m}^3/\text{s}$ )、含沙量( $\text{kg}/\text{m}^3$ ), 缺测值用负值表示。
<i>QW</i>	实型	1	<i>O</i>	指定时间内的径流量( $\times 10^8 \text{m}^3$ )
<i>SW</i>	实型	1	<i>O</i>	指定时间内的输沙量( $\times 10^4 \text{t}$ )
<i>OBT</i>	实型	<i>N</i>		中间变量
<i>OB</i>	实型	<i>N</i>		中间变量
<i>OBS</i>	实型	<i>N</i>		中间变量

注: 内部调用了直线插值程序 ZXCZ。

### 十三、流量序列的特征值统计

#### 1. 功能

用于统计流量序列的最大值和最小值及其它们出现的时间、最大若干天径流量、某一流量值以上的水量。

#### 2. 计算程序

```

SUBROUTINE TZTJ(YEAR,N,MDH,OBQ,QMAX,TMAX,
+ QMIN,TMIN,NTD,WN,IT,GQ,GQW,NGQH,OBT,M,
+ TIME,TQ)
INTEGER * 4 TMAX,TMIN,YEAR,TIME
DIMENSION MDH(4,N),OBQ(N),OBT(N),TIME(M),TQ(M),
+ IT(2)
DIMENSION MON(12),MONDAY(12)
DATA MON/31,28,31,30,31,30,31,31,30,31,30,31/
DATA MONDAY/0,31,59,90,120,151,181,212,243,

```

```

+ 273,304,334/
  IF (YEAR/4 * 4. NE. YEAR) GOTO 20
  MON(2) = 29
  DO 10 I = 3, 12
10  MONDAY(I) = MONDAY(I) + 1
20  CONTINUE
    QMAX = OBQ(1)
    I1 = 1
    QMIN = OBQ(1)
    I2 = 1
    DO 150 I = 2, N
      QI = OBQ(I)
      IF (QMAX. GE. QI) GOTO 130
      QMAX = QI
      I1 = I
130  CONTINUE
      IF (QMIN. LE. QI) GOTO 140
      QMIN = QI
      I2 = I
140  CONTINUE
150  CONTINUE
      TMAX = MDH(1, I1) * 100000 + MDH(2, I1) * 1000
      + + MDH(3, I1) * 10 + MDH(4, I1) / 6.
      TMIN = MDH(1, I2) * 100000 + MDH(2, I2) * 1000
      + + MDH(3, I2) * 10 + MDH(4, I2) / 6.
      DO 210 I = 1, N
        IMM = MDH(1, I)
        IDD = MDH(2, I)
        IHH = MDH(3, I)
        IFF = MDH(4, I)
        OBT(I) = (MONDAY(IMM) + IDD - 1) * 24. + IHH + IFF / 60.

```

```

210  CONTINUE
      IM=MDH(1,1)
      ID=MDH(2,1)
      IH=MDH(3,1)
      IF(MDH(4,1).EQ.0)GOTO 215
      IH=IH+1
      IF(IH.LT.24)GOTO 215
      IH=IH-24
      ID=ID+1
      IF(ID.LE.MON(IM))GOTO 215
      ID=ID-MON(IM)
      IM=IM+1
215  CONTINUE
      IEM=MDH(1,N)
      IED=MDH(2,N)
      IEH=MDH(3,N)
      ND=MONDAY(IEM)-MONDAY(IM)+IED-ID
      M=ND*24+IEH-IH+1
      IMM=IM
      IDD=ID
      IHH=IH
      KK=0
      IF(IHH.NE.0)GOTO 218
      KK=1
      KDI=1
218  CONTINUE
      DO 230 I=1,M
      IF(IHH.LT.24)GOTO 220
      IF(KK.EQ.0)KDI=I
      KK=KK+1
      IHH=IHH-24

```

```

      IDD=IDD+1
      IF (IDD. LE. MON (IMM)) GOTO 220
      IDD=IDD--MON (IMM)
      IMM=IMM+1
220  CONTINUE
      T=(MONDAY (IMM)+IDD-1)*24.+IHH
      CALL ZXCZ(N,OBT,OBQ,1,T,Q)
      TQ(I)=Q
      TIME(I)=IMM*100000+IDD*1000+IHH*10
      IHH=IHH+1
230  CONTINUE
250  KND=(M-KDI+1)/24--NTD+1
      IF (KND. LE. 0) THEN
        WN=0
        IT(1)=0
        IT(2)=0
      ELSE
        M1=KDI
        M2=24*NTD+M1-1
        DO 270 I=1,KND
          W=0
          DO 260 J=M1,M2
            W=W+TQ(J)*0.36
260  CONTINUE
          OBT(I)=W
          M1=M1+24
          M2=M2+24
270  CONTINUE
          DO 275 I=1,KND
275  OBT(I)=OBT(I)*0.0001
      II=1

```

```

      WN=OBT(1)
      DO 290 I=1,NND
      Q=OBT(I)
      IF(WN.GE.Q)GOTO 280
      WN=Q
      I1=I
280  CONTINUE
290  CONTINUE
      I1=KDI+(I1-1)*24
      I1=TIME(I1)
      IT(1)=I1/100000
      I1=MOD(I1,100000)
      IT(2)=I1/1000
      ENDIF
300  GQW=0
      NGQH=0
      DO 350 I=1,M
      IF(TQ(I).GT.GQ)THEN
      NGQH=NGQH+1
      GQW=GQW+(TQ(I)-GQ)*0.36
      ELSE
      ENDIF
350  CONTINUE
      GQW=GQW/10000.
      RETURN
      END

```

### 3. 程序说明

变量	类型	大小	I/O	内 容
YEAR	整型	1	I	计算的年份
N	整型	1	I	流量序列的个数

<i>MDH</i>	整型	$4 \times N$	<i>I</i>	流量序列对应的月、日、时、分
<i>OBQ</i>	实型	$N$	<i>I</i>	流量序列
<i>QMAX</i>	实型	1	<i>O</i>	最大流量值( $\text{m}^3/\text{s}$ )
<i>TMAX</i>	整型	1	<i>O</i>	<i>QMAX</i> 对应的时间 形式为: <i>MMDDHHF</i> 其中 <i>MM</i> ——月; <i>DD</i> ——日; <i>HH</i> ——时; <i>F</i> ——分(十进制)
<i>QMIN</i>	实型	1	<i>O</i>	最小流量值( $\text{m}^3/\text{s}$ )
<i>TMIN</i>	整型	1	<i>O</i>	<i>QMIN</i> 对应的时间,表示形式 同 <i>TMAX</i>
<i>NTD</i>	整型	1	<i>I</i>	要统计的最大若干天水量的天 数。如要统计最大 3 天水量,此 值为 3
<i>WN</i>	实型	1	<i>O</i>	最大 <i>NTD</i> 天的水量( $\times 10^8 \text{m}^3$ )
<i>IT</i>	整型	2	<i>O</i>	最大 <i>NTD</i> 天水量起始月、日
<i>GQ</i>	实型	1	<i>I</i>	给定的标准流量
<i>GQW</i>	实型	1	<i>O</i>	<i>GQ</i> 流量以上的水量( $\times 10^8 \text{m}^3$ )
<i>NGQH</i>	整型	1	<i>O</i>	大于 <i>GQ</i> 流量值的历时(h)
<i>OBT</i>	实型	$N$		内部使用数组
<i>M</i>	整型	1	<i>O</i>	整个流量过程的历时小时数
<i>TIHE</i>	整型	$M$		内部数组
<i>TQ</i>	实型	$M$		内部数组

注:内部调用了直线程序 ZXCZ。

## 附 录

### 黄河洪水预报及调度系统简介

#### 一、系统综述

黄河洪水预报及调度系统(YRFFCS)容纳了前面各章介绍的所有算法程序。除此之外,还包括了一些适合于黄河特点的专用算法,是微机上开发的一个集洪水预报、洪水调度、参数率定、实用程序为一体的水文软件包。


YRFFCS 采用了模型、方法、数据的分离和计算机编程覆盖技术,可通过调用方法库中不同方法和数据库中的各类数据组合成许多作业,方便地进行下述工作:①实时暴雨洪水预报;②设计洪水计算;③历史洪水放大模拟;④防洪工程的实时调度和仿真;⑤率定模型参数;⑥水文特征值统计、绘图等。另外,还可根据用户需要,加入新方法,进行其它功能的运算。

经过几年的实践证明,黄河洪水预报及调度系统具有以下特点:①通用性强、灵活、易扩充;②输入、输出多路径;③交互性强而灵活,便于使用;④有较强的实时校正能力;⑤运算速度快。系统的结构设计符合计算机应用软件开发规范,代表了软件工程学的发展趋势。

#### 二、数据库及其管理

数据库由固定参数库、时变参数库、历史数据典型库、实时数据库和临时数据库五个子数据库组成。各子库的结构和功能是:  
①固定参数库:存贮模型固定参数,它采用了层次结构,数据库

的访问通过方法代码和区域代码两组关键字来实现。②时变参数库：存贮模型的状态参数，其结构与固定参数库相同。数据的访问通过方法代码、区域代码和日期三组关键字来实现。③历史数据典型库：存贮一些典型的历史洪水数据，该库由数据文件和相应的索引文件构成。数据的访问通过站号、数据类型、开始时间、结束时间、时间间隔五组关键字实现。④临时数据库：临时数据库的结构和访问方式同历史数据典型库，它的作用是存贮中间计算结果和一些其它临时数据。⑤实时数据库：存贮实时日雨量、时段雨量、水位、流量、遥测雨量等数据，它建立在共享服务器上，访问方式同历史数据典型库。除了以上五类数据库外，数据还可存取于指定的数据文本文件。



附图—1 数据库管理框图

对于以上五个数据库，除实时数据库是单独管理以外，其它四个数据库采用了统一管理。数据库管理模块的功能包括数据库初始化、数据增加、数据修改、数据检索、数据打印、数据删除等（见附图—1）。

### 三、方法库及其管理

方法是为实现一个或几个目标的具体算法程序。YRFFCS 方法库中的方法分为两类，即通用方法和专用方法。

通用方法可用于任何流域，方法本身不需要改变。专用方法是描述某区特别属性而编制的一些专用算法。方法库中的每个方法都包含输入模块、核心模块、输出模块三部分。输入模




块主要功能是根据模型需要，从指定处获得数据，然后把这些数据正确地赋给核心模块中的诸变量。核心模块是为解决某一具体问题的算法程序，是方法的实体。该程序设计很通用，模块化程度高。输出模块的功能是把核心模块计算的结果输出到指定位置，以备其它方法使用和检索。显然，方法库中的方法之间是相互独立的。目前，YRFFCS 的方法库中已装有萨克模型、新安江（三水源）模型等 20 多个算法，方法库中各种方法及其方法编号、代码等有关信息如附表—1 所示。随着系统的发展，还可不断地加入新方法。

方法库是许多方法的集合。方法库的管理是模仿数据库的思想，对库内的元件进行检索、增加、删除、修改（见附图—2）。不同之处在于数据库是对数据进行管理，而方法库管理对象是程序文件。这些文件可以是源程序、目标文件，也可以是执行文件。考虑到微机的特点，YRFFCS 采用了目标文件入库的方式，并使用了覆盖技术。

#### 四、模型库及其管理

模型就是为解决某一实际问题，通过调用方法库中的有关算法和数据

库中的相应数据而组合成的作业。不同的区域，模型亦不同。即使对于同一区域，由于采用的方法不同，组合成的模型也不一样，所以，YRFFCS 中的方法构成了许多模型。把一些常用模型进行统一管理，形成模型库。现行的 YRFFCS 中的模型库里的模型包括①预报模型：它包括三门峡以上的五个洪水预报模型（吴堡至龙门的预报，临潼至华县的预报，




附图—2 方法库管理框图

附表—1

YRFFCS 方法库内方法索引表

方法编号	方法代码	方法名称	说明
1. 01	FXAJ	新安江模型	通用方法
1. 02	FSAC	萨克拉门托模型	通用方法
1. 03	FTAN	水箱模型	通用方法
1. 04	FSAB	陕北模型	通用方法
1. 05	FPAR	降雨径流相关图法	通用方法
1. 06	FMSK	马斯京根连续演算法	通用方法
1. 07	FLYM	分层马斯京根法	通用方法
1. 08	FNLM	非线性马斯京根法	通用方法
1. 09	FRSV	水库调洪计算法	通用方法
1. 10	FDJQ	反馈模拟实时校正	通用方法
1. 11	FADS	加/减计算法	通用方法
2. 01	FWBL	吴堡至龙门预报方法	专用方法
2. 02	FLNH	临潼至华县预报方法	专用方法
2. 03	FLHZ	龙、华、河、湫至潼关预报方法	专用方法
2. 04	FTCL	伊洛河夹滩地区处理方法	专用方法
2. 05	FDPH	东平湖水库分洪计算方法	专用方法
3. 01	UMAP	面平均雨量计算	通用方法
3. 02	UHQS	径流量、输沙量计算	通用方法
3. 03	UEXQ	等时间间隔流量(水位)序列生成	通用方法
3. 04	UEXP	等时间间隔雨量序列生成	通用方法
3. 05	UMXN	求最大最小值	通用方法
3. 06	UPLT	高分辨率过程线显示	通用方法
3. 07	UPFG	打印降水、流量、水位过程线	通用方法
3. 08	UWTC	流量过程线特征值统计	通用方法
3. 09	UHQC	水位流量相互转换	通用方法
3. 10	UERR	误差统计与分析	通用方法
3. 11	UADD	流量合成法	通用方法




附图—3 模型管理系统

龙门、华县、河津、猗头至潼关的预报,非线性马法预报,三门峡水库出库预报);三门峡至花园口区间的八个洪水预报模型(陆浑水库入库与出库预报,故县水库入库与出库预报,三门峡至花园口区间的产汇流预报,三门峡至小浪底区间的产汇流预报,陆浑、故县、三门峡、润城至花园口区间的产汇流预报,龙门镇、白马寺、小浪底、五龙口、山路平至花园口的预报,黑石关、小浪底、武陟至花园口的预报,非线性马法预报);花园口以下的四个洪水预报模型(洪峰相关法,蓄率中线法,非线性马法,分层马法)。

②调度模型:它包括三门峡、故县、陆浑三库联合调度模型,黄河下游东平湖等防洪工程的联合调度模型。

③率定模型:包括人工率定和自动优选两种率定模型。

④实用模型:实用模型是YRFFCS的一个重要分支,也是预报、调度模型的和率定模型的辅助模块,它包括面平均雨量计算和误差统计等实用模型(见附图一3)。



附图一4 模型库、方法库、数据库三者的关系

限于当前计算机的理论水平,目前模型的增加、删除、修改等管理,都是通过编辑系统完成的,然后加入模型库管理文件。通过检索了解模型库内现有模型的情况,它包括模型编号、模型名称、模型输入文件名、模型输出文件名等信息。

### 五、三库间的关系

YRFFCS之所以灵活方便,功能齐全,主要是采用了模型、方法、数据的分离和计算机编程的覆盖技术。实现了系统内加入新方法基本上不增加机器内存,模型库、方法

库、数据库之间既相互独立、又相互依赖。打个形象的比喻，方法如砖头，数据似水泥，模型则是由砖头和水泥搭成的各类不同建筑物（见附图—4）。

## 参 考 文 献

- [1] 赵人俊. 流域水文模型——新安江模型与陕北模型. 北京: 水利电力出版社, 1984
- [2] 庄一筠, 林三益. 水文预报. 北京: 水利电力出版社, 1986
- [3] 张文华. 实用暴雨洪水预报理论与方法. 北京: 水利电力出版社, 1990
- [4] 翟家瑞. 用分层马斯京根法作河道洪水演算. 人民黄河, 1992(7)
- [5] 翟家瑞. 马斯京根法几种不同应用形式之浅析. 人民黄河, 1994(4)
- [6] 长江流域规划办公室主编. 水文预报方法. 北京: 水利出版社, 1982
- [7] 孙桂华. 实用水文预报中反馈模拟实时校正的应用. 水文, 1991(1)
- [8] 翟家瑞. 流域单位线的推求方法——分析试错法. 水文, 1982(5)
- [9] 童明生. IBM PC FORTRAN77 程序设计语言. 北京: 北京科技出版社, 1989
- [10] 谭浩强, 田淑清. FORTRAN 语言. 北京: 清华大学出版社, 1983

前言  
目录

目 录

前言

第一章新安江（三水源）模型

第一节模型结构

一、蒸散发计算

二、产流量计算

三、水源划分

四、汇流计算

五、不同时段长度的转化

第二节模型计算程序

一、计算程序

二、程序说明

第三节实例

一、主程序

二、实例说明

第二章萨克拉门托模型

第一节模型结构

一、产流量计算

二、蒸散发计算

三、水分交换计算

四、下渗计算

五、模型参数的意义

六、几个主要参数对洪水的影响

第二节模型计算程序

一、计算程序

二、程序说明

第三节实例

一、主程序

二、实例说明

第三章水箱模型

第一节模型结构

第二节模型计算程序

一、计算程序

二、程序说明

第三节实例

一、主程序

二、实例说明

第四章陕北模型

第一节模型结构

一、点降雨径流模型

二、降雨和下垫面空间分布不均匀的考虑

三、陕北模型推流步骤

第二节模型计算程序

	一、 计算程序
	二、 程序说明
第三节实例	
	一、 主程序
	二、 实例说明
第五章降雨径流相关图法	
第一节降雨径流相关图	
	一、 概述
	二、 $P \sim R$ 的计算
	三、 相关图推流计算
第二节 $P \sim P \sim R$ 计算程序	
	一、 $P \sim P \sim R$ 程序
	二、 程序说明
第三节实例	
	一、 主程序
	二、 实例说明
第六章水库调洪演算	
第一节计算方法	
第二节水库调洪计算程序	
	一、 计算程序
	二、 程序说明
第三节实例	
	一、 主程序
	二、 实例说明
第七章马斯京根法	
第一节基本马法	
第二节分段马法	
	一、 方法介绍
	二、 分段马法计算程序
	三、 实例
第三节变参数马法	
	一、 方法介绍
	二、 变参数马法计算程序
	三、 实例
第四节非线性马法	
	一、 非线性马法计算公式
	二、 非线性马法计算程序
	三、 实例
第五节分层马法	
	一、 方法简介
	二、 方法特点
	三、 分层马法计算程序
	四、 实例
第六节讨论	
	一、 各方法间的关系



	二、区间来（引）水的处理
	三、马法的预见期和预报误差
第八章	纳须单位线法
	第一节 基本概念
	一、纳须瞬时单位线
	二、纳须时段单位线
	三、纳须单位线 $n$ 、 $k$ 的计算
	第二节 计算程序
	一、求 $n$ 和 $k$ 的计算程序
	二、纳须时段单位线的计算程序
	第三节 实例
第九章	反馈模拟实时校正
	第一节 模型简介
	第二节 实时校正计算程序
	一、计算程序
	二、程序说明
	第三节 实例
第十章	其它常用水文计算子程序
	一、直线插值
	二、抛物线插值
	三、水位与流量的换算
	四、单位线推流计算
	五、时段单位线的转换
	六、马斯京根演算参数 $k$ 与 $x$ 的计算
	七、马斯京根汇流系数的计算
	八、误差统计与分析
	九、等时段流量（或水位）序列的生成
	十、等时段雨量序列的生成
	十一、打印过程线
	十二、径流量、输沙量的计算
	十三、流量序列的特征值统计
附录	黄河洪水预报及调度系统简介
	一、系统综述
	二、数据库及其管理
	三、方法库及其管理
	四、模型库及其管理
	五、三库间的关系
参考文献	